

تم التطوير من يوليو 2009 حتى مايو 2010  
من أعمال مهندس البرمجيات  
محمود سمير فايد

# لغة البرمجة سوبرنوبا

## Supernova Programming Language

الإصدار 1.3 Version

منتج حر مفتوح المصدر

Free-Open Source

زهرة من صناعة البرمجيات العربية في عام 2010  
تحصل على نسختك الكاملة منها مجانا

<http://supernova.sourceforge.net>

# إهداء

إلى والدتي الحبيبة (اجمل واطيب ماما فى الدنيا كلها)  
التي لو خرجت من الدنيا برضاها فاني قد اعطيت اكثر مما اتمنى

إلى والدي الحبيب المهندس سمير (بابا حبيبي وحشتني اوى اوى)  
مصدر ما انا فيه من سعادة وامان والرجل الذي اطمح ان اكون مثله

الى اخي الغالى المهندس محمد (وحشتني جدا جدا)  
استاذي الكبير فهو الذي علمني باسلوب التشجيع والمنافسة الشريفة

الى اخي الحبيب المهندس احمد (ربنا يرجعك لينا بالسلامة ان شاء الله)

الى اخي الغالى المهندس ابراهيم (تحب ناكل ايه النهارده يا هيما؟)

الى ابن عمي الغالى الدكتور فادي سعيد (ماتبقاش تختفى كثير يامان)

الى ابن عمي الحبيب الدكتور شادي سعيد (وحشني الخناق معاك ياراجل)

الى صديقي المخلص ضابط مهندس ماهر القط ( انا بس بصبح ياباشا!!)

الى اصدقائي العرب فى فريق تقنية البرمجة بدون كود  
الاستاذ سامح كامل و الاستاذ ابو السعود عبدالرؤف من مصر  
الاخ الحبيب امين دبابسية – والاخ الحبيب جردى مسعود من الجزائر

الى كل راغب فى العلم ومحب للمعرفة ويطمح ان يساهم فى الرقى  
بالمستوى المعرفى للمجتمع العربى الاسلامى على الصعيد الدولى

# بِسْمِ اللّٰهِ الرَّحْمٰنِ الرَّحِیْمِ

اللهم اجعل عملي خالصا لوجهك الكريم يارب العرش العظيم  
واجعلني من عبادك المخلصين الصالحين وانفع بي امة الاسلام  
انك على كل شيء قدير وانك انت يارحمن يارحيم اكرم الاكرمين

اللهم امين يارب العالمين



## الفهرس

الصفحة	الموضوع	مسلسل
5	Introduction to Supernova مدخل إلى السوبرنوفات	1
18	Expressions التعبيرات	2
35	Do Statements جمل التنفيذ	3
43	Date and Time الوقت والتاريخ	4
56	Strings السلاسل الحرفية	5
93	Arrays المصفوفات	6
100	Files الملفات	7
110	Delivery Application تطبيق الديلفري	8
116	Macro الماكرو او المختصر	9
124	DLL Functions دوال مكتبات الربط الديناميكية	10
127	Sound Player Library مكتبة تشغيل الصوت	11
130	Sound Recorder Library مكتبة تسجيل الصوت	12
133	Movie Player Library مكتبة تشغيل الفيديو	13
136	Programming With Fiction Description البرمجة بوصف الخيال	14
177	GUI واجهة المستخدم الرسومية	15
246	ADO و مثال يستعمل الـ COM إستعمال تقنية الـ	16
252	تجربة العمل على Notepad++ لكتابة الشفرة المصدرية	17
255	رسالة من المؤلف	18





## مدخل إلى السوبرنوبا

### Introduction to Supernova

لمدة سنوات كان هناك خلاف كبير في المنتديات العربية حول بناء لغة برمجة عربية الهوية من ناحية البناء والاستخدام بمعنى ان يتم بنائها من قبل مبرمجين عرب وان تكون الشفيرة المصدرية (الكود البرمجي المكتوب بهذه اللغة) باللغة العربية.

السوبرنوبا لغة برمجة عربية الهوية من ناحية البناء اما من ناحية الاستخدام فهي تحقق المعادلة الصعبة حيث تتيح ان تكون الشفيرة المصدرية باللغة العربية او باللغة الانجليزية او خليط من اللغة العربية والانجليزية مما يعنى الحفاظ على الهوية العربية وفي نفس الوقت التواصل على النطاق العالمى من خلال اللغة الانجليزية

السوبرنوبا اضافة الى العلم وبداية من حيث انتهى الآخرون وليست اعادة اختراع للعجلة فهي لغة تم بنائها باستخدام اساليب الذكاء الاصطناعى لتكون احدى لغات البرمجة من الجيل الخامس فهي تصنف ضمن فرع معالجة اللغات الطبيعية وتعد اضافة ملموسة الى هذا الفرع من العلم فهي لغة قريبة جدا الى لغة الانسان بنسبة كبيرة تفوق لغات البرمجة الاخرى المتعارف عليها قبل بناء السوبرنوبا.



## ما هي التقنية المستخدمة في بناء لغة السوبرنوبا ؟

تم تطوير السوبرنوبا باستخدام تقنية البرمجة بدون كود PWCT وهي منتج عربي مجاني مفتوح المصدر يمكن ان تحصل على نسختك منه مجاناً من خلال الرابط التالي <http://doublesvsoop.sourceforge.net>

## ما هي الملامح الاساسية في لغة البرمجة سوبرنوبا ؟

- لغة برمجة من الجيل الخامس اكثر اجيال لغات البرمجة قربا الى لغة الانسان و يضم هذا الجيل اللغات الطبيعية وانظمة الذكاء الاصطناعي
- لغة صغيرة الحجم مساحتها تحت 10 ميجا بايت
- تقدم مفهوم البرمجة بوصف الخيال Programming With Fiction Description
- توفر مرونة عالية في الكتابة Writing Flexibility
- تتسم بأنها تقبل تغير اللغة (عربي/انجليزي/خليط)
- تدعم بناء تطبيقات بواجهة رسومية GUI Applications تعمل تحت منصة مايكروسوفت ويندوز MS-Windows
- تتيح توزيع التطبيقات على صورة ملفات تنفيذية EXE
- تتعامل مع المتغيرات بدون تعريف مسبق وتسد لها قيم ابتدائية بشكل تلقائي حسب سياق الاستخدام.
- لاتجعلك تهتم بنوع المتغير (حرفي/رقمي...إلخ) وتختار المناسب حسب سياق استخدام المتغير.
- تشتمل على تعليمات لمعالجة السلاسل الحرفية بمرونة.
- تشتمل على تعليمات للتعامل مع الوقت والتاريخ.
- تشتمل على تعليمات للتعامل مع العمليات الحسابية والمنطقية.
- تشتمل على تعليمات للتفرع المشروط والتكرار.
- تتيح التعامل مع المصفوفات ذات الابعاد والاحجام المختلفة.
- تتيح تعريف دوال جديدة Functions او احداث جديدة Events.
- تتيح التعامل مع الملفات بانشائها وقرائنها وكتابتها وتتيح تشغيل البرامج الخارجية.
- تتيح استدعاء دوال في مكتبات ربط ديناميكية DLL.
- تشتمل على مكتبات جاهزة للاستخدام مثل مكتبة تشغيل الصوت ومكتبة تسجيل الصوت ومكتبة تشغيل الفيديو.

# برنامج Hello World

عند تقديم لغات البرمجة لأول مرة فانه من الشائع ان يتم كتابة برنامج ترحيبي باستخدام اللغة وشرح طريقة تشغيله حتى يتعرف الدارس على طريقة استعمال اللغة بشكل مبسط وهذا ماسوف نقوم به الان باذن الله رب العالمين.

بداية يتم كتابة الشفرة المصدرية في ملف يحمل الامتداد Nova وليكن مثلا اسم الملف Hello.nova ويحتوى على الشفرة المصدرية التالية

Hello.nova

**I want window and the window title is Hello World.**

نلاحظ وجود نقطة (.) فى نهاية Hello World لتبين انها نهاية البيانات.

يمكن ان يتم كتابة الشفرة المصدرية باللغة العربية وليكن مثلا كالتالى

انا اريد نافذة والنافذة تحت عنوان هو مرحبا.

لتشغيل البرنامج نقوم بتشغيل مفسر ومترجم لغة السوبرنوفانا Nova.exe ثم نختار الملف Hello.nova (تم التطبيق على المثال الاول الذى يحتوى على شفرة مصدرية باللغة الانجليزية وعنوان النافذة Hello World )



## ماذا يفعل Nova.exe عند تشغيله ؟


يتيح لنا اختيار ملف من نوع Nova او Star لتشغيله فى حالة كان الملف من النوع Nova (ملف يشتمل على الشفرة المصدرية) فانه يتم عمل ترجمة لهذه الشفرة Compile ثم يتم توليد ملف مترجم يحمل الامتداد Star ويتم تشغيل البرنامج بعد عملية الترجمة بشكل تلقائى. فى حالة كان الملف من النوع Star (ملف مترجم جاهز للتنفيذ) فانه يتم عمل تشغيل للبرنامج مباشرة من خلال هذا الملف ويتم توفير وقت الترجمة حيث يتم التشغيل مباشرة مما يعطى سرعة انطلاق فى تشغيل التطبيقات.

## هل هناك طريقة اسرع لتشغيل البرامج ؟

نعم حيث يمكن تحديد Nova.exe على انه التطبيق الافتراضى لفتح الملفات من النوع Nova او Star وبالتالي عند الضغط مرتين على اى ملف من النوع Nova او من النوع Star يتم فتحه بواسطة Open With مفسر ومترجم اللغة بشكل تلقائى فمثلا يتم تحديد اى ملف من النوع Nova وليكن question.nova ثم يتم الضغط بالزر الايمن للفارة على هذا الملف ثم يتم اختيار فتح بواسطة Open With ثم اختيار برنامج Choose Program... ثم نختار Nova.exe









### كيف اقوم بتوزيع برامجي على صورة ملفات تنفيذية EXE ؟

- يتم تشغيل البرنامج باستخدام مفسر اللغة لتوليد ملف Star مترجم
- يتم انشاء مجلد جديد ونضع فيه ملف Star مع Nova.exe
- يتم اعادة تسمية ملف Star و Nova.exe ليكونوا بنفس الاسم
- نفترض ان لدينا برنامج Flash.nova ونريد توزيع هذا البرنامج على صورة ملف Flash.exe للتشغيل مباشرة من قبل المستخدم.
- يتم تشغيل Flash.nova وبالتالي نحصل على ملف Star بالاسم Flash.nova.star
- يتم انشاء مجلد جديد وليكن Myapp ونضع فيه Nova.exe و Flash.nova.star ويتم اعادة تسمية Nova.exe الى Flash.exe و Flash.nova.star الى Flash.star



**Generated Object File  
(Flash.nova.star)**



## كيف اقوم بكتابة الشفيرة المصدرية لبرامجى باستخدام السوبرنوبا ؟

تتكون البرامج من مجموعة من القطع Paragraphs والتي تتكون من مجموعة من الجمل Sentences والتي هي عبارة عن مجموعة من الكلمات Words المرتبة تبعا لقواعد Rules والتي تكون عبارة عن ترتيب تصنيفات Classification الكلمات معا والتي تصنف الى نطاقات مختلفة حيث يشتمل كل نطاق مجموعة من الكلمات.

مثلا القاعدة PVC تعنى Pronoun – Verb – Class حيث Pronoun هو الضمير و Verb هو الفعل و Class هو الفئة و كل من P و V و C ثلاث نطاقات لتصنيف الكلمات Words والنطاق P يحتوى على ( I ) بالانجليزية و ( انا ) بالعربية والنطاق V يحتوى على ( Want ) بالانجليزية و ( اريد ) بالعربية والنطاق C يحتوى على ( Window ) بالانجليزية و ( نافذة ) بالعربية

وبالتالى فان الامر

I want window

وكذلك الامر

انا اريد نافذة

كل منهما يتبع القاعدة PVC اي Pronoun Verb Class ولايعنى ذلك ان P يشتمل على الضمائر فقط وانما يشتمل على الضمائر + كلمات اخرى وكذلك V لايشتمل على الافعال فقط بل يشتمل على الافعال + كلمات اخرى ونفس الشيء بالنسبة لـ C حيث يشتمل هذا النطاق على اسماء الفئات + كلمات اخرى وضعت داخل هذا النطاق.

ان تكوين تعليمات باستخدام القواعد المعرفة داخل اللغة يعنى ان هذه التعليمات صحيحة من الناحية النحوية كـ Grammar متفق عليه ومحدد ومقبول داخل اللغة ولكن ليس بالضرورة ان صحة القواعد تعنى ان الامر قابل للتنفيذ فقد يكون الامر غير معرف داخل تعليمات اللغة.

اي ان جميع تعليمات اللغة صحيحة من ناحية اتباعها لقاعدة معينة من قواعد اللغة ولكن الصحة من ناحية اتباع القواعد شرط ضرورى وليس كافيا حتى يكون الامر قابل للتنفيذ.

ايضا من القواعد المعرفة داخل اللغة

## القاعدة : CAVD

حيث C سبق التعرف عليها وتعنى نطاق الفئات Class وغيرها من الكلمات بينما A تعنى نطاق الخصائص Attribute وغيرها من الكلمات و V سبق التعرف عليها وتعنى نطاق الافعال Verb وغيرها من الكلمات و D تعنى البيانات Data و هو نطاق لامحدود مسموح لنا بتحديد محتواه حسب رغبتنا وحسب السياق والاستخدام فلا قيود على هذا النطاق اللامحدود من ناحية المحتوى حتى تقبل التعليمات.

ان الامر

Window title is Hello World.

وكذلك الامر

نافذة تحت عنوان هو مرحبا.

يتبعان القاعدة CAVD حيث النطاق C يضم كل من (Window) و (نافذة) والنطاق A يضم كل من (title) و (تحت عنوان) والنطاق V يضم كل من (is) و (هو) والنطاق D يقبل كل من (Hello World.) و (مرحبا.) حيث ان النقطة الموجودة في نهاية البيانات هي التي تشير الى النطاق D وبالتالي فان الحروف المجهول تصنيفها قبل هذه النقطة يتم تصنيفها داخل هذا النطاق.

## القاعدة I

النطاق I يعنى Instructions وهو نطاق التعليمات وغالبا مايضم تعليمات اللغة المعرفة بشكل سريع.

ويضم هذا النطاق كل من (and) و (the) و ( و ) و (يا) و (الـ) وغيرها من الكلمات والجمل. ومن هنا تظهر لنا تعليمات بدون تاثير وقت التنفيذ مثل ( and ) و ( the ) وهى تعليمات تسمح بدمج الجمل مع بعضها البعض لتكوين القطع مثلا كلا الامرين

I want window  
Window title is Hello World.

او الامرين

انا اريد نافذة  
نافذة تحت عنوان هو مرحبا.

يمكن دمجها لتكوين جملة اكبر باستخدام تعليمات يتم تجاهلها وقت التنفيذ  
على الشكل التالي

I want window and the Window title is Hello World.

او

انا اريد نافذة و النافذة تحت عنوان هو مرحبا.

ويكون التصنيف كما بالجدول التالي

English word الكلمة الانجليزية	Arabic Word الكلمة العربية	Class/Range التصنيف تحت نطاق
I	انا	P (Pronoun)
Want	اريد	V (Verb)
Window	نافذة	C (Class)
and	و	I (Instruction)
the	ال	I (Instruction)
Window	نافذة	C (Class)
Title	تحت عنوان	A (Attribute)
is	هو	V (Verb)
Hello World.	مرحبا.	D (Data)

ومن القواعد الاساسية فى اللغة ايضا

القاعدة DI

القاعدة DVD

وبالتالى يكون لدينا فى لغة السوبرنوبا خمسة قواعد جوهرية تتكون منها الجمل.  
والجدول التالي يشتمل على القواعد الجوهرية الخمسة المصنف عليها تعليمات اللغة

القاعدة Rule	Example	مثال
PVC	I want window	انا اريد نافذة
CAVD	Window title is hello.	نافذة تحت عنوان هو مرحبا.
I	Set window auto show off	امنع الاظهار التلقائي للنافذة
DI	Manager. Instructions are	مدير. تعليماته هي
DVD	[x]. = (1).	[س]. = (1).

### ماهى مرونة الكتابة المتوفرة للشفيرة المصدرية باستخدام السوبرنوبا ؟

الاجابة على هذا السؤال تشتمل نطاق واسع من المعلومات فتكاد تكون الاجابة الكاملة هى الدراسة الشاملة للغة السوبرنوبا

ولكن هناك اطار عام من الملامح التى تصف مرونة الكتابة ومن هذا الاطار

- اللغة ليست حساسة لحالة الحروف Not Case Sensitive بمعنى لافرق بين كتابة حروف اللغة الانجليزية بصورتها الكبيرة UPPER case او الصغيرة lower case
- اللغة ليست حساسة للمسافات بين الكلمات – وايضا ليست حساسة للمسافات بين حروف الكلمة الواحدة وهذه خاصية مميزة للغة السوبرنوبا بشكل كبير
- اللغة ليست حساسة لنهاية الاسطر بمعنى يمكن ترك سطر والبداية فى سطر جديد فى اى وقت – يعنى ممكن ان نكتب نصف كلمة فى سطر ونصفها الاخر فى السطر التالى


هذه مرونة منقطة النظير فى الكتابة حتى ان الشفيرة المصدرية التالية مقبولة من قبل مترجم اللغة

```

i
w a n t w i n d o w
a n d
t h e
w i n d o w t i t l
e i s i want window.

```

ونتيجة تنفيذ البرنامج كالتالى



ونلاحظ شيئاً مهماً ان عنوان النافذة هو I want window وهذا هو نفس الامر المستخدم لإنشاء النافذة !!

والسؤال الان

كيف فرق مترجم السوبرنوفيا بين I want window الاولى المستخدمة كأمر وبين I want window الثانية المستخدمة كعنوان للنافذة ؟

الامر يعود الى الذكاء الاصطناعي المتوفر داخل اللغة والذي يتم من قبل المترجم هو توقع تصنيف الكلمة القادمة والتي سيقوم المترجم بقراءتها مستدلاً بالقواعد الخمسة الاساسية التي تقوم عليها اللغة مع تذكر اخر تصنيف تم تدوينه لآخر كلمة تم قرائتها من قبل المترجم.

حيث من القواعد الجوهرية الخمسة يتم استنتاج القواعد التالية

- بداية الجملة تكون بكلمة تصنف داخل النطاق I او P او C او D
- كلمات النطاق I تاتي فقط في بداية الجملة او بعد كلمة من النطاق D
- كلمات النطاق P تاتي فقط في بداية الجملة
- كلمات النطاق V تاتي فقط بعد كلمة من النطاق P او D او A
- كلمات النطاق C تاتي فقط في بداية الجملة او بعد كلمة من النطاق V
- كلمات النطاق A تاتي فقط بعد كلمة من كلمات النطاق C

ومن خلال هذه القواعد التي تم استنتاجها وعند وضعها للتطبيق من قبل مترجم اللغة فإنه يخرج بالنتيجة التي تجعلك تشعر ان المترجم يفهم ماكتبه له ويدركه ولكنه مبرمج باساليب الذكاء الاصطناعي التي ترتقى بالالة وتجعلها تحاكي ذكاء الانسان فى بعض النقاط.

والسؤال الان

بما ان السوبرنوبا ليست حساسة للمسافات فهل الشفيرة التالية صحيحة ؟

IwantwindowandtheWindowtitleisHello World.

نعم صحيحة

اذن كيف يفرق مترجم السوبرنوبا بين الكلمات حتى يصنفها ؟

هذا تحدى كبير اثناء تصميم السوبرنوبا حيث لا يكون هناك نهاية للكلمات وانما تتوقع اللغة مع قراءة كل حرف جديد ان هذا الحرف هو نهاية الكلمة ثم تبدا بالبحث عن هذه الكلمة فى النطاقات المسوح بها تبعا لنطاق اخر كلمة تم قراءتها فاذا وجدت الكلمة فانها تعتبر ان هذا الحرف هو نهاية الكلمة اما ان لم تجدها فانها تواصل القراءة باستقبال حرف جديد وتعيد البحث مرة اخرى وهكذا حتى تصل الى الكلمة او الى نهاية البرنامج بدون جدوى من البحث وبالتالي نحصل على رسالة خطأ من المترجم.

و عملية البحث المستمرة فى قاموس اللغة عند قراءة كل حرف تؤدي الى مايكاد ان يكون انهيار فى اداء المترجم والحصول على بطء شديد فى عملية الترجمة ولكن مايقذف الموقف هو ان الكلمات المتوقعة تكون فى نطاقات محددة مما يقلل فضاء عملية البحث وبالتالي زيادة الاداء نسبيا عما كان سيكون عليه الحال اذا تم البحث فى فضاء اكبر.

لايقف الامر عند هذا الحد من التعقيد بل يمتد الى دقة عالية فى اختيار كلمات اللغة حتى لا يكون هناك تداخل بين الكلمات الموجودة فى النطاقات المسموح بها فى نفس الوقت فمثلا ( I ) فى النطاق P اى Pronoun تتداخل مع ( is ) الموجود فى النطاق V اى Verb ولكن لا يحدث تعارض لانه من خلال القواعد الجوهرية الخمسة لا يمكن ان تتوقع (كلمة من النطاق P او كلمة من النطاق V) فى نفس الوقت.



وعند وجود احتمال للتعارض فان نظام الذكاء الاصطناعي المتوفر لدى مترجم اللغة يمتلك القدرة على اتخاذ القرار المناسب فعلى سبيل المثال هناك تعارض بين امكانية ان تكون الكلمة من النطاق P او من النطاق D فى نفس الوقت

مثلا كما فى الشفرة التالية

**Window title is I want window.**

حيث بعد ( is ) التى تصنف ضمن النطاق V يمكن ان ياتى كلمة من النطاق P او النطاق D وهذا يسبب تعارض فى تصنيف ( I ) القادمة بعد ( is ) هل تصنف على انها P ام تصنف على انها D

هنا يتدخل الذكاء الاصطناعي ليرى مانتيجه كلا الاحتمالين فيجد النظام ان فى حالة التصنيف على انها P سوف يكون هناك خطأ فى البرنامج لانه ستكون هناك مخالفة للقواعد الجوهرية الخمسة لان القاعدة CAVPD غير معرفة ضمن قواعد اللغة الاساسية ولهذا يتخذ مترجم اللغة القرار السليم بتصنيف ( I ) ضمن D وبذلك تكون لدينا القاعدة CAVD وهى ضمن القواعد الجوهرية الخمسة ويكون البرنامج سليم.

وحتى لا يكون الباب مفتوح على مصراعيه دائما للكثير من المشاكل الناتجة عن المرونة فى الكتابة للكلمات والجمل التى تدخل تحت النطاق D وامكانية تداخلها وتعارضها مع كلمات معرفة فى النطاقات الاخرى المتوقعة مع النطاق D فى نفس الوقت – فان السوبرنوفاتوفر اسلوب لحل هذا النزاع وهو الاعلان عن النطاق D من البداية وليس كما يتم فى النهاية من خلال نقطة حيث يتم الاعلان بوضع الكلمات داخل علامتى تنصيص فردية او مزدوجة مثل معظم لغات البرمجة الاخرى.

**I want window and the window title is "Hello World"**

**I want window and the window title is 'Hello World'**

ويمكن استخدام هذا الاسلوب عند الحاجة فقط.



## التعبيرات Expressions

- التعبيرات تدخل ضمن نطاق البيانات Data اي يرمز لها فى القواعد بـ D وللتفرقة بينها وبين القيمة التى يتم تحديدها مباشرة فانه يتم تصنيف البيانات الى نوعين الاول بيانات كقيمة Data as value والنوع الثانى بيانات كتعبير Data as expression
- يتم كتابة التعبيرات كالتالى:-
  - يتم وضع اسماء المتغيرات بين اقواس مربعة
    - مثلا [variable name] او [ اسم المتغير ]
  - يتم وضع القيم بين اقواس
    - مثلا ( Value ) او ( قيمة )
  - يمكن كتابة القيم بشكل اخر للسماح بكتابة الاقواس
    - | Value | او | قيمة |
  - يتم وضع المعاملات Operators المختلفة بين المتغيرات والقيم

مثال على Data as Value

I want window and the window title is "Hello World"

أمثلة على Data as Expression

I want window and the window title from (Hello World).

I want window and the window title from "(Hello World)"

[mytitle]. = (Hello World).  
I want window and the window title from [mytitle].



## ما المعاملات Operators التي يمكن استخدامها في التعبيرات Expressions ؟

في الواقع تشتمل لغة البرمجة سوبرنوبا على العديد من المعاملات التي تصنف الى احدى المجالات التالية

- المعاملات الرياضية Arithmetic Operators
- المعاملات المنطقية Logical Operators
- المعاملات العلاقية Relational Operators

أولا / المعاملات الرياضية

المعامل Operator	الغرض والاستخدام
* او x او X	الضرب
/	القسمة
%	باقي القسمة
+	الجمع
-	الطرح

ثانيا / المعاملات المنطقية

المعامل Operator	الغرض والاستخدام
AND	نفس الوظيفة المتوقعة والمعروفة من خلال دراسة الدوائر المنطقية النتيجة صحيحة True فقط في حالة صحة كل من اطراف الدخل
OR	النتيجة صحيحة True في حالة صحة اى طرف من اطراف الدخل
NOT	النتيجة هي نفي الدخل – فلو كان الدخل True تكون النتيجة False العكس صحيح اى لو كان الدخل False تكون النتيجة True

### ثالثا / المعاملات العلاقية

المعنى	المعامل
أكبر من	>
أصغر من	<
أكبر من او يساوى	>=
أصغر من او يساوى	<=
يساوى	==
لا يساوى	!=

### أسبقية المعاملات فى التنفيذ Precedence of operators

اولوية التنفيذ	المعاملات فى المستوى
1	* x X / %
2	+ -
3	< > <= >=
4	!= ==
5	Not
6	And Or

مثال

```
I want window and the window title is Assignment.
The window height is 300. and window back color = {30,200,100}.
I want listbox and the listbox name is mylist1.
i want listbox and the listbox name is mylist2.
listbox left is 250.
[v1]. is ( Hello )( World ).
[v2]. = (15)x(3).
[v3]. equal (Country).
[v4]. = (Hello)[v1].
[v5]. = (12) and (7) or (1) not (13).
[v6]. = | (1) [One] (2) [Two] |.
[v7]. = (Hello )[GoodName].
mylist2. listbox selected
listbox must add item "[v1]. is ( Hello )( World )."
listbox must add item "[v2]. = (15)x(3)."
```

نتيجة تشغيل هذا البرنامج كالتالى



لفهم هذا المثال هناك عدة نقاط تحتاج الى التوضيح حيث يجب ان تعلم فى البداية ان جمل المساواة التى تسند قيم الى المتغيرات Assignment Statements تكتب على صيغة القاعدة DVD حيث D هي Data as expression أى بيانات كتعبير اما V فهى اما علامة المساواة = او تكتب is او تكتب equal وهذا ماتم فى الجمل التالية على سبيل المثال

```
[v1]. is ( Hello )( World ).  
[v2]. = (15)x(3).  
[v3]. equal (Country).
```

ايضا نلاحظ ان دمج السلاسل الحرفية يتم من خلال سردها فى التعبير على التوالى بدون استخدام اى معاملات مثل

```
[v1]. is ( Hello )( World ).
```

حيث تم دمج كل من ( Hello ) و ( World ) بسردهما على التوالى بدون معاملات وكذلك فى المثال التالى حيث يتم دمج كل من القيمة ( Hello ) مع المتغير [v1] وذلك بسردهما ايضا على التوالى بدون معاملات

```
[v4]. = (Hello) [v1].
```

ومن النقاط المهمة ان السوبرنونا تفترض وجود اى متغير وتتيح استخدامه مباشرة بدون الحاجة الى تعريفه وايضا بدون الحاجة الى اسناد اى قيمة الى المتغير قبل استخدامه والسطر التالى يبين ذلك


```
[v7]. = (Hello ) [GoodName].
```

حيث في هذا السطر من البرنامج تم استخدام المتغير [GoodName] قبل تعريفه وايضا قبل اسناد اى قيمة له ولان المتغير تم استخدامه مباشرة لأول مرة في الدمج مع سلسلة حرفية وهى القيمة ( Hello ) فان السياق يجعل القيمة الافتراضية التى يتم اسنادها الى المتغير [GoodName] هى ( NULL ) ومن هنا نلاحظ ان نظام الذكاء الاصطناعى المتوفر داخل مترجم السوبرنوفافا يضع قيمة افتراضية للمتغيرات بشكل تلقائى عند استعمالها مباشرة وقبل اسناد اى قيمة اليها وهذه القيمة الافتراضية يتم وضعها حسب سياق استخدام المتغير حيث ان هذه القيمة تكون:-

- ( NULL ) فى حالة الدمج مع السلاسل الحرفية
- ( 1 ) فى حالة عمليات الضرب او القسمة
- ( 0 ) فى حالة عمليات الجمع او الطرح

```
I want window
The window title is Looking To The future.
I want listbox and the listbox top = 60.
window width = 700. and listbox width = 650.
The listbox must add item Supernova know that any variable may
exist at any time in the future.
The listbox must add from [x].
The listbox must add item Supernova suggest that the initial
[x] value is (1) when you write [x] x (5).
The listbox must add from [x] x (5).
The listbox must add item Supernova suggest that the initial
[x] value is (0) when you write [x] + (5).
The listbox must add from [x] + (5).
```

لاحظ ان النصوص الظاهرة على اكثر من سطر فى المثال السابق هى فى الواقع تم كتابتها فى الشفيرة المصدرية فى سطر واحد.



وهذا مثال اخر

```
I want window contains button
Button name is btn1. and button caption = Multiply.
The window title is Looking To The future.

I want listbox and the listbox top = 60.


Btn1 mouseclick. instructions are
:note: "Using [X] in the right side without
        declaration and/or assigning a value to it.
        [x] automatically take the value of (1) because there are
        Multiply operation (x)"

[x]. = [x] x (5).

The listbox must add from [x].

End of instructions
```

ونلاحظ في هذا المثال انه لكتابة ملاحظات او تعليقات Comments داخل الشفيرة المصدرية فانه يتم استخدام :note: يليها الملاحظات المطلوبة على صيغة Data وللتوضيح فان :note: يقع ضمن النطاق I وبالتالي القاعدة المستخدمة هي ID وهي ليست قاعدة جوهرية ثابتة وانما هي قاعدة ثانوية (قاعدة تقبل في حالات خاصة فقط) وبالتالي يتم تحويلها الى القاعدة الجوهرية DI من قبل المترجم عند التعرف عليها. وفيما يلي نتيجة تشغيل هذا البرنامج



والمثال التالي بنفس الفكرة ولكن يطبق عملية القسمة

(15). decimal places allowed

I want window contains button  
Button name is btn1. and button caption = Division.  
The window title is Looking To The future.

I want listbox and the listbox top = 60.


Btn1 mouseclick. instructions are  
:note: "Using [X] in the right side without  
declaration and/or assigning a value to it.  
[x] automatically take the value of (1) because there are  
Division operation (/)"

[x]. = [x] / (5).

The listbox must add from [x].

End of instructions

وتنفيذ البرنامج يعطى النتيجة التالية



والان بعد ان تم توضيح عملية اسناد القيمة الافتراضية الى المتغيرات حسب السياق ننتقل الى نقطة اخرى.



المثال التالي يستخدم المعاملات العلاقية لتحديد هل يقع رقم في نطاق معين ام لا

```
i want window contains listbox
the window title is the test is true when 5 < thevalue < 10.

i want textbox and i want button
the button left is 300.
the button caption is test value.
the button name is btn1.
the listbox top is 50.

btn1 mouseclick. instructions are
    store the textbox value to [x].
    [mytest]. = [x] > (5) and [x] < (10).

ok

mytest true. instructions are
listbox must add item True.
ok

mytest false. instructions are
listbox must add item False.
ok
```

ولفهم هذا المثال ينبغي التعرض لمفهوم احداث الحالة State Events ببساطة هي تنفيذ تعليمات معينة عندما تصل قيمة متغير الى قيمة معينة وفي المثال السابق فان التعليمات التالية يتم تنفيذها عندما تكون قيمة المتغير mytest هي true وبالتالي يتم اضافة العنصر True الى مربع القائمة المعرف في النافذة

```
mytest true. instructions are
listbox must add item True.
ok
```


بينما يتم تنفيذ التعليمات التالية عندما تكون قيمة المتغير mytest هي false

```
mytest false. instructions are
listbox must add item False.
ok
```

حيث يتم اضافة العنصر الذي يحمل النص False الى مربع القائمة المعرف في النافذة

وايضا ينبغي معرفة ان هناك تعليمات او كلمات في اللغة يمكن ان تستخدم كبديلا لتعليمات او كلمات اخرى وتعطى نفس النتيجة مثلا سبق التعرض ان علامة = او is او equal لهما نفس التأثير في جملة الاسناد Assignment statement ويضاف الى ذلك ان كل من end of instructions او ok لهما نفس التأثير في تعريف التعليمات instructions.

ونتيجة تنفيذ المثال السابق كالتالى



حيث تم التجربة بادخال القيمة 7 والتي تقع داخل النطاق الرقمى من 5 الى 10 وبالتالي تم اضافة عنصر يحمل النص True الى مربع القائمة المعرف فى النافذة.

والان لدينا مثال باللغة العربية يوضح عملية الجمع

التطبيق تحت عنوان هو عملية الجمع.

انا اريد نافذة و النافذة تحت عنوان هو عملية الجمع.

يانافذة الاتجاه هو بداية من اليمين لليسار.

يانافذة لون الخلفية هو {255,255,255}.

يانافذة العرض هو 530.

يانافذة الارتفاع هو 200.

انا اريد مربع نص ويا مربع نص البداية اليسرى هي 100.

يامربع نص الاسم هو ت1.

انا اريد مربع نص ويا مربع نص البداية اليسرى هي 100.

يامربع نص البداية العلوية هي 50.

يامربع نص الاسم هو ت2.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 100.  
يامربع نص الاسم هو ت3.

انا اريد زر  
يازر الاسم هو "زر1"  
يازر النص هو "خروج من البرنامج"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد زر  
يازر الاسم هو "زر2"  
يازر النص هو "اجمع الرقمين"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.  
يازر البداية العلوية هي 100.

"زر2 النقر بالفارة" تعليماته هي

ت1. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم1].  
ت2. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم2].  
[النتيجة] = [رقم1] + [رقم2].  
ت3. مربع نص تم تحديده ويامربع نص القيمة من [النتيجة].

نهاية التعليمات

ونتيجة تنفيذ هذا البرنامج كالتالي

خروج من البرنامج	33
	216
اجمع الرقمين	249

## والمثال التالي باللغة العربية يوضح عملية الطرح

التطبيق تحت عنوان هو عملية الطرح.

انا اريد نافذة و النافذة تحت عنوان هو عملية الطرح.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص الاسم هو ت1.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 50.  
يامربع نص الاسم هو ت2.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 100.  
يامربع نص الاسم هو ت3.

انا اريد زر  
يازر الاسم هو "زر1"  
يازر النص هو "خروج من البرنامج"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد زر  
يازر الاسم هو "زر2"  
يازر النص هو "اطرح"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.  
يازر البداية العلوية هي 100.

"زر2 النقر بالفارة" تعليماته هي  
ت1. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم1].  
ت2. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم2].  
[النتيجة] = [رقم1] - [رقم2].  
ت3. مربع نص تم تحديده ويامربع نص القيمة من [النتيجة].  
نهاية التعليمات

## والمثال التالي باللغة العربية يوضح عملية الضرب

التطبيق تحت عنوان هو عملية الضرب.

انا اريد نافذة و النافذة تحت عنوان هو عملية الضرب.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص الاسم هو ت1.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 50.  
يامربع نص الاسم هو ت2.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 100.  
يامربع نص الاسم هو ت3.

انا اريد زر  
يازر الاسم هو "زر1"  
يازر النص هو "خروج من البرنامج"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد زر  
يازر الاسم هو "زر2"  
يازر النص هو "اضرب"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.  
يازر البداية العلوية هي 100.

"زر2 النقر بالفارة" تعليماته هي  
ت1. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم1].  
ت2. مربع نص تم تحديده  
يامربع نص القيمة الى [رقم2].  
[النتيجة]. = [رقم1] x [رقم2].  
ت3. مربع نص تم تحديده ويامربع نص القيمة من [النتيجة].  
نهاية التعليمات

## المثال التالي باللغة العربية يوضح عملية القسمة

التطبيق تحت عنوان هو عملية القسمة.

انا اريد نافذة و النافذة تحت عنوان هو عملية القسمة.

يانافذة الاتجاه هو بداية من اليمين لليسار.

يانافذة لون الخلفية هو {255,255,255}.

يانافذة العرض هو 530.

يانافذة الارتفاع هو 200.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.

يامربع نص الاسم هو ت1.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.

يامربع نص البداية العلوية هي 50.

يامربع نص الاسم هو ت2.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.

يامربع نص البداية العلوية هي 100.

يامربع نص الاسم هو ت3.

انا اريد زر

يازر الاسم هو "زر1"

يازر النص هو "خروج من البرنامج"

يازر البداية اليسرى هي 350.

يازر اسم الخط هو arial.

يازر حجم الخط هو 14.

يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

لوسمحت اغلق النافذة

نهاية التعليمات

انا اريد زر

يازر الاسم هو "زر2"

يازر النص هو "اقسم"

يازر البداية اليسرى هي 350.

يازر اسم الخط هو arial.

يازر حجم الخط هو 14.

يازر العرض هو 150.

يازر البداية العلوية هي 100.

"زر2 النقر بالفارة" تعليماته هي

ت1. مربع نص تم تحديده

يامربع نص القيمة الى [رقم1].

ت2. مربع نص تم تحديده

يامربع نص القيمة الى [رقم2].

[النتيجة] = [رقم1] / [رقم2].

ت3. مربع نص تم تحديده ويامربع نص القيمة من [النتيجة].

نماعة التعليمات

## وفيما يلي مثال باللغة العربية على دمج السلاسل الحرفية

التطبيق تحت عنوان هو برنامج الترحيب.

انا اريد نافذة و النافذة تحت عنوان هو برنامج الترحيب.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص الاسم هو ت1.  
يامربع نص اسم الخط هو arial.  
يامربع نص حجم الخط هو 14.

انا اريد مربع نص ويامربع نص البداية اليسرى هي 100.  
يامربع نص البداية العلوية هي 50.  
يامربع نص الاسم هو ت2.  
يامربع نص اسم الخط هو arial.  
يامربع نص حجم الخط هو 14.

انا اريد زر  
يازر الاسم هو "زر1"  
يازر النص هو "خروج من البرنامج"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد زر  
يازر الاسم هو "زر2"  
يازر النص هو "قل مرحبا؟"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.  
يازر البداية العلوية هي 100.

"زر2 النقر بالفارة" تعليماته هي  
ت1. مربع نص تم تحديده  
يامربع نص القيمة الى [الاسم].  
ت2. مربع نص تم تحديده  
يامربع نص القيمة من (مرحبا) [الاسم].  
نهاية التعليمات

## ونتيجة تنفيذ البرنامج السابق كالتالى



والجدير بالذكر ان الحرية متوفرة فى تحديد اسماء المتغيرات ويمكن ان يشتمل اسم المتغير على مسافات او رموز ويمكن ان يكون الاسم عربى او انجليزى بل يمكن ان يكون اسم المتغير طويل جدا ويمكن ان يكون اكثر من سطر وهى حرية قد تصل الى درجة كبيرة للغاية بمعنى انها حرية زائدة عن الحاجة حيث من غير العملى ان يكون اسم المتغير طويل جدا واكثر من سطر ولكنها امكانية متوفرة ومدعومة من قبل مترجم السوبرنوفال.

ونضيف الى الكلمات التى لها نفس التأثير فى اللغة كلمة You و Please لهما نفس التأثير وكذلك انت و لوسمحت لهما نفس التأثير

بمعنى ان السطر البرمجة

```
You show window
```

يمكن ان يكتب

```
Please show window
```

وكذلك السطر البرمجة

```
انت اعرض النافذة
```

يمكن ان يكتب

```
لوسمحت اعرض النافذة
```



والان نعود الى احداث الحالة State Events ولكن بمثال متقدم وللتذكرة فان احداث الحالة هي تنفيذ تعليمات معينة عندما تصل قيمة متغير الى قيمة معينة.

```
I want window contains button
Button name is btn1. and button caption = Create button.
The window title is Looking To The future.
Window width = 700.

:note: "Btn2 event before creating btn2 !!!"
Btn2 mouseclick. instructions are
    I want window and window title is wow.
End of instructions

:note: " x = 450 event before creating [x] !!!"
X 450. instructions are
    btn1. button selected
    The button caption is Close.
end of instructions

:note: " x = 600 event before creating [x] !!!"
X 600. instructions are you close application ok

Btn1 mouseclick. instructions are
:note: "Using [X] in the right side without
    declaration and/or assigning a value to it.
    [x] automatically take the value of (0) because there are
    sum operation (+)"

    [x]. = [x] + (150).


    I want button and the button name is btn2.
    button left from [x].

End of instructions
```


في هذا المثال يتم انشاء زر وعند الضغط عليه نحصل على زر اخر وهكذا باستمرار حتى يكون لدينا في النافذة ثلاثة ازرار عندما يتم الضغط على الزر فانه يتم انهاء التطبيق.



وفيما يلي متابعة العمل على هذا المثال حيث عند انشاء الزر الثالث يتغير نص الزر الاساسى الى close وعند الضغط عليه يتم انهاء التطبيق.



بينما عند الضغط على اى زر من الازرار التى تم انشائها فاننا نحصل على نافذة جديدة تحت عنوان هو wow.





## جمل التنفيذ Do Statements

جمل التنفيذ تستخدم للتحكم فى سير البرنامج حيث تتيح التفرع المشروط والتكرار وايضا التنظيم من خلال تقسيم البرنامج الى اجزاء.

تشتمل لغة البرمجة سوبرنوبا على أربعة جمل تنفيذ اساسية وهى

- Do IF للتفرع المشروط وباللغة العربية نفذ فى حالة  
○ نهاية التعليمات هى End IF وباللغة العربية نهاية الحالة
- Do While للتكرار وباللغة العربية نفذ طالما  
○ نهاية التعليمات هى End While وباللغة العربية نهاية طالما
- Do Function/Do Procedure لاستدعاء اجراء او دالة.
- Do File لكى يشتمل التطبيق على ملف اخر يحتوى دوال او اجراءات

يلى كل من Do IF و Do While بيانات كتعبير Data as expression وهى الشرط الذى يتم اختباره فاذا كان صحيحا يتم تنفيذ التعليمات اما اذا كان غير صحيح فان التنفيذ ينتقل الى مابعد End IF فى حالة Do IF او ينتقل الى مابعد End While فى حالة Do While



## انظر الى البرنامج التالى

```
Set window auto show off

Application title = Do Statements.

i want window contains listbox
window title = Do Statements.
Window Width = 270. listbox Width = 250.
Window Height = 500. listbox Height = 450.
listbox font size = 14.
listbox back color = {20,255,100}.

[x]. = (1). and do while [x] <= (7).
    listbox must add from [x].
    Do if [x] == (5).
        [t]. = .
        [y]. = (1).
        Do while [y] <= (5).
            [t]. = [t] [y].
            listbox must add from [t].
            [y]. = [y] + (1).
        End while
    End if
    [x]. = [x] + (1).
End while
[t]. = .
[y]. = (1).
do while [y] <= (7).
    [t]. = [t] [y].
    listbox must add from [t].
    [y]. = [y] + (1).
    Do if [y] == (6).
        listbox must add item *****.
    End if
End while
```

You show window


## تلاحظ من هذا البرنامج ان Do While و Do IF تاخذ الشكل التالى

```
[1] Do While ....Data as Expression....
    ....Instructions....
End While

[2] Do If ....Data as Expression....
    ....Instructions....
End If
```

ومسموح لنا بتداخل هذه التركيبات داخل بعضها البعض للتحكم بعمق فى سير البرنامج.

وننتيجة تنفيذ البرنامج كالتالي



ولدينا مثال اخر مقسم الى ملفين

DoFunctionStatement.nova

```
application title = "Do Function"

do file "mylib.nova"

do procedure myuserinterface.
window title = "Do Function"

[input]. = (5). do function myfunc. listbox must add from [output].
[input]. = (3). do function myfunc2. listbox must add from [output].

You show window

Function myfunc.
    [output]. = [input]x[input].
End function

Function myfunc2. and [output]. = [input] x (2). ok
```

## MyLib.nova

```
procedure myuserinterface.  
  
    i want window contains listbox  
    window height = 450.  
    listbox height = 400.  
    window width = 230.  
    window back color = {255,255,255}.  
  
end procedure
```

وبالتالى نلاحظ ان التعليمات تتبع الصورة

```
[3] Do Function ....function name....
```

```
Function ....function name....  
    ....Instructions....  
End Function
```

Note: We can use OK | End of instructions instead of End Function


```
[4] Do Procedure ....procedure name....
```

```
Procedure ....procedure name....  
    ....Instructions....  
End Procedure
```

Note: We can use OK | End of instructions instead of End Procedure

```
[5] Do File ...."FileName.nova"....
```

ونتيجة التنفيذ كالتالى




## وفيما يلي مثال باللغة العربية

التطبيق تحت عنوان هو احب العربي.  
انا اريد نافذة و النافذة تحت عنوان هو مرحبا.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.

انا اريد زر و يازر النص هو خروج.  
يازر البداية اليسرى هي 300.  
يازر الاسم هو "زر1"  
"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد مربع القائمة  
[س]. = (1).  
نفذ طالما [س] => (100).  
يامربع القائمة يجب ان تضيف من [س].  
نفذ في حالة [س] == (3).  
يامربع القائمة يجب ان تضيف عنصر هذا هو رقم 3.  
نهاية الحالة  
[س]. = [س] + (1).  
نهاية طالما

## ونتيجة تنفيذ البرنامج كالتالي



وفيما يلي المثال السابق ولكن باللغة الانجليزية

I want window and the window title is hello.  
The window back color is {255,255,255}.


I want listbox

I want button and the button caption is Close.  
The button left is 300. and the button name is btn1.

```
[x]. = (1).  
Do While [x] <= (100).  
    listbox must add from [x].  
    Do if [x] == (3).  
        Listbox must add item This is number three.  
    End if  
    [x]. = [x] + (1).  
End While
```

```
btn1 mouseclick. instructions are  
    You close window  
End of instructions
```

ونتيجة تنفيذ البرنامج كالتالي





يمكن كتابة نفس البرنامج مرتين مرة عربي ومرة انجليزي في نفس الوقت (خليط)

انا اريد نافذة و النافذة تحت عنوان هو مرحبا.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.

انا اريد زر و يازر النص هو خروج.  
يازر البداية اليسرى هي 300.  
يازر الاسم هو "زر1"

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

انا اريد مربع القائمة  
[س]. = (1).  
نفذ طالما [س] => (100).  
يامربع القائمة يجب ان تضيف من [س].  
نفذ في حالة [س] == (3).  
يامربع القائمة يجب ان تضيف عنصر هذا هو رقم 3.  
نهاية الحالة  
[س]. = [س] + (1).  
نهاية طالما

I want window and the window title is hello.  
The window back color is {255,255,255}.

I want listbox


I want button and the button caption is Close.  
The button left is 300. and the button name is btn1.

```
[x]. = (1).  
Do While [x] <= (100).  
    listbox must add from [x].  
    Do if [x] == (3).  
        Listbox must add item This is number three.  
    End if  
    [x]. = [x] + (1).  
End While
```

```
btn1 mouseclick. instructions are  
    You close window  
End of instructions
```

و عند تنفيذ البرنامج وهو بهذا الشكل نحصل على نافذتين.  
النافذة الاولى ناتجة عن الشفيرة المصدرية المكتوبة باللغة العربية .  
والنافذة الثانية ناتجة عن الشفيرة المصدرية المكتوبة باللغة الانجليزية.

والنتيجة كما بالشكل التالي



وبذلك نكون قد تعرضنا لجمل التنفيذ Do Statements مع عرض امثلة بكل من اللغة العربية والانجليزية و كليهما معا.



## الوقت والتاريخ Date and Time

- تتيح لنا السوبرنوفيا بشكل سريع معرفة معلومات عن الوقت والتاريخ واليوم والشهر والزمن المستغرق منذ تشغيل الحاسب بالثواني لكي يستعمل في اجراء عمليات حساب الزمن الذي مر خلال اداء عملية معينة على سبيل المثال

```
[1] The Date
    [mydate]. = the date.

[2] The Time
    [mytime]. = the time.

[3] The Seconds
    [myseconds]. = the seconds.

[4] The Day
    [myday]. = the day.

[5] The Month
    [mymonth]. = the month.
```


حيث يمكن في جمل الاسناد Assignment Statements ان تستخدم the data او التاريخ لمعرفة التاريخ – كما يستعمل the time او الزمن لمعرفة الوقت و يستعمل the seconds او الثواني لمعرفة الثواني المستغرقة منذ تشغيل الحاسب و يستعمل the day او اليوم لمعرفة اسم اليوم بينما يستعمل the month او الشهر لمعرفة اسم الشهر.



وفيما يلي مثال يعرض التاريخ the date


```
Application title = The Date.  
I want window and window title = The Date.  
window back color = {255,255,255}.  
window width = 330. window height = 150.  
I want label and label caption from the date.  
The label font size = 40. and label width = 300.  
label height = 100.
```

حيث نلاحظ استعمال الامر label caption ليكون محتوى العنوان ناتج عن بيانات كتعبير data as expression وهذا التعبير هو the date اي التاريخ.




ونفس البرنامج يمكن ان يتم كتابته باللغة العربية كالتالي

```
التطبيق تحت عنوان هو التاريخ.  
انا اريد نافذة والنافذة تحت عنوان هو التاريخ.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330.  
يانافذة الارتفاع هو 250.  
انا اريد عنوان ويا عنوان النص من التاريخ.  
يا عنوان حجم الخط هو 40.  
يا عنوان العرض هو 300.  
يا عنوان الارتفاع هو 100.
```



## والان لدينا مثال لمعرفة اليوم The day

```
Application title = The Day.  
I want window and window title = The Day.  
window back color = {255,255,255}.  
window width = 330. window height = 150.  
I want label and label caption from the Day.  
The label font size = 40. and label width = 300.  
label height = 100.
```



ويمكن كتابة نفس البرنامج باللغة العربية كالتالى

```
التطبيق تحت عنوان هو اليوم.  
انا اريد نافذة والنافذة تحت عنوان هو اليوم.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330.  
يانافذة الارتفاع هو 250.  
انا اريد عنوان وياعنوان النص من اليوم.  
ياعنوان حجم الخط هو 40.  
ياعنوان العرض هو 300.  
ياعنوان الارتفاع هو 100.
```



## ولمعرفة الشهر the month

```
Application title = The Month.
```

```
I want window and window title = The Month.  
window back color = {255,255,255}.  
window width = 330. window height = 150.
```

```
I want label and label caption from the Month.  
The label font size = 40. and label width = 300.  
label height = 100.
```



ويمكن كتابة البرنامج باللغة العربية كالتالى

التطبيق تحت عنوان هو الشهر.

انا اريد نافذة والنافذة تحت عنوان هو الشهر.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330.  
يانافذة الارتفاع هو 250.

انا اريد عنوان وياعنوان النص من الشهر.  
ياعنوان حجم الخط هو 40.  
ياعنوان العرض هو 300.  
ياعنوان الارتفاع هو 100.



## ولمعرفة الزمن بالثواني the seconds

```
Application title = The Seconds.  
  
I want window and window title = The Seconds.  
window back color = {155,155,155}.  
window width = 330. window height = 150.  
  
I want label and label caption from the seconds.  
The label font size = 40. and label width = 300.  
label height = 100.
```



## ويمكن كتابة البرنامج باللغة العربية

التطبيق تحت عنوان هو الوقت بالثواني.

انا اريد نافذة والنافذة تحت عنوان هو الوقت بالثواني.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330.  
يانافذة الارتفاع هو 250.

انا اريد عنوان ويا عنوان النص من الثواني.  
يا عنوان حجم الخط هو 40.  
يا عنوان العرض هو 300.  
يا عنوان الارتفاع هو 100.



## والمثال التالي لمعرفة الوقت the time

```
Application title = The Time.
```

```
I want window and window title = The Time.  
window back color = {155,155,155}.  
window width = 330. window height = 150.
```

```
I want label and label caption from the time.  
The label font size = 40. and label width = 300.  
label height = 100.
```



## والبرنامج باللغة العربية كالتالي

التطبيق تحت عنوان هو الوقت.

انا اريد نافذة والنافذة تحت عنوان هو الوقت.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330.  
يانافذة الارتفاع هو 250.

انا اريد عنوان ويا عنوان النص من الوقت.  
يا عنوان حجم الخط هو 40.  
يا عنوان العرض هو 300.  
يا عنوان الارتفاع هو 100.





والان لدينا مثال يعرض الزمن بالثواني باستمرار

```
Application title = The Seconds.  
  
I want window and window title = The Seconds.  
window back color = {5,155,255}.  
window width = 330. window height = 250.  
window name = win1.  
  
I want label  
The label font size = 40. and label width = 300.  
label height = 100.  
  
I want button and button caption = Close.  
the button name is btn1.  
button top = 150. and button left = 150.  
  
do while (True).  
    label caption from ( ) the Seconds.  
    [x]. = the seconds. and do while "the seconds - [x] < (0.1)"  
        do windows events  
    end while  
end while  
  
btn1 mouseclick. instructions are  
    You close application  
end of instructions
```

في هذا المثال تم استخدام الامر do windows events لتنفيذ احداث الويندوز المتعلقة بالاستجابة للدخل القادم من الفارة ولوحة المفاتيح وغيرها من الاحداث حتى لا تتعطل واجهة التطبيق عن الاستجابة والتفاعل مع المستخدم.

وفيما يلي صورة التطبيق اثناء العمل



## والان نتعرض لنفس التطبيق ولكن باللغة العربية

التطبيق تحت عنوان هو الوقت بالثواني.

ملاحظة:

"

يتم إلغاء العلامات العشرية  
لان الوقت المعروض يتحدث عن مرور ثانية

"

(0). علامات عشرية متاحة

انا اريد نافذة والنافذة تحت عنوان هو الوقت بالثواني.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330. و يانافذة الارتفاع هو 250.

انا اريد عنوان وياعنوان النص من الوقت.  
ياعنوان حجم الخط هو 40.  
ياعنوان العرض هو 300. و ياعنوان الارتفاع هو 100.

انا اريد زر ويازر النص هو خروج. و يازر الاسم هو "زر1"  
يازر البداية العلوية هي 150. و يازر البداية اليسرى هي 150.

نفذ طالما (1).

ياعنوان النص من ( ) الثواني.

[س] = الثواني. و نفذ طالما الثواني - [س] > (1).

نفذ احداث الويندوز

نهاية طالما

نهاية طالما

"زر1 انقر بالفارة" تعليماته هي لوسمحت اغلق النافذة حسنا

ونتيجة تنفيذ التطبيق كالتالى



وباستخدام نفس الفكرة في المثال السابق نتعرض لمثال يعرض الوقت باستمرار

```
Application title = The Time.

I want window and window title = The Time.
window back color = {5,155,255}.
window width = 330. window height = 250.
window name = win1.

I want label
The label font size = 40. and label width = 300.
label height = 100.

I want button and button caption = Close.
the button name is btn1.
button top = 150. and button left = 150.

do while (True).
    label caption from ( ) the time.
    [x]. = the seconds. and do while the seconds - [x] < (1).
        do windows events
    end while
end while

btn1 mouseclick. instructions are
    You close application
end of instructions
```

وفيما يلي التطبيق وقت التشغيل حيث يتم عرض الوقت وتحديثه بالشكل مستمر على آخر وضع لمعرفة الزمن باستمرار وفي أي وقت عند الضغط على الزر Close فإن التطبيق يتم انهائه.



## والان نتعرض لنفس التطبيق ولكن باللغة العربية

التطبيق تحت عنوان هو الوقت.

انا اريد نافذة والنافذة تحت عنوان هو الوقت.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 330. و يانافذة الارتفاع هو 250.

انا اريد عنوان وياعنوان النص من الوقت.  
ياعنوان حجم الخط هو 40.  
ياعنوان العرض هو 300. و ياعنوان الارتفاع هو 100.

انا اريد زر ويازر النص هو خروج. و يازر الاسم هو "زر1"  
يازر البداية العلوية هي 150. و يازر البداية اليسرى هي 150.

نفذ طالما (1).

ياعنوان النص من ( ) الوقت.

[س]. = الثواني. و نفذ طالما الثواني - [س] > (1).


نفذ احداث الويندوز

نهاية طالما

نهاية طالما

"زر1 انقر بالفارة" تعليماته هي لوسمحت اعلق النافذة حسنا

## والتطبيق وقت التشغيل كالتالى




والحال كما هو فى التطبيق السابق حيث يتم عرض الوقت وتحديثه بالشكل مستمر على اخر وضع لمعرفة الزمن باستمرار وفى اى وقت عند الضغط على الزر خروج فان التطبيق يتم انهاءه.

والان ننتقل الى تطبيق مشجع بشكل اكبر وهو تطبيق يعرض نص يظهر ويختفى  
وفكرة هذا التطبيق ماخوذة من مثال تم شرحه بالصوت والصورة و قام به  
Stephen France – كندا حيث كان يعرض تقنية البرمجة بدون كود PWCT وتم  
تطبيق نفس الفكرة الان بالسوبرنوا Supernova.

```
Application title = Hello!!.  
  
I want window and window title = Hello!!.  
Window back color = {5,155,255}.  
Window width = 330. window height = 250.  
Window name = win1.  
  
I want button and button caption = Be my friend , click to say yes !!.  
The button top = 20. button width = 300.  
The button name = btn1.  
  
I want button the button top = 80.  
Button caption = . and the button name = btn2.  
  
I want button the button top = 130.  
Button caption = . and the button name = btn3.  
  
btn1 mouseclick. instructions are  
    Btn2. button selected and button caption = Really !! ? .  
End of instructions  
  
btn2 mouseclick. instructions are  
Do while (True).  
    Btn3. button selected and button caption = Hurrah.  
    [x]. = the seconds. and do while "the seconds - [x] < (0.5)"  
        Do windows events  
    End while  
    Btn3. button selected and button caption = .  
    [x]. = the seconds. and do while "the seconds - [x] < (0.5)"  
        Do windows events  
    End while  
End while  
End of instructions  
  
btn3 mouseclick. instructions are  
    You close window  
End of instructions
```

فكرة هذا المثال ليست صعبة !!  
حيث يتم عمل حلقة Loop تعمل باستمرار لان شرط استمرارها هو True اي  
تستمر للابد حيث ان نتيجة الشرط دائما صحيحة.  
و داخل هذه الحلقة يتم مسح النص الذي يشكل عنوان الزر  
ثم يتم الانتظار لمدة ما  
ثم يتم عرض نص يشكل عنوان الزر ثم يتم الانتظار لمدة ما

ونتيجة تنفيذ التطبيق كالتالي



Flash





## السلاسل الحرفية Strings

تتيح لنا السوبرنوفنا بشكل سريع اجراء عمليات معالجة على السلاسل الحرفية وبالتالي تفتح المجال لبناء العديد من التطبيقات التي تقوم على معالجة النصوص

```
[1] Len ....String....
    [mylen]. = len (Hello World).

[2] ....Letters Count.... Left ....String....
    [myleft]. = (5) left (Hello World).

[3] ....Letters Count.... Right ....String....
    [myright]. = (5) right (Hello World).

[4] Alltrim ....String....
    [mymsg]. = alltrim ( Hello World ).

[5] RTrim ....String....
    [mymsg]. = rtrim ( Hello World ).

[6] LTrim ....String....
    [mymsg]. = ltrim ( Hello World ).

[7] Upper ....String....
    [upper]. = upper (Hello World).

[8] Lower ....String....
    [lower]. = lower (Hello World).

[9] ....Count.... Replicate ....String....
    [newstr]. = (3) replicate ( wow ).

[10] ....Index.... Inside ....String....
    [myletter]. = (3) inside (Hello World).

[11] ....SubString.... At ....String....
    [pos]. = (110) at (Hello World).

[12] ....(RangeFrom:RangeTo).... SubStr ....String....
    [substr]. = (3:5) substr (Hello World).

[13] ....(SubStringFrom:SubStringTo).... Transform ....String....
    [transform]. = (World)(:)(RealWorld) transform (Hello World).
```





المثال التالي يستخدم Len من اجل تحديد طول (عدد حروف) سلسلة حرفية

```
Application title = StrLen.  
I want window contains textbox and window title = Get String Length.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
button caption = get string length. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = len [mytext].  
    label caption from [myvar].  
end of instructions
```

فى هذا المثال تم استعمال len يليها المتغير [mytext] والذى يشتمل على السلسلة الحرفية التى يقوم بادخالها المستخدم فى مربع النص المعرف داخل نافذة التطبيق

ونتيجة تنفيذ هذا البرنامج كالتالى

حيث تم ادخال السلسلة الحرفية Hello World المكونة من 11 حرف حيث كلمة Hello خمسة حروف ثم هناك مسافة تحسب بحرف ثم كلمة World لتضيف خمسة حروف اخرى وبالتالي يكون طول السلسلة الحرفية  $5+1+5 = 11$  حرف



## والشفيرة المصدرية لهذا التطبيق باللغة العربية كالتالى

التطبيق تحت عنوان هو الطول.

انا اريد نافذة والنافذة تحت عنوان هو معرفة الطول.

انا اريد مربع نص  
انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1"  
يازر البداية اليسرى هي 300.

يازر النص هو احصل على طول سلسلة حرفية.  
يازر العرض هو 150.

"زر1 انقر بالفارة" تعليماته هي  
يا مربع نص القيمة الى [النتيجة].  
[النتيجة]. = طول [النتيجة].  
يا عنوان النص من [النتيجة].  
نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

حيث تم ادخال السلسلة الحرفية (السلام عليكم) وهي تشتمل على 6 حروف لكلمة  
(السلام) + 1 حرف للمسافة + 5 حروف لكلمة (عليكم) وبالتالي يكون طول العبارة  
الحرفية = 6 + 1 + 5 = 12 حرف



الآن ننتقل إلى استعمال `left` والتي تستخدم للحصول على عدد معين من الحروف داخل سلسلة حرفية بداية من جهة اليسار

```
Application title = StrLeft.  
  
I want window contains textbox and window title = Get Sub String From  
Left.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Get 5 letters from left. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [mytext]. = (5) left [mytext].  
    label caption from [mytext].  
end of instructions
```

في هذا المثال تم الحصول على خمسة حروف من السلسلة الحرفية المسجلة في المتغير `[mytext]` بداية من جهة اليسار وتم تسجيل هذه الحروف الخمسة في نفس المتغير `[mytext]` أي جعلنا المتغير يشتمل على خمسة حروف فقط ثم قمنا بعرض هذه الحروف الخمسة من خلال عنوان معرف داخل نافذة التطبيق.

والتطبيق وقت التشغيل كالتالي حيث تم ادخال السلسلة الحرفية `Hello World` وعند عرض خمس حروف من جهة اليسار من هذه السلسلة حصلنا على كلمة `Hello` كما هو واضح من الصورة.



## ويمكن كتابة التطبيق باللغة العربية كما يلي

التطبيق تحت عنوان هو "اليسار"  
انا اريد نافذة والنافذة تحت عنوان هو "اليسار"

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو "احصل على 5 حروف من اليسار"

يازر العرض هو 150.

انا اريد مربع نص

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [النص1].  
[النص1] = (5) حروف يسار [النص1].  
ياعنوان النص من [النص1].

نهاية التعليمات


## والتطبيق وقت التشغيل كالتالى



والان لدينا مثال متقدم قليلا عن المثال السابق حيث الان نستعمل Left مع عدد متغير من الحروف التي يتم تحديدها من خلال قيمة رقمية يتم ادخالها في مربع نص داخل النافذة وفور تغير هذه القيمة الرقمية يتغير عنوان الزر الذي يظهر النتيجة ليشير الى عدد الحروف التي سيتم الحصول عليها بداية من جهة اليسار.

```
Application title = StrLeft2.  
I want window contains textbox and window title = Get Sub String From Left.  
The textbox name is text1.  
I want textbox and textbox top is 50.  
The textbox name is text2.  
I want label and label top is 100.  
I want button and button name is btn1. and button left = 300.  
Button caption = Get n letters from left. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext1].  
    text2. textbox selected and textbox value to [mytext2].  
    [mytext]. = [mytext2] left [mytext1].  
    label caption from [mytext].  
End of instructions  
text2 change. instructions are  
    Textbox value to [mytext].  
    The button caption from (Get ) [mytext] ( letters from left).  
End of instructions
```

والتطبيق وقت التشغيل كالتالي حيث تم ادخال السلسلة الحرفية "How Are you ?" وتم تحديد عدد 8 حروف للحصول عليها من جهة اليسار وبالتالي تم الحصول على السلسلة الحرفية "How Are" ونلاحظ انها تشتمل مسافتين منها واحدة في النهاية.



## والان نرى الشفيرة المصدرية لنفس التطبيق ولكن باللغة العربية

التطبيق تحت عنوان هو اليسار 2.

انا اريد نافذة والنافذة تحت عنوان هو اليسار 2.

انا اريد مربع نص

يامربع نص الاسم هو نص 1.

انا اريد مربع نص

يامربع نص البداية العلوية هي 50.

يامربع نص الاسم هو نص 2.

انا اريد عنوان وياعنوان البداية العلوية هي 100.

انا اريد زر ويازر الاسم هو "زر 1" ويازر البداية اليسرى هي 300.

يازر النص هو احصل على ن حروف من اليسار.

يازر العرض هو 150.

"زر 1 النقر بالفارة" تعليماته هي

نص 1. مربع نص تم تحديده ويامربع نص القيمة الى [ك1].

نص 2. مربع نص تم تحديده ويامربع نص القيمة الى [ك2].

[النتيجة] = [ك2] حروف يسار [ك1].

ياعنوان النص من [النتيجة].

نهاية التعليمات

"نص 2 التغير" تعليماته هي

يامربع نص القيمة الى [عدد الحروف].

يازر النص من (احصل على ) [عدد الحروف] ( حروف من اليسار).

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

اليسار 2

How Are you ?

احصل على 8 حروف من اليسار

8

How Are

والآن ننتقل الى **Right** والتي تتيح لنا الحصول على عدد معين من الحروف ولكن من جهة اليمين.

```
Application title = StrRight.  
  
I want window contains textbox and window title = Get Sub String  
From Right.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Get 5 letters from right. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [mytext]. = (5) right [mytext].  
    label caption from [mytext].  
end of instructions
```

والتطبيق وقت التشغيل كالتالى

حيث تم ادخال Hello World وتم الحصول على كلمة World التي تمثل خمس حروف من جهة اليمين.



## والتطبيق باللغة العربية كالتالى

التطبيق تحت عنوان هو "اليمين"

انا اريد نافذة والنافذة تحت عنوان هو "اليمين"

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو "احصل على 5 حروف من اليمين"  
يازر العرض هو 150.

انا اريد مربع نص

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [النص1].  
[النص1]. = (5) حروف يمين [النص1].  
ياعنوان النص من [النص1].

نهاية التعليمات

والتطبيق وقت التشغيل كالتالى  
حيث كما فى المثال المقدم باللغة الانجليزية تم ادخال Hello World وتم  
الحصول على كلمة World التى تمثل خمس حروف من جهة اليمين.





وبنفس الطريقة كما حدث مع left يمكن تقديم مثال متقدم عن المثال السابق حيث الان نستعمل Right مع عدد متغير من الحروف التي يتم تحديدها من خلال قيمة رقمية يتم ادخالها في مربع نص داخل النافذة وفور تغير هذه القيمة الرقمية يتغير عنوان الزر الذي يظهر النتيجة ليشير الى عدد الحروف التي سيتم الحصول عليها بداية من جهة اليمين.

```
Application title = StrRight2.  
I want window contains textbox and window title = Get Sub String From Right.  
the textbox name is text1.  
I want textbox and textbox top is 50.  
the textbox name is text2.  
I want label and label top is 100.  
I want button and button name is btn1. and button left = 300.  
Button caption = Get n letters from Right. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext1].  
    text2. textbox selected and textbox value to [mytext2].  
    [mytext]. = [mytext2] right [mytext1].  
    label caption from [mytext].  
end of instructions  
text2 change. instructions are  
    textbox value to [mytext].  
    the button caption from (Get ) [mytext] ( letters from right).  
end of instructions
```

والتطبيق وقت التشغيل كالتالى



## ويمكن كتابة نفس التطبيق باللغة العربية كالتالى

التطبيق تحت عنوان هو اليمين 2.

انا اريد نافذة والنافذة تحت عنوان هو اليمين 2.

انا اريد مربع نص

يامربع نص الاسم هو نص 1.

انا اريد مربع نص

يامربع نص البداية العلوية هي 50.

يامربع نص الاسم هو نص 2.

انا اريد عنوان ويا عنوان البداية العلوية هي 100.

انا اريد زر ويازر الاسم هو "زر 1" ويازر البداية اليسرى هي 300.

يازر النص هو احصل على ن حروف من اليمين.

يازر العرض هو 150.

"زر 1 النقر بالفارة" تعليماته هي

نص 1. مربع نص تم تحديده ويامربع نص القيمة الى [ك1].

نص 2. مربع نص تم تحديده ويامربع نص القيمة الى [ك2].

[النتيجة] = [ك2] حروف يمين [ك1].

يا عنوان النص من [النتيجة].

نهاية التعليمات

"نص 2 التغير" تعليماته هي

يامربع نص القيمة الى [عدد الحروف].

يازر النص من (احصل على ) [عدد الحروف] ( حروف من اليمين).

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى



والآن نتعرف على Alltrim والتي تستخدم لحذف جميع المسافات على يمين ويسار سلسلة حرفية

```
Application title = StrAlltrim.  
I want window contains textbox and window title = Alltrim.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
button caption = Alltrim. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = Alltrim [mytext].  
    label caption from [myvar].  
end of instructions
```

والتطبيق وقت التشغيل كالتالي



والشفيرة المصدرية باللغة العربية كالتالي

التطبيق تحت عنوان هو قص المسافات.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو قص المسافات.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو قص كل المسافات. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

يا مربع نص القيمة الى [نص1].

[نص1]. = قص كل المسافات [نص1].

يا عنوان النص من [نص1].

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالي



ويمكن استخدام RTrim لقص المسافات على يمين السلسلة الحرفية فقط  
وايضا يمكن استخدام LTrim لقص المسافات على يسار السلسلة الحرفية فقط

## والمثال التالي يوضح استخدام RTrim

```
Application title = StrRtrim.  
I want window contains textbox and window title = Rtrim.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
button caption = Rtrim. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = Rtrim [mytext].  
    label caption from [myvar].  
end of instructions
```

## والمثال التالي يوضح استخدام LTrim

```
Application title = StrLtrim.  
I want window contains textbox and window title = Ltrim.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
button caption = Ltrim. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = Ltrim [mytext].  
    label caption from [myvar].  
end of instructions
```

وفيما يلي الشفرة المصدرية للمثالين ولكن باللغة العربية

هذا هو المثال المتعلق بحذف المسافات على يسار سلسلة حرفية

التطبيق تحت عنوان هو قص المسافات اليسرى.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو "قص المسافات اليسرى"

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر

يازر النص هو "حذف المسافات"

يازر الاسم هو "زر1"

يازر البداية اليسرى هي 300.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].

[ت1]. = قص المسافات يسار [ت1].

ياعنوان النص من [ت1].

نهاية التعليمات

وهذا هو المثال المتعلق بحذف المسافات على يمين سلسلة حرفية

التطبيق تحت عنوان هو قص المسافات يمين.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو "قص المسافات يمين"

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر

يازر النص هو "حذف المسافات"

يازر الاسم هو "زر1"

يازر البداية اليسرى هي 300.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].

[ت1]. = قص المسافات يمين [ت1].

ياعنوان النص من [ت1].


نهاية التعليمات

والان ناتي الى التحويل من الحروف الكبيرة UPPER case الى الحروف الصغيرة lower case والعكس (في اللغة الانجليزية)  
ببساطة يتم استعمال UPPER للتحويل الى الحروف الكبيرة  
بينما يتم استعمال lower للتحويل الى الحروف الصغيرة

المثال التالي يستعمل UPPER (التحويل الى الحروف الانجليزية الكبيرة)

```
Application title = StrUpper.  
  
I want window contains textbox and window title = Upper.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Upper. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = Upper [mytext].  
    label caption from [myvar].  
end of instructions
```

وهذه هي صورة التطبيق وقت التشغيل  
نلاحظ انه تم ادخال Hello World وعند اجراء عملية التحويل حصلنا على  
السلسلة الحرفية HELLO WORLD التي تشتمل على نفس الحروف ولكن بعد  
تحويلها جميعا الى الحروف الانجليزية الكبيرة




## المثال التالي يستعمل lower (التحويل الى الحروف الانجليزية الصغيرة)

```
Application title = StrLower.  
  
I want window contains textbox and window title = Lower.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
  
button caption = Lower. and button width = 150.  
  
btn1 mouseclick. instructions are  
  
    textbox value to [mytext].  
    [myvar]. = Lower [mytext].  
    label caption from [myvar].  
  
end of instructions
```

وفيما يلي صور التطبيق وقت التشغيل

في هذا المثال تم ادخال السلسلة الحرفية Hello World وعند اجراء عملية التحويل حصلنا على السلسلة الحرفية hello world والتي تشتمل على نفس الحروف ولكن بعد التحويل الى الحروف الانجليزية الصغيرة.



ويمكن كتابة التطبيقين باللغة العربية كالتالى

## التحويل الى الحروف الكبيرة

التطبيق تحت عنوان هو التحويل للحروف الكبيرة.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو التحويل للحروف الكبيرة.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو التحويل للحروف الكبيرة. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

يا مربع نص القيمة الى [ت1].

[ت1]. = حروف كبيرة [ت1].

يا عنوان النص من [ت1].

نهاية التعليمات

والتطبيق وقت التشغيل كالتالى

حيث تم ادخال العبارة Hello World

وتم الحصول على النتيجة HELLO WORLD

اى تم النجاح فى تحويل الحروف الى الحروف الانجليزية الكبيرة كما هو متوقع.





وفيما يلي المثال باللغة العربية للتحويل الى الحروف الانجليزية الصغيرة

التطبيق تحت عنوان هو التحويل للحروف الصغيرة.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو التحويل للحروف الصغيرة.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو التحويل للحروف الصغيرة. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

يا مربع نص القيمة الى [ت1].

[ت1]. = حروف صغيرة [ت1].

يا عنوان النص من [ت1].

نهاية التعليمات

وفيما يلي صورة التطبيق وقت التشغيل


وقد تمت التجربة بادخال السلسلة الحرفية Hello World  
وكما هو متوقع حصلنا على النتيجة hello world اي تم تحويل حروف  
السلسلة الحرفية التي تم ادخالها الى الحروف الانجليزية الصغيرة.



## والآن ننتقل الى Replicate هي ببساطة تستخدم لتكرار سلسلة حرفية عدد معين من المرات

```
Application title = StrReplicate.  
  
I want window contains textbox and window title = Replicate String.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Replicate String. and button width = 150.  
  
btn1 mouseclick. instructions are  
  
    textbox value to [mytext].  
    [mytext]. = (3) Replicate [mytext].  
    label caption from [mytext].  
  
end of instructions
```

والتطبيق وقت التشغيل كالتالي  
حيث تم ادخال السلسلة الحرفية WOW  
وتم تكرارها ثلاث مرات لنحصل على السلسلة الحرفية WOWWOWWOW



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو التكرار.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو التكرار.

انا اريد عنوان وياعنوان البداية العلوية هي 50.  
ياعنوان العرض هو 300.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو كرر 3 مرات. ويازر العرض هو 150.

"زر1 انقر بالفارة" تعليماته هي

يا مربع نص القيمة الى [ت1].  
[ت1] = (3) مرات كرر [ت1].  
يا عنوان النص من [ت1].

نهاية التعليمات


والتطبيق وقت التشغيل كالتالى  
حيث تم ادخال السلسلة الحرفية wow  
وتم تكرارها ثلاث مرات لنحصل على النتيجة wowwowwow كما هو متوقع.



ويمكن تقديم المثال بشكل متقدم حيث يتم تحديد عدد مرات التكرار من خلال قيمة رقمية يتم ادخالها في مربع نص معرف داخل نافذة التطبيق كالتالي

```
Application title = StrReplicate2.  
  
I want window contains textbox and window title = Replicate String.  
The textbox name is text1.  
  
I want textbox and textbox top is 50.  
The textbox name is text2.  
  
I want label and label top is 100. and label width = 400.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = Replicate string n times. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext1].  
    text2. textbox selected and textbox value to [mytext2].  
    [mytext]. = [mytext2] replicate [mytext1].  
    label caption from [mytext].  
End of instructions  
  
text2 change. instructions are  
    textbox value to [mytext].  
    the button caption from (Replicate String ) [mytext] ( times).  
End of instructions
```

والتطبيق وقت التشغيل كالتالي



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو كرر2.  
انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو كرر2.  
يامربع نص الاسم هو نص1.  
انا اريد مربع نص ويمربع نص البداية العلوية هي 50.  
يامربع نص الاسم هو نص2.  
انا اريد عنوان وياعنوان البداية العلوية هي 100.  
ياعنوان العرض هو 400.  
انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو كرر عبارة ن مرة. ويازر العرض هو 150.  
"زر1 النقر بالفارة" تعليماته هي  
نص1. مربع نص تم تحديده ويمربع نص القيمة الى [ت1].  
نص2. مربع نص تم تحديده ويمربع نص القيمة الى [ت2].  
[النتيجة]. = [ت2] مرات كرر [ت1].  
ياعنوان النص من [النتيجة].  
نهاية التعليمات  
"نص2 التغير" تعليماته هي  
يامربع نص القيمة الى [ت1].  
يازر النص من (كرر عبارة) [ت1] (مرة).  
نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

Screenshot of a Windows application window titled "كرر 2". The window contains a text input field with the text "باغالى", a button labeled "كرر عبارة 3 مرة", and another text input field with the text "3". Below these is a label "باغالى باغالى باغالى".

والآن ناتي الى **Inside** والتي تستخدم للحصول على حرف معين داخل سلسلة  
حرفية من خلال ترتيب هذا الحرف داخل السلسلة من جهة اليسار.

```
Application title = StrInside.  
  
I want window contains textbox and window title = Char Inside String.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Get Char Inside String. and button width = 150.  
  
btn1 mouseclick. instructions are  
  
    textbox value to [mytext].  
    [mytext]. = (3) Inside [mytext].  
    label caption from [mytext].  
  
end of instructions
```

والتطبيق وقت التشغيل كالتالي  
حيث تم تحديد الحرف الثالث من جهة اليسار داخل السلسلة  
وهذا الحرف من السلسلة هو رقم 3



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو حرف داخل عبارة.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو حرف داخل عبارة.

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو الحصول على الحرف الثالث. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].  
[ت1]. = (3) حرف داخل [ت1].  
ياعنوان النص من [ت1].

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

حيث تم ادخال العبارة الحرفية Mahmoud

لكى نحصل على الحرف الثالث من جهة اليسار وهو الحرف h كما هو متوقع.



## والان يمكن ان نرى مثال اكثر تقدما يوضح استخدام Inside

```
Application title = StrInside2.
```

```
I want window contains textbox and window title = Get Char n Inside String.  
the textbox name is text1.
```


```
I want textbox and textbox top is 50.  
the textbox name is text2.
```

```
I want label and label top is 100. and label width = 400.
```

```
I want button and button name is btn1. and button left = 300.  
Button caption = Get Char n Inside String. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext1].  
    text2. textbox selected and textbox value to [mytext2].  
    [mytext]. = [mytext2] Inside [mytext1].  
    label caption from [mytext].  
end of instructions
```

```
text2 change. instructions are
```

## والتطبيق وقت التشغيل كالتالى





## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو حرف داخل عبارة حرفية 2.  
انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو حرف داخل عبارة حرفية 2.  
يامربع نص الاسم هو نص 1.  
انا اريد مربع نص ويامربع نص البداية العلوية هي 50.  
يامربع نص الاسم هو نص 2.  
انا اريد عنوان وياعنوان البداية العلوية هي 100 وياعنوان العرض هو 400.  
انا اريد زر ويازر الاسم هو "زر 1" ويازر البداية اليسرى هي 300.  
يازر النص هو حرف ن داخل العبارة الحرفية. ويازر العرض هو 150.  
"زر 1 النقر بالفارة" تعليماته هي  
نص 1. مربع نص تم تحديده ويامربع نص القيمة الى [ت 1].  
نص 2. مربع نص تم تحديده ويامربع نص القيمة الى [ت 2].  
[النتيجة] = [ت 2] حرف داخل [ت 1].  
ياعنوان النص من [النتيجة].  
نهاية التعليمات  
"نص 2 التغير" تعليماته هي  
يامربع نص القيمة الى [نص 1].  
يازر النص من (حرف) [نص 1] ( داخل العبارة الحرفية).  
نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

حرف داخل عبارة حرفية 2

Good Man

حرف 6 داخل العبارة الحرفية

6

M

ولدينا ايضا مثال اخر اكثر تقدما يوضح استخدام Inside لعرض حروف سلسلة  
حرفية حرف حتى نهايتها

```
Application title = StrInside3.  
I want window contains textbox and window title = Letters Inside String.  
I want listbox and listbox top is 100.  
I want button and button name is btn1. and button left = 300.  
Button caption = Letters Inside String. and button width = 150.  
btn1 mouseclick. instructions are  
  the textbox value to [mytext1].  
  [mysize]. = len [mytext1].  
  [x]. = (1). do while [x] <= [mysize].  
    [mytext]. = [x] Inside [mytext1].  
    listbox must add from [mytext].  
    [x]. = [x] + (1).  
  end while  
End of instructions
```

والتطبيق وقت التشغيل كالتالي حيث تم ادخال السلسلة الحرفية Hello World  
وكانت النتيجة ان حصلنا على حروف هذه السلسلة من خلال اضافة هذه الحروف  
كعناصر في مربع القائمة المعرف داخل نافذة التطبيق.



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو حرف داخل عبارة3.

انا اريد نافذة تحتوى مربع نص و النافذة تحت عنوان هو "حرف داخل عبارة 3"

انا اريد مربع القائمة و يامربع القائمة البداية العلوية هي 70.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو حروف داخل عبارة. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].

[الطول]. = طول [ت1].

[س]. = (1). نفذ طالما [س] => [الطول].

[ح1]. = [س] حرف داخل [ت1].

يامربع القائمة يجب ان تضيف من [ح1].

[س]. = [س] + (1).

نهاية طالما

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى



والآن نتعرف على At ببساطة هي تبحث عن سلسلة حرفية داخل سلسلة حرفية اخرى وتعود لنا بنتيجة البحث والتي تكون رقم يعبر عن بداية السلسلة التي نبحث عنها وفي حالة الفشل فى ايجادها (اي فى حالة كونها غير موجودة) تعود لنا بالرقم صفر

```
Application title = StrAt.
```


```
I want window contains textbox and window title = Get Pos of Sub String.  
the textbox name is text1.
```

```
I want textbox and textbox top is 50.  
the textbox name is text2.
```

```
I want label and label top is 100. and label width = 400.
```

```
I want button and button name is btn1. and button left = 300.  
Button caption = Get Pos of Sub String. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext1].  
    text2. textbox selected and textbox value to [mytext2].  
    [mytext]. = [mytext2] at [mytext1].  
    label caption from [mytext].  
end of instructions
```

والتطبيق وقت التشغيل كالتالى حيث تم ادخال سلسلة حرفية وهى Welcome to Supernova وتم البحث عن to لتحديد مكانها وبالفعل تم ذلك وحصلنا على النتيجة وهو بداية من الحرف رقم 9



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو مكان عبارة حرفية داخل اخرى.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو مكان عبارة حرفية داخل اخرى. ويامربع نص الاسم هو نص1.

انا اريد مربع نص ويامربع نص البداية العلوية هي 50. ويامربع نص الاسم هو نص2.  
انا اريد عنوان وياعنوان البداية العلوية هي 100.  
وياعنوان العرض هو 500.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو تحديد المكان. ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

نص1. مربع نص تم تحديده ويامربع نص القيمة الى [ت1].  
نص2. مربع نص تم تحديده ويامربع نص القيمة الى [ت2].  
[النتيجة]. = [ت2] ما هي بدايتها فى [ت1].  
ياعنوان النص من [النتيجة].

نهاية التعليمات


## والتطبيق وقت التشغيل كالتالى



والآن ياتي دور التعرف على Substr ببساطة تستخدم للحصول على جزء من سلسلة حرفية (عدد معين من الحروف داخلها بداية من حرف معين يتم تحديده من جهة اليسار)

```
Application title = StrSubStr.  
  
I want window contains textbox and window title = Get Sub String.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
button caption = Get Sub String. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = (3:5) substr [mytext].  
    label caption from [myvar].  
end of instructions
```

والتطبيق وقت التشغيل كالتالي حيث تم ادخال السلسلة الحرفية Hello World وتم الحصول على جزء منها بداية من الحرف رقم 3 نحصل على خمسة حروف وبالتالي نحصل على Ilo W حيث هناك مسافة تشكل حرف ايضا.



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو جزء من عبارة.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو جزء من عبارة.

انا اريد عنوان وياعنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو مطلوب 5 حروف بداية من الحرف الثالث. ويازر العرض هو 180.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].

[النتيجة]. = (3:5) حروف داخل [ت1].

ياعنوان النص من [النتيجة].

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

وبنفس الطريقة تم ادخال السلسلة الحرفية Hello World وتم الحصول على جزء

منها بداية من الحرف رقم 3 نحصل على خمسة حروف وبالتالي نحصل على llo

W حيث هناك مسافة تشكل حرف ايضا.



ويمكن تقديم المثال بشكل متقدم حيث يتم تحديد نقطة البداية وعدد الحروف من خلال متغيرات كالتالي

```
Application title = StrSubStr2.  
I want window contains textbox and window title = Get Sub String.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
Button caption = Get Sub String. and button width = 150.  
btn1 mouseclick. instructions are  
  
    textbox value to [mytext].  
    [x1]. = (5).  
    [x2]. = (5).  
    [range]. = [x1](:)[x2].  
    [myvar]. = [range] substr [mytext].  
    label caption from [myvar].  
  
End of instructions
```

ونحصل على نفس النتيجة المتوقعة كالمثال السابق والشفيرة المصدرية باللغة العربية كالتالي

التطبيق تحت عنوان هو جزء من عبارة حرفية2.  
انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو جزء من عبارة حرفية.  
انا اريد عنوان ويا عنوان البداية العلوية هي 50.  
انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو مطلوب 5 حروف بداية من الحرف الخامس. ويازر العرض هو 180.  
"زر1 النقر بالفارة" تعليماته هي

```
يا مربع نص القيمة الى [ت1].  
[س1] = (5).  
[س2] = (5).  
النطاق = [س1](:)[س2].  
النتيجة = [النطاق] حروف داخل [ت1].  
يا عنوان النص من [النتيجة].
```


نهاية التعليمات



والان ناتي الى Transform والتي تستخدم لتحويل جزء من سلسلة حرفية الى محتوى اخر.

```
Application title = StrTransform.  
I want window contains textbox and window title = Transform.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
button caption = Sub String Transform. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [myvar]. = (nova:*nova*) transform [mytext].  
    label caption from [myvar].  
end of instructions
```


والتطبيق وقت التشغيل كالتالي



ويمكن اعادة تقديم المثال بشكل اكثر تقدما كالتالى

```
Application title = StrTransform.  
I want window contains textbox and window title = Transform.  
textbox name is text1.  
textbox value is wow nova wow.  
I want label and label top is 150. and label width = 400.  
I want textbox and textbox top = 50.  
textbox name is text2.  
textbox value is nova.  
I want textbox and textbox top = 100.  
textbox name is text3.  
textbox value is *nova*.  
I want button and button name is btn1. and button left = 300.  
button caption = Sub String Transform. and button width = 150.  
btn1 mouseclick. instructions are  
    text1. textbox selected and textbox value to [mytext].  
    text2. textbox selected and textbox value to [str1].  
    text3. textbox selected and textbox value to [str2].  
    [myvar]. = [str1][:][str2] transform [mytext].  
    label caption from [myvar].  
end of instructions
```

والتطبيق وقت التشغيل كالتالى



ويمكن اعادة تقديم المثالين السابقين من خلال شيفرة مصدرية باللغة العربية كالتالى

### المثال الاول

التطبيق تحت عنوان هو تحويل عبارة حرفية.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو تحويل عبارة حرفية.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

يامربع نص القيمة هي نجم.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.  
يازر النص هو تحويل عبارة حرفية. ويازر العرض هو 150.

"زر1 انقر بالفارة" تعليماته هي

يامربع نص القيمة الى [ت1].  
[النتيجة]. = (نجم:\*نجم\*) حول [ت1].  
يا عنوان النص من [النتيجة].

نهاية التعليمات

والتطبيق وقت التشغيل كالتالى



## وبالنسبة للمثال الثاني (متقدم)

- التطبيق تحت عنوان هو حول2.
- انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو حول2.
- يامربع نص الاسم هو نص1.
- يامربع نص القيمة هي نجم.
- انا اريد عنوان وياعنوان البداية العلوية هي 150.
- ياعنوان العرض هو 400.
- انا اريد مربع نص ويامربع نص البداية العلوية هي 50.
- يامربع نص الاسم هو نص2.
- يامربع نص القيمة هي نجم.
- انا اريد مربع نص ويامربع نص البداية العلوية هي 100.
- يامربع نص الاسم هو نص3.
- يامربع نص القيمة هي \*نجم\*.
- انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.
- يازر النص هو تحويل عبارة حرفية. ويازر العرض هو 150.
- "زر1 النقر بالفارة" تعليماته هي
- نص1. مربع نص تم تحديده ويامربع نص القيمة الى [ت1].
- نص2. مربع نص تم تحديده ويامربع نص القيمة الى [ت2].
- نص3. مربع نص تم تحديده ويامربع نص القيمة الى [ت3].
- [النتيجة] = [ت2]([ت3] حول [ت1]).
- ياعنوان النص من [النتيجة].
- نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى

حول2

نجم

نجم

\*نجم\*

\*نجم\*

تحويل عبارة حرفية



## المصفوفات Arrays

السوبرنوتا تتيح لنا التعامل مع المصفوفات باحجام مختلفة وابعاد مختلفة حيث يتم استخدام متغير على انه اسم المصفوفة ويتم الفصل بين اسم المصفوفة ورقم العنصر الذى يتم ادخاله على شكل قيمة او متغير باستخدام \ و يمكن ان يتم ادخال العنصر دفعة واحدة على انه متغير والفصل بين اسم المصفوفة ورقم العنصر من خلال علامة - ويتم دعم المصفوفات ذات الابعاد المختلفة من خلال استخدام اكثر من \ و نظام الذكاء الاصطناعي فى السوبرنوتا لا يشترط تحديد عدد عناصر المصفوفة حيث يمكن استخدام العنصر مباشرة فى اى وقت بدون الاهتمام بالمصفوفة و العنصر داخل المصفوفة هو اولا واخيرا متغير يسهل الوصول اليه من خلال اسم المصفوفة + رقم العنصر.

```
[array name]\[element index as variable]
```

```
[array name]\(element index as value)
```

```
[array name-element index as value]
```

```
Multi-dimension arrays supported by using nested \
```



لدينا الان مثال بسيط يوضح كيفية التعامل مع المصفوفات

في هذا المثال يتم استخدام مصفوفة myarray ويتم اسناد قيم الى عشرة عناصر باستخدام حلقة نفذ طالما Do While Loop ثم يتم عرض قيم هذه العناصر العشرة باستخدام ثلاثة طرق مختلفة وبسيطة للاشارة الى العناصر في المصفوفة

```
Application title = "Array"

I want window contains listbox and the window title is Array.
listbox height = 500.
window height = 550. window width = 230.

[x]. = (1). and Do while [x] <= (10).
    [myarray]\[x]. = [x] x [x].
    [x]. = [x] + (1).
End while

listbox must add item Using the first method.

listbox must add from [myarray-1].
listbox must add from [myarray-2].
listbox must add from [myarray-3].
listbox must add from [myarray-4].
listbox must add from [myarray-5].
listbox must add from [myarray-6].
listbox must add from [myarray-7].
listbox must add from [myarray-8].
listbox must add from [myarray-9].
listbox must add from [myarray-10].

listbox must add item Using the second method.

listbox must add from [myarray]\(1).
listbox must add from [myarray]\(2).
listbox must add from [myarray]\(3).
listbox must add from [myarray]\(4).
listbox must add from [myarray]\(5).
listbox must add from [myarray]\(6).
listbox must add from [myarray]\(7).
listbox must add from [myarray]\(8).
listbox must add from [myarray]\(9).
listbox must add from [myarray]\(10).

listbox must add item Using the third method.


[x]. = (1). and Do while [x] <= (10).
    listbox must add from [myarray]\[x].
    [x]. = [x] + (1).
End while
```

والتطبيق وقت التشغيل كالتالى

حيث نلاحظ انه تم اسناد القيم 1 4 9 16 25 36 49 64 81 100 الى عشرة عناصر داخل المصفوفة

ونلاحظ ان اسناد القيم العشرة للعناصر تم بالمرور على الارقام من 1 الى 10 ثم تسجيل حاصل ضرب الرقم فى نفسه كقيمة للعنصر داخل المصفوفة

ثم تم عرض هذه القيم العشرة بثلاثة طرق مختلفة للاشارة الى اسم العنصر داخل المصفوفة وحصلنا على نفس النتيجة فى كل مرة مما يدل على انه لا اختلاف حدث عندما تغيرت الطريقة المستخدمة للاشارة الى اسم العنصر تماما كما هو متوقع .



```
Array
Using the first method
1
4
9
16
25
36
49
64
81
100
Using the second method
1
4
9
16
25
36
49
64
81
100
Using the third method
1
4
9
16
25
36
49
64
81
100
```

## والان نرى الشفيرة المصدرية باللغة العربية

التطبيق تحت عنوان هو "المصفوفة"

انا اريد نافذة و النافذة تحت عنوان هو "المصفوفة"  
يانافذة الارتفاع هو 550.  
يانافذة العرض هو 230.

انا اريد مربع القائمة  
يامربع القائمة الارتفاع هو 500.

[س]. = (1). و نفذ طالما [س] => (10).  
[مصفوفة1] \ [س] = [س] x [س].  
[س] = [س] + (1).  
نهاية طالما

يامربع القائمة يجب ان تضيف عنصر "باستخدام الطريقة الاولى"  
يامربع القائمة يجب ان تضيف من [مصفوفة1-1].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-2].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-3].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-4].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-5].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-6].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-7].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-8].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-9].  
يامربع القائمة يجب ان تضيف من [مصفوفة1-10].

يامربع القائمة يجب ان تضيف عنصر "باستخدام الطريقة الثانية"

يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (1).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (2).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (3).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (4).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (5).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (6).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (7).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (8).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (9).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ (10).

يامربع القائمة يجب ان تضيف عنصر "باستخدام الطريقة الثالثة"

[س]. = (1). و نفذ طالما [س] => (10).  
يامربع القائمة يجب ان تضيف من [مصفوفة1] \ [س].  
[س] = [س] + (1).



## والتطبيق وقت التشغيل كالتالى

ان ما تم فى نفس التطبيق المقدم باللغة الانجليزية هو الذى تم الان حيث نلاحظ انه تم اسناد القيم 1 4 9 16 25 36 49 64 81 100 الى عشرة عناصر داخل المصفوفة

ونلاحظ ان اسناد القيم العشرة للعناصر تم بالمرور على الارقام من 1 الى 10 ثم تسجيل حاصل ضرب الرقم فى نفسه كقيمة للعنصر داخل المصفوفة

ثم تم عرض هذه القيم العشرة بثلاثة طرق مختلفة للاشارة الى اسم العنصر داخل المصفوفة وحصلنا على نفس النتيجة فى كل مرة مما يدل على انه لا اختلاف حدث عندما تغيرت الطريقة المستخدمة للاشارة الى اسم العنصر تماما كما هو متوقع .



والان يمكن ان نتعرف على مثال اكثر تقدما فى التعامل مع المصفوفات وذلك من خلال استخدام مصفوفة ذات بعدين وتمثل جدول يشتمل على عشرة صفوف + ثلاثة اعمدة.


```
Application title = "Array2"

I want window and the window title is Array2.
Window width = 350. window height = 300.
i want button and button caption = close. and button name = btn1.
button top = 230. and button width = 330.
btn1 mouseclick. instructions are you close window ok
i want listbox and the listbox left = 10. and listbox name = list1.
listbox width = 100.
i want listbox and the listbox left = 120. and listbox name = list2.
listbox width = 100.
i want listbox and the listbox left = 240. and listbox name = list3.
listbox width = 100.

[x]. = (1). and Do while [x] <= (10).
    [myarray]\[x]\(1). = [x].
    [myarray]\[x]\(2). = [x] x [x].
    [myarray]\[x]\(3). = [x] x [x] x [x].
    [x]. = [x] + (1).
End while

[x]. = (1). and Do while [x] <= (10).
    list1. listbox selected
    listbox must add from [myarray]\[x]\(1).
    list2. listbox selected
    listbox must add from [myarray]\[x]\(2).
    list3. listbox selected
    listbox must add from [myarray]\[x]\(3).
    [x]. = [x] + (1).
End while
```


والتطبيق وقت التشغيل كالتالى



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو "مصفوفة2"  
 انا اريد نافذة والنافذة تحت عنوان هو "مصفوفة2"  
 ينافذة العرض هو 350. وينافذة الارتفاع هو 300.  
 انا اريد زر ويازر النص هو خروج. ويازر الاسم هو "زر1"  
 يازر البداية العلوية هي 230. ويازر العرض هو 330.  
 "زر1 النقر بالفارة" تعليماته هي لوسمحت اغلق النافذة حسنا  
 انا اريد مربع القائمة ويامربع القائمة البداية اليسرى هي 10.  
 يامربع القائمة الاسم هو "قائمة1"  
 يامربع القائمة العرض هو 100.  
 انا اريد مربع القائمة ويامربع القائمة البداية اليسرى هي 120.  
 يامربع القائمة الاسم هو "قائمة2"  
 يامربع القائمة العرض هو 100.  
 انا اريد مربع القائمة ويامربع القائمة البداية اليسرى هي 240.  
 يامربع القائمة الاسم هو "قائمة3"  
 يامربع القائمة العرض هو 100.  
 [س]. = (1). و نفذ طالما [س] => (10).  
 [مصفوفة1][س]\(1) = [س].  
 [مصفوفة1][س]\(2) = [س] x [س].  
 [مصفوفة1][س]\(3) = [س] x [س] x [س].  
 [س] = [س] + (1).  
 نهاية طالما  
 [س]. = (1). و نفذ طالما [س] => (10).  
 "قائمة1" مربع قائمة تم تحديدها  
 يامربع القائمة يجب ان تضيف من [مصفوفة1][س]\(1).  
 "قائمة2" مربع قائمة تم تحديدها  
 يامربع القائمة يجب ان تضيف من [مصفوفة1][س]\(2).  
 "قائمة3" مربع قائمة تم تحديدها  
 يامربع القائمة يجب ان تضيف من [مصفوفة1][س]\(3).  
 [س]. = [س] + (1).  
 نهاية طالما

## والتطبيق وقت التشغيل كالتالى





## الملفات Files

السوبرنونا تتيح لنا التعامل مع الملفات (الانشاء - القراءة - الكتابة)

وتتيح لنا ايضا تشغيل البرامج الخارجية وفتح الملفات بشكل مباشر وسريع.

كما تتيح لنا عرض النافذة الشهيرة التي تطلب من المستخدم تحديد ملف للفتح او للحفظ وهو امر شائع الاستخدام فى تطبيقات مايكروسوفت ويندوز.

```
Execute file ....data as value....  
Execute from ....data as expression....  
....data as expression.... memory to file ....data as expression....  
....data as expression.... file to memory ....data as expression....  
....data as expression.... file to open  
....data as expression.... file to save
```



والان لدينا مثال بسيط يقوم بفتح الملفات للتشغيل – فمثلا يتم فتح ملف من نوع الشفيرة المصدرية للسوبرنوبا (احد البرامج المكتوبة بالسوبرنوبا) ويتم فتح ملف صورة ايضا.

```
I want window and the window title is Execute file.
The window back color is {255,0,0}.


I want button and the button caption is Run "welcome2.nova"
The button top is 60. and the button left is 20.
The button width is 300. and the button name is btn1.

I want button and the button caption is Run "open test.jpg"
The button top is 100. and the button left is 20.
The button width is 300. and the button name is btn2.

Btn1 MouseClick. instructions are
    :note: " execute file command parameter is data as value"
    execute file "..\..\test\welcome2.nova"
End of instructions

Btn2 MouseClick. instructions are
    :note: " execute from command parameter is data as expression"
    execute from "..\..\test\test.jpg"
End of instructions
```

والتطبيق وقت التشغيل كالتالى



## والشفيرة المصدرية باللغة العربية كالتالى

انا اريد نافذة والنافذة تحت عنوان هو "تشغيل ملف"  
يانافذة لون الخلفية هو ازرق.  
يانافذة العرض هو 350. يانافذة الارتفاع هو 300.  
انا اريد عنوان وياعنوان النص هو مرحبا.  
ياعنوان حجم الخط هو 50.  
ياعنوان العرض هو 300. ياعنوان الارتفاع هو 80.  
ياعنوان لون الكتابة هو ابيض.  
ياعنوان لون الخلفية هو ازرق.  
انا اريد زر ويازر النص هو "فتح الصورة"  
يازر البداية العلوية هي 100.  
يازر البداية اليسرى هي 20.  
يازر العرض هو 300.  
يازر الاسم هو "زر1"  
يازر لون الكتابة هو الارجوان.  
يازر لون الخلفية هو ابيض.  
"زر1 النقر بالفارة" تعليماته هي  
ملاحظة: "تشغيل الملف تاخذ القيمة عبارة عن بيانات Data"  
تشغيل الملف "cover.jpg"  
نهاية التعليمات  
انا اريد زر ويازر النص هو "فتح تطبيق نوبا"  
يازر البداية العلوية هي 150.  
يازر البداية اليسرى هي 20.  
يازر العرض هو 300.  
يازر الاسم هو "زر2"  
يازر لون الكتابة هو الارجوان.  
يازر لون الخلفية هو ابيض.  
"زر2 النقر بالفارة" تعليماته هي  
ملاحظة: "تشغيل من تاخذ القيمة عبارة عن تعبير Expression"  
تشغيل من "(ArabicCode\_Picture.nova)"  
نهاية التعليمات

## والتطبيق وقت التنفيذ كالتالى




## والان لدينا مثال يقوم بحفظ محتوى متغير في الذاكرة memory الى ملف File

```
I want window and the window title is memory to file.
The window back color is purple.

I want button and the button caption is Run "create test.txt"
The button top is 60. and the button left is 20.
The button width is 300. and the button name is btn1.
The button back color is white.

Btn1 MouseClick. instructions are
[mystr]. = "(Hello
How are you ?
I hope that you are fine.
bye.)"
[mystr]. memory to file "(test.txt)"
End of instructions
```

والتطبيق وقت التشغيل كالتالي حيث بمجرد الضغط على الزر المعرف داخل نافذة التطبيق يتم انشاء الملف test.txt وكتابة محتوى المتغير mystr بداخله ويمكن فتح الملف باستخدام برنامج المفكرة Notepad.exe المتوفرة في مايكروسوفت ويندوز MS-Windows لمشاهدة محتواه.



## والشفيرة المصدرية باللغة العربية كالاتى

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو حفظ نص فى ملف.  
يانافذة الاتجاه هو بداية من اليمين لليسا.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

انا اريد زر  
يازر النص هو "اكتب النص فى ملف mytext.txt"  
يازر العرض هو 180.  
يازر الاسم هو "زر1"

انا اريد مربع نص  
يامربع نص البداية اليسرى هي 200.

"زر1 النقر بالفارة" تعليماته هي

يامربع نص القيمة الى [س].  
[س]. ذاكرة يحفظ محتواها فى ملف "(mytext.txt)"

نهاية التعليمات

والتطبيق وقت التشغيل كالتالى حيث نجد مربع نص معرف داخل النافذة وفى التجربة التى قمنا بها تم كتابة سلسلة حرفية فى مربع النص وهى (السلام عليكم ورحمة الله وبركاته) ثم بالضغط على الزر المعرف داخل النافذة يتم حفظ السلسلة الحرفية المكتوبة فى مربع النص المعرف داخل النافذة فى الملف mytext.txt





والان ننتقل لمثال جديد يوضح كيفية قراءة محتوى ملف وتسجيله في متغير موجود في الذاكرة Memory و يتم ذلك باستخدام الامر file to memory كما يتضح من خلال المثال الذى يقوم بقراءة الملف test2.txt ووضع محتواه في مربع تحرير editbox معرف داخل نافذة التطبيق.


```
I want window contains editBOX
The window title is file to memory.
The window back color is purple.

I want button and the button caption is Run "read test2.txt"
The button top is 260. and the button left is 20.
The button width is 300. and the button name is btn1.
The button back color is white.

Btn1 MouseClick. instructions are
    "(test2.txt)" file to memory [x].
    editbox value FROM [x].

End of instructions
```

والتطبيق وقت التشغيل كالتالى



## والشفيرة المصدرية باللغة العربية كالتالى

امنع الاظهار التلقائى للنافذة  
امنع اعادة تحجيم النافذة  
انا اريد نافذة و النافذة تحت عنوان هو قراءة ملف وعرض محتواه.  
يانافذة الاتجاه هو بداية من اليمين لليساار.

انا اريد مربع تحرير

"(test.txt)" ملف يحفظ محتواه فى الذاكرة [س].  
يامربع تحرير القيمة من [س].

لوسمحت قم بتوسيط النافذة  
لوسمحت اعرض النافذة

## والتطبيق وقت التشغيل كالتالى


حيث تم قراءة محتوى الملف test.txt  
وتم نقل هذا المحتوى الذى تم قراءته الى متغير [س]  
ثم نقل محتوى هذا المتغير الى مربع التحرير المعرف داخل نافذة التطبيق



والان لدينا مثال متقدم نوعا ما عند المقارنة بالمثال السابق حيث تم نقل محتوى الملف في متغير ثم تم قراءة محتوى هذا المتغير حرف حرف للحصول على مابداخله سطر سطر ليتم اضافة كل سطر على حده كعنصر في مربع القائمة listBox المعرف داخل نافذة التطبيق

```
I want window contains listBox
The window title is file to memory2.
The window back color is purple.
I want button and the button caption is Run "read test2.txt"
The button top is 260. and the button left is 20.
The button width is 300. and the button name is btn1.
The button back color is white.
Btn1 MouseClick. instructions are
"(test2.txt)" file to memory [x].
[lineend]. = (
).
    [mymax]. = len [x].
    [myline]. = .
    [y]. = (1).
    Do while [y] <= [mymax].
        [myadd]. = [y] inside [x].
        [myline]. = [myline] [myadd].
        [mycheck]. = (2) right [myline].
        Do if [mycheck] == [lineend] or [y] == [mymax].
            listBox must add from [myline].
            [myline]. = .
        End if
        [y]. = [y] + (1).
    End while
End of instructions
```

والتطبيق وقت التشغيل كالتالى



والان نتعرف من خلال الامثلة البسيطة على كيفية اظهار نافذة (فتح ملف) او (حفظ باسم) حتى يختار المستخدم ملف ما او يحدد اسم الملف و ببساطة يتم ذلك من خلال الامرين ( نافذة فتح ملف ) و الامر file to save (نافذة حفظ باسم)

```
I want window and the window title is file to open.  
[x]. file to open  
The window title from [x].
```

```
i want window and the window title is file to save.  
[x]. file to save  
the window title from [x].
```

```
I want window and the window title is file to open2.  
[x]. file to open  
  
[check]. = [x] == ().  
Do if [check].  
    the window title is no file selected.  
End if  
Do if not [check].  
    the window title from [x].  
End if
```

## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو "ملف لكى نفتحته"  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.




[س]. ملف لكى نفتحته  
النافذة تحت عنوان من [س].

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو "ملف لكى نفتحته"  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

[س]. ملف لكى نحفظه  
النافذة تحت عنوان من [س].

## والامثلة وقت التشغيل كالتالى





## تطبيق الديلفرى Delivery Application

فى هذا التطبيق نتعرف بشكل افضل كيف تتناغم عناصر اللغة المختلفة اثناء بناء وتطوير التطبيقات – انه تطبيق بسيط للغاية ولكنه جيد يوضح كيفية الاستفادة من اوامر التعامل مع السلاسل الحرفية Strings لاجراء عملية البحث Search ويوضح كيف يتم بناء واجهة التطبيق الرسومية GUI بالاضافة الى استخدام الاحداث Events لاستدعاء اجراءات Procedures تقوم بوظائف التطبيق كما ان هناك استخدام لاوامر التعامل مع الوقت والتاريخ Date & Time لتسجيل معلومات تفصيلية عن الطلب Order كما هناك استخدام لاوامر التعامل مع الملفات Files حيث يتم حفظ الطلب فى ملف ثم فتحه للعرض باستخدام المفكرة .Notepad



الشفيرة المصدرية للتطبيق كالتالى وهى سهلة الفهم ان شاء الله فى حالة استعاب ماتم شرحه فى الفصول السابقة باذن الله رب العالمين.

```
The application title is "Delivery"
Set window auto show off
Set window resize off

[myarray]. = (3).
[myarray]\(1). = (Tea).
[myarray]\(2). = (Cafe).
[myarray]\(3). = (Sandwich).

I want window and the window title is "Delivery"
Window width = 730. and window height = 530.

I want label
The label width is 700. and label height is 120.
The label font size is 20.
The label back color = white.
The label caption is
"
    What do you want ?
"

I want label and label top is 140.
Label caption is Title :.
Label font name is "arial"
Label font size is "14"

I want textbox
Textbox left is 80.
Textbox top is 140.
Textbox width is 500.
Textbox font name is arial.
Textbox font size is 14.

I want button
Button top is 140.
Button left is 600.
Button caption is Search.
Button font name is arial.
Button font size is 14.
Button name is BtnSearch.

I want listbox and listbox top is 180.
Listbox width is 700. and listbox height is 250.
The listbox font name is arial.
The listbox font size is 14.

[x]. = (1). and do while [x] <= [myarray].
The listbox must add from [myarray]\[x].
[x]. = [x] + (1).
End while

I want button and the button top is 450.
The button left is 240. and button caption = "Ok"
The button name is btn1.
The button back color is white.

I want button and the button top is 450.
The button left is 360. and button caption = "Close"
```

## تابع الشفيرة المصدرية

```
The button name is btnclose.
The button fore color is white.
The button back color is red.
btnclose mouseclick. instructions are you close application ok

listbox value is 1.

You centre window and You show window

btn1 mouseclick. instructions are
    The listbox value to [x].
    Do if not [x] == (0).
        [myitem]. = [myarray]\[x].
        [mystr]. = "(
            Delivery Application
            -----
            Date :) thedate (
            Time :) thetime (
            Month :) themonth (
            Day :) theday (
            Item :-----> ) [myitem] (
            -----)"
        [mystr]. memory to file "(delivery.txt)"
        execute file "delivery.txt"
    End if
End of instructions

btn search mouse click. instructions are
    textbox value to [mystr].
    [mystr]. = lower [mystr].
    [x]. = (1).
    Do while [x] <= [myarray].
        [myitem]. = lower [myarray]\[x].
        [check]. = [mystr] at [myitem].
        Do if not [check] == (0).
            listbox value from [x].
        End if
        [x]. = [x] + (1).
    End while
End of instructions
```

ويمكن اعادة كتابة التطبيق وتطويره ليكون بشكل افضل مما هو عليه الان فمثلا يمكن الغاء زر البحث Search بحيث يكون البحث تفاعلى اى يتم بشكل ديناميكى وتلقائى بمجرد كتابة او تعديل عبارة البحث التى يتم ادخالها فى مربع النص.

كما يمكن تحسين الجزء من الشفيرة الخاص بتوليد تقرير الطلب بحيث يتم تجهيز التقرير باستخدام تعبير expression يتم ارساله الى الملف مباشرة بدون استخدام متغير وسيط كما تم



## وبهذا تكون الشفرة المصدرية للتطبيق كالتالى

```
The application title is "Delivery2"
Set window auto show off
Set window resize off

[myarray]. = (3).
[myarray]\(1). = (Tea).
[myarray]\(2). = (Cafe).
[myarray]\(3). = (Sandwich).

I want window and the window title is "Delivery"
Window width = 730. and window height = 530.

I want label
The label width is 700. and label height is 120.
The label font size is 20.
The label back color = white.
The label caption is
"
    What do you want ?
"

I want label and label top is 140.
label caption is Title :.
label font name is "arial"
label font size is "14"

I want textbox
textbox left is 80.
textbox top is 140.
textbox width is 500.
textbox font name is arial.
textbox font size is 14.
textbox name is text1.

I want listbox and listbox top is 180.
Listbox width is 700. and listbox height is 250.
The listbox font name is arial.
The listbox font size is 14.

[x]. = (1). and do while [x] <= [myarray].
The listbox must add from [myarray]\[x].
[x]. = [x] + (1).
end while

I want button and the button top is 450.
The button left is 240. and button caption = "Ok"
The button name is btn1.
The button back color is white.

I want button and the button top is 450.
The button left is 360. and button caption = "Close"
The button name is btnclose.
The button fore color is white.
The button back color is red.
btnclose mouseclick. instructions are you close application ok

listbox value is 1.


You centre window and You show window
```

## تابع الشفيرة المصدرية


```
btn1 mouseclick. instructions are
    The listbox value to [x].
    Do if not [x] == (0).
        "(
            Delivery Application
            -----
            Date :) thedate (
            Time :) thetime (
            Month :) themonth (
            Day :) theday (
            Item :-----> ) [myarray]\[x] (
            -----) "
            Memory to file "(delivery.txt)"
            Execute file "delivery.txt"
        End if
    End of instructions

text1 change. instructions are
    textbox value to [mystr].
    [mystr]. = lower [mystr].
    [x]. = (1).
    Do while [x] <= [myarray].
        [myitem]. = lower [myarray]\[x].
        [check]. = [mystr] at [myitem].
        Do if not [check] == (0).
            listbox value from [x].
        End if
        [x]. = [x] + (1).
    End while
End of instructions
```


## والتطبيق وقت التشغيل كالتالى



وفيما يلي مثال على تقرير الطلب الناشئ من التطبيق



وفيما يلي التطبيق بعد تعديل الشفيرة المصدرية واستخدام البحث التفاعلي.





## الماكرو او المختصر Macro

هو اشارة باختصار من متغير الى متغير اخر لاستخدامه – او من متغير الى دالة او اجراء لاستدعائه

بمعنى اوضح يكون لدينا متغير وهذا المتغير يشتمل على اسم متغير اخر نريد الوصول اليه او هذا المتغير يشتمل على اسم دالة او اجراء نريد استدعائه

وللماكرو استخدامات عديدة و متقدمة في البرمجة لما يوفره من مستوى اعلى من المرونة والديناميكية في التنفيذ المبني على المجهول من المتغيرات والدوال.

تشتمل السوبرنوتا على امرين لهذا الغرض

```
Get Variable ....Variable/Value Point to Variable Name....
```

```
Do Function ....Variable/Value Point to Function Name....
```

الامر الاول هو Get Variable والذي يستخدم للوصول الى متغير من خلال متغير اخر قيمته هي اسم هذا المتغير الذي نريد الوصول اليه اما الامر الثانى فهو الامر Do Function والذي يستخدم لاستدعاء دالة او اجراء اسمه مخزن داخل متغير.

مما يميز الماكرو انه قابل للاستخدام فى اى expression وبالتالي فهو منفذ السوبرنوتا لاستدعاء الدوال داخل الـ expressions بشكل رسمى معلن



والان نبدا بمثال عملي للتعرف على الماكرو واستخدامه للوصول الى متغير

```
Application title = GetVariable.  
  
I want window contains textbox and window title = GetVariable.  
Window height = 130.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = GetVariable. and button width = 150.  
btn1 mouseclick. instructions are  
    Textbox value to [mytext].  
    Label caption from get variable (mytext).  
End of instructions
```

ونلاحظ من هذا المثال ان (get variable (mytext) هي نفسها تعادل [mytext] حيث ان الماكرو توصل الى اسم المتغير مباشرة من خلال قيمة ولكن كان من الممكن ان يكون الطريق اطول وان يتوصل الماكرو الى المتغير بدلالة متغير اخر يحتوى على اسم المتغير المطلوب

ونتيجة تنفيذ هذا التطبيق كالتالى



ونتابع الان لنرى مثال اخر

```
Application title = GetVariable2.  
  
I want window contains textbox and window title = GetVariable2.  
Window height = 130.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = GetVariable. and button width = 150.  
btn1 mouseclick. instructions are  
    Textbox value to [mytext].  
    Label caption from get variable (mytext) x get variable(mytext).  
End of instructions
```

ونتيجة تنفيذ هذا المثال كالتالي



والشفيرة المصدرية باللغة العربية كالتالي

التطبيق تحت عنوان هو احصل على متغير.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو احصل على متغير.  
بنافذة الارتفاع هو 150.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو "الحصول على متغير" ويازر العرض هو 150.  
"زر1 النقر بالفارة" تعليماته هي

يا مربع نص القيمة الى [ت1].

يا عنوان النص من المتغير (ت1) x المتغير (ت1).

نهاية التعليمات

والتطبيق وقت التشغيل كالتالي




وفى هذا المثال تم الحصول على قيمة من المستخدم من خلال مربع النص المعروف داخل النافذة ومن ثم تم تخزين هذه القيمة فى متغير والذي تم الحصول على القيمة التى بداخله باستخدام الماكرو ثم تم معاملتها على انها قيمة رقمية وتم ضرب القيمة الرقمية فى نفسها ثم اظهار الناتج

وفى التجربة تم ادخال الرقم 5 وكانت النتيجة هي الرقم 25 تماما كما هو متوقع

## والان نتعرف على مثال اخر

```
Application title = Pointer2Variable.  
  
I want window contains textbox and window title = Pointer2Variable.  
Window height = 130.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = GetVariable. and button width = 150.  
btn1 mouseclick. instructions are  
    textbox value to [mytext].  
    [mytext]. = Alltrim [mytext].  
    [myselector]. = (mytext).  
    label caption from get variable [myselector].  
End of instructions
```


فى هذا المثال هناك القليل من اللف والدوران حيث تم الوصول الى المتغير [mytext] من خلال المتغير [myselector] باستخدام الماكرو ونتيجة تنفيذ هذا البرنامج كالتالى:-



## والان نرى مثال يستعمل الماكرو لتنفيذ دالة

```
Application title = DoFunction.  
  
I want window contains textbox and window title = DoFunction.  
Window height = 130.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = DoFunction. and button width = 150.  
btn1 mouseclick. instructions are  
    label caption from do function (myfunc).  
End of instructions  
  
Function myfunc.  
    textbox value to [mytext].  
    [output]. = [mytext] x [mytext].  
End function
```


## والتطبيق فى وقت التشغيل كالتالى



## ونواصل لرؤية مثال اكثر تقدما

```
Application title = DoFunction2.  
I want window contains textbox and window title = DoFunction2.  
Window height = 130.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
Button caption = DoFunction. and button width = 150.  
btn1 mouseclick. instructions are  
    label caption from do function (myfunc) + do function  
    (myfunc).  
End of instructions  
Function myfunc.  
    textbox value to [mytext].  
    [output]. = [mytext] x [mytext].  
End function
```

## والتطبيق وقت التشغيل كالتالى




## ونتابع مع مثال اخر اكثر تقدما

```
Application title = Pointer2Function.  
I want window contains textbox and window title = Pointer2Function.  
Window height = 130.  
I want label and label top is 50.  
I want button and button name is btn1. and button left = 300.  
Button caption = DoFunction. and button width = 150.  
btn1 mouseclick. instructions are  
    [myfunctionname]. = (myfunc).  
    label caption from do function [myfunctionname].  
End of instructions  
Function myfunc.  
    textbox value to [mytext].  
    [output]. = [mytext] x [mytext].  
End function
```



## والتطبيق وقت التشغيل كالتالى



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو العدد مضروب فى نفسه.

انا اريد نافذة تحتوى مربع نص والنافذة تحت عنوان هو العدد مضروب فى نفسه.  
يانافذة الارتفاع هو 130.

انا اريد عنوان ويا عنوان البداية العلوية هي 50.

انا اريد زر ويازر الاسم هو "زر1" ويازر البداية اليسرى هي 300.

يازر النص هو "اضرب العدد فى نفسه" ويازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي

[اسم الدالة] = (دالتى).

يا عنوان النص من نفذ الدالة [اسم الدالة].

نهاية التعليمات

تعريف الدالة دالتى.

بامربع نص القيمة الى [ت1].

[الخرج] = [ت1] x [ت1].

نهاية الدالة

## والتطبيق وقت التنفيذ كالتالى

حيث تم ادخال الرقم 5 وتم الحصول على العدد مضروب فى نفسه  
اي  $25 = 5 \times 5$  كما هو متوقع



والان نرى مثال جيد يعطى مضروب رقم من خلال دالة لاتقوم باستدعاء نفسها اى  
Do While Loop وذلك عن طريق استعمال حلقة نفذ طالما No recursive

```
Application title = NoRecursive.  
  
I want window contains textbox and window title = NoRecursive.  
Window height = 130.  
  
I want label and label top is 50.  
  
I want button and button name is btn1. and button left = 300.  
Button caption = DoFunction. and button width = 150.  
  
btn1 mouseclick. instructions are  
  
    textbox value to [input].  
    label caption from do function (myfunc).  
  
End of instructions  
  
Function myfunc.  
  
    [output]. = [input].  
  
    Do while [input] > (1).  
  
        [input]. = [input] - (1).  
        [output]. = [output] x [input].  
  
    End while  
  
End function
```

والتطبيق وقت التشغيل كالتالى

حيث تم ادخال الرقم 5 وبالتالي حصلنا على النتيجة وهى مضروب الرقم خمسة  
وهو يساوى  $120 = 1 \times 2 \times 3 \times 4 \times 5$



والان لدينا مثال اخر لننهي به هذا الفصل حيث نقوم بكتابة برنامج يعطى مضروب رقم ولكن من خلال استدعاء الدالة لنفسها Recursive

```
Application title = Recursive.

I want window contains textbox and window title = Recursive.
Window height = 130.

I want label and label top is 50.

I want button and button name is btn1. and button left = 300.
Button caption = DoFunction. and button width = 150.
btn1 mouseclick. instructions are
    Textbox value to [input].
    Label caption from do function (myfunc).
End of instructions

Function myfunc.

    [output]. = [output] x [input].
    [input]. = [input] - (1).

    Do if [input] > (1).
        Do function myfunc.
    End if
End function
```

والتطبيق وقت التشغيل كالتالى

حيث كما فى المثال السابق تم ادخال الرقم 5 وبالتالي حصلنا على النتيجة وهى مضروب الرقم خمسة وهو يساوى  $120 = 1 \times 2 \times 3 \times 4 \times 5$





## دوال مكتبات الربط الديناميكية DLL Functions

هي الطريق المفتوح لزيادة امكانيات السوبرنوفيا حيث يتوفر لنا الالاف بل الملايين من الدوال المخزنة بداخل العديد من مكتبات الربط الديناميكية المتوفرة مع نظام التشغيل مايكروسوفت ويندوز MS-Windows والآخرى المنتشرة عبر الانترنت وتستخدم في مختلف الاغراض.

....data as expression.... DLL library to use  
....data as value.... DLL function to call with parameters ....data as value....  
With respect to DLL functions which return a value  
The value will be stored in [Output] variable.

وطريقة الاستخدام تكمن في تحديد مكتبة الربط الديناميكية (اسم ملف DLL) وذلك عن طريق الامر DLL Library to use ثم بعد ذلك يتم استدعاء دوال موجودة داخل هذه المكتبة عن طريق الامر DLL Function to call with parameters وهذا الامر يرسل المعطيات الى الدالة عن طريق مصفوفة ذات بعدين حيث تشتمل على عدد صفوف يساوى عدد المعطيات وتشتمل على اثنين من الاعمدة حيث ان العمود الاول يشتمل على نوع المتغير هل هو رقمى number ام حرفى string بينما العمود الثانى يشتمل على القيمة value .




والان نرى مثال يوضح ذلك حيث فى هذا المثال نستخدم مكتبة الربط الديناميكية winmm.dll والتي تشتمل على العديد من الدوال ونحن نستخدم الان الدالة PlaySound لتشغيل ملف صوت وهذا الملف هو sample.wav تم استعمال مصفوفة هى [myarray] وتم الاشارة الى انها تشتمل على ثلاثة عناصر اى تم تحديد عدد عناصر المصفوفة وتلك المعلومة مطلوبة من قبل الامل dll function to call with parameters حتى يقوم بتجميع المعطيات وارسالها الى الدالة بالصيغة المناسبة حسب نوع كل من المعطيات على حده هل هو حرفى string ام رقمى number ورغم ان السوبرنوكا لاتجعلك تلتفت الى هذه الامور (عدد عناصر المصفوفة - نوع المتغير) الا انها ضرورية هنا لاننا نتعامل مع وسط خارجى بعيد عن عالم السوبرنوكا وهو عالم مكتبات الربط الديناميكية القابلة للاستخدام من قبل معظم لغات البرمجة.

```
I want window
The window title is "Call Dll - Play Sound , File Sample.wav"

[myarray]. = (3).
[myarray]\(1)\(1). = (string). [myarray]\(1)\(2). = "(sample.wav)"
[myarray]\(2)\(1). = (number). [myarray]\(2)\(2). = (0).
[myarray]\(3)\(1). = (number). [myarray]\(3)\(2). = (0).
"(winmm.dll)" Dll Library to use
"PlaySound" dll function to call with parameters "myarray"
```

والتطبيق وقت التشغيل كالتالى حيث تظهر نافذة ويتم سماع الصوت



## والشفيرة المصدرية باللغة العربية كالتالى

التطبيق تحت عنوان هو استدعاء دالة من مكتبة ربط ديناميكى.

انا اريد نافذة و النافذة تحت عنوان هو "استدعاء دالة من مكتبة ربط ديناميكى"  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

[المصفوفة] = (3).  
[المصفوفة] (1) (1) = (حروف). [المصفوفة] (1) (2) = "sample.wav"  
[المصفوفة] (2) (1) = (رقم). [المصفوفة] (2) (2) = (0).  
[المصفوفة] (3) (1) = (رقم). [المصفوفة] (3) (2) = (0).

"(winmm.dll)" مكتبة ربط ديناميكى لكى نستعملها

"PlaySound" استدعاء دالة فى مكتبة ربط ديناميكى بالمعطيات "المصفوفة"

## والتطبيق وقت التشغيل كالتالى

حيث يتم اظهار نافذة التطبيق وتشغيل ملف الصوت





## مكتبة تشغيل الصوت Sound Player Library

هي عبارة عن مكتبة تشتمل مجموعة من الدوال التي تستخدم في تشغيل ملف صوت مع التحكم في عملية التشغيل من خلال الايقاف ومتابعة التشغيل.

Sound Player Library procedures

```
* MySoundOpen [input]. = "(filename.mp3)"  
* MySoundPlay  
* MySoundPause  
* MySoundResume  
* MySoundClose
```

وفي هذا الجزء من الكتاب سوف نتعرف اولا على الشفيرة المصدرية لمكتبة تشغيل الصوت والتي تم كتابتها بالسوبرنوكا وهي عبارة عن استدعاء مباشر لمجموعة من الدوال في ملف DLL ثم بعد ذلك نعرض مثال يستعمل المكتبة ويوضح خصائصها ان شاء الله رب العالمين



## والان نرى الشفرة المصدرية لمكتبة تشغيل الصوت SoundPlayerLib.nova

```
Procedure MySoundOpen.  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string).  
[temparray]\(1)\(2). = 'open "[input]" type mpegvideo alias MediaFile)'  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(nil)'  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
"(winmm.dll)" Dll Library to use  
"mciSendString" dll function to call with parameters "temparray"  
End procedure  
  
Procedure MySoundPlay.  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(play MediaFile)'  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(nil)'  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
"(winmm.dll)" Dll Library to use  
"mciSendString" dll function to call with parameters "temparray"  
End procedure  
  
Procedure MySoundPause.  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(pause MediaFile)'  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(nil)'  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
"(winmm.dll)" Dll Library to use  
"mciSendString" dll function to call with parameters "temparray"  
End procedure  
  
Procedure MySoundClose.  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(close MediaFile)'  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(nil)'  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
"(winmm.dll)" Dll Library to use  
"mciSendString" dll function to call with parameters "temparray"  
End procedure  
  
Procedure MySoundResume.  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(resume MediaFile)'  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(nil)'  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
"(winmm.dll)" Dll Library to use  
"mciSendString" dll function to call with parameters "temparray"  
End procedure
```

وينبغي الإشارة الى ان استيعاب محتوى الشفرة المصدرية للمكتبة ليس ضروريا حتى تتمكن من استخدامها في بناء وتطوير تطبيقاتك الخاصة.



## والان مع مثال يوضح كيفية استعمال المكتبة والاستفادة من خصائصها

```
Application title is Sound Player Library Demo.
```

```
Do file "SoundPlayerLib.nova"
```

```
I want window
```

```
The window title is "Sound Player Library Demo"
```

```
I want button and button caption is "open test.mp3" and button top is 50.
```

```
Button name is btn1.
```

```
I want button and button caption is play. and button top is 100.
```

```
Button name is btn2.
```

```
I want button and button caption is pause. and button top is 150.
```

```
Button name is btn3.
```

```
I want button and button caption is resume. and button top is 200.
```

```
Button name is btn4.
```

```
I want button and button caption is close. and button top is 250.
```

```
Button name is btn5.
```

```
btn1 mouse click. instructions are
```

```
    [input]. = "(test.mp3)" and do procedure MySoundOpen.
```

```
End of instructions
```

```
btn2 mouse click. instructions are
```

```
    do procedure MySoundPlay.
```

```
End of instructions
```

```
btn3 mouse click. instructions are
```

```
    do procedure MySoundPause.
```

```
End of instructions
```

```
btn4 mouse click. instructions are
```

```
    do procedure MySoundResume.
```

```
End of instructions
```

```
btn5 mouse click. instructions are
```

```
    do procedure MySoundClose.
```

```
End of instructions
```

## والتطبيق وقت التشغيل كالتالى





## مكتبة تسجيل الصوت Sound Recorder Library

هي عبارة عن مكتبة تشتمل على دوال لتسجيل الصوت وحفظه في ملف للاستماع اليه فيما بعد.

Sound Recorder Library procedures

---

```
* MySoundRecord  
* MySoundSave [input]. = "(filename.wav)"
```

وفي هذا الجزء من الكتاب سوف نتعرف اولا على الشفيرة المصدرية لمكتبة تسجيل الصوت والتي تم كتابتها بالسوبرنوبا وهي عبارة عن استدعاء مباشر لمجموعة من الدوال في ملف DLL ثم بعد ذلك نعرض مثال يستعمل المكتبة ويوضح خصائصها ان شاء الله رب العالمين



## والآن مع الشفيرة المصدرية لمكتبة تسجيل الصوت SoundRecorderLib.nova

```
Procedure MySoundrecord.  
  
  "(winmm.dll)" Dll Library to use  
  
  [temparray]. = (4).  
  
  [temparray]\(1)\(1). = (string).  
  [temparray]\(1)\(2). = '(open new Type waveaudio Alias recsound)'  
  
  [temparray]\(2)\(1). = (string). [temparray]\(2)\(2). = '()'  
  
  [temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
  
  [temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
  
  "mciSendString" dll function to call with parameters "temparray"  
  
  [temparray]\(1)\(1). = (string).  
  [temparray]\(1)\(2). = '(record recsound)'  
  
  "mciSendString" dll function to call with parameters "temparray"  
  
End procedure  
  
Procedure MySoundSave.  
  
  "(winmm.dll)" Dll Library to use  
  
  [temparray]. = (4).  
  
  [temparray]\(1)\(1). = (string).  
  [temparray]\(1)\(2). = '(save recsound ) [input]'  
  
  [temparray]\(2)\(1). = (string). [temparray]\(2)\(2). = '()'  
  
  [temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(0)'  
  
  [temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0)'  
  
  "mciSendString" dll function to call with parameters "temparray"  
  
  [temparray]\(1)\(1). = (string).  
  [temparray]\(1)\(2). = '(close recsound)'  
  
  "mciSendString" dll function to call with parameters "temparray"  
  
End procedure
```

وينبغي الإشارة الى ان استيعاب محتوى الشفيرة المصدرية للمكتبة ليس ضروريا حتى تتمكن من استخدامها في بناء وتطوير تطبيقاتك الخاصة.

## والان نرى مثال يستعمل المكتبة

```
Application title is Sound Recorder Library Demo.
```

```
Do file "SoundRecorderLib.nova"
```

```
I want window
```

```
The window title is "Sound Recorder Library Demo"
```


```
I want button and button caption is "Record" and button top is 50.  
Button name is btn1.
```

```
btn1 mouse click. instructions are  
do procedure MySoundRecord.  
End of instructions
```

```
I want button and button caption is "Save to test.wav" and button top is 100.  
Button name is btn2.
```

```
btn2 mouse click. instructions are  
[input]. = "(test.wav)" do procedure MySoundSave.  
End of instructions
```

والتطبيق وقت التشغيل كالتالي حيث عند الضغط على زر Record نقوم بالتحدث من خلال المايك وسوف يتم تسجيل الصوت من قبل التطبيق وعند الضغط على Save to test.wav سوف يتم حفظ الصوت في هذا الملف للاستماع اليه لاحقا عند الرغبة في ذلك





## مكتبة تشغيل الفيديو Movie Player Library

هي عبارة عن مكتبة تشتمل على دوال لتشغيل الفيديو مع التحكم في عملية التشغيل من خلال الايقاف ومتابعة التشغيل.

### Movie Player Library procedures

---

- \* MyMovieOpen  
[input] : filename.avi  
[input2]: Alias, MyMovie for example  
[input3]: window handle, use command (Window handle to)
- \* MyMoviePlay  
[input] : Alias  
[input2] : Movie region as : left top width height
- \* MyMoviePause  
[input] : Alias
- \* MyMovieResume  
[input] : Alias , MyMovie for example
- \* MyMovieClose  
[input] : Alias , MyMovie for example

وفي هذا الجزء من الكتاب سوف نتعرف اولا على الشفيرة المصدرية لمكتبة تشغيل الفيديو والتي تم كتابتها بالسوبرنوفيا وهي عبارة عن استدعاء مباشر لمجموعة من الدوال في ملف DLL ثم بعد ذلك نعرض مثال يستعمل المكتبة ويوضح خصائصها ان شاء الله رب العالمين



## والان مع الشفيرة المصدرية لمكتبة تشغيل الفيديو MoviePlayerLib.nova

```
Procedure MyMovieOpen.  
:note: " [input] = filename.avi [input2] = Alias [input3] = window handle"  
"(winmm.dll)" Dll Library to use  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(open ) [input] (" type  
AVIvideo alias ) [input2] ( style child parent ) [input3] '  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(0) '  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(128) '  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0) '  
"mciSendString" dll function to call with parameters "temparray"  
end procedure
```

```
Procedure MyMoviePlay.  
:note: " [input] = Alias [input2] = movie region entered as left top width  
height"  
"(winmm.dll)" Dll Library to use  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(put ) [input] (  
window client at ) [input2] '  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(0) '  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(128) '  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0) '  
"mciSendString" dll function to call with parameters "temparray"  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(play ) [input] '  
"mciSendString" dll function to call with parameters "temparray"  
End procedure
```

```
Procedure MyMoviePause.  
:note: " [input] = Alias "  
"(winmm.dll)" Dll Library to use  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(pause ) [input] '  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(0) '  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(128) '  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0) '  
"mciSendString" dll function to call with parameters "temparray"  
End procedure
```

```
Procedure MyMovieResume.  
:note: " [input] = Alias "  
"(winmm.dll)" Dll Library to use  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(resume ) [input] '  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(0) '  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(128) '  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0) '  
"mciSendString" dll function to call with parameters "temparray"  
End procedure
```

```
Procedure MyMovieClose.  
:note: " [input] = Alias "  
"(winmm.dll)" Dll Library to use  
[temparray]. = (4).  
[temparray]\(1)\(1). = (string). [temparray]\(1)\(2). = '(close ) [input] '  
[temparray]\(2)\(1). = (number). [temparray]\(2)\(2). = '(0) '  
[temparray]\(3)\(1). = (number). [temparray]\(3)\(2). = '(128) '  
[temparray]\(4)\(1). = (number). [temparray]\(4)\(2). = '(0) '  
"mciSendString" dll function to call with parameters "temparray"  
End procedure
```

## والان نرى مثال يوضح خصائص مكتبة تشغيل الفيديو

```
Application title is Movie Player Library Demo.
Do file "MoviePlayerLib.nova"
I want window and window back color is white. and window width is 700.
The window title is "Movie Player Library Demo"
I want button and button caption is "open test.avi" and button top is 50.
Button name is btn1.
I want button and button caption is play. and button top is 100.
Button name is btn2.
I want button and button caption is pause. and button top is 150.
Button name is btn3.
I want button and button caption is resume. and button top is 200.
Button name is btn4.
I want button and button caption is close. and button top is 250.
Button name is btn5.
btn1 mouse click. instructions are
    [input]. = "(test.avi)"
    [input2]. = (mymovie).
    window handle to [input3].
    do procedure MyMovieOpen.
End of instructions
btn2 mouse click. instructions are
    [input]. = (mymovie).
    [input2]. = (150 10 400 300).
    do procedure MyMoviePlay.
End of instructions
btn3 mouse click. instructions are
    [input]. = (mymovie). and do procedure MyMoviePause.
End of instructions
btn4 mouse click. instructions are
    [input]. = (mymovie). and do procedure MyMovieResume.
End of instructions
btn5 mouse click. instructions are
    [input]. = (mymovie). and do procedure MyMovieClose.
End of instructions
```

## والتطبيق وقت التشغيل كالتالى





## البرمجة بوصف الخيال

### Programming With Fiction Description

ان البرمجة بوصف الخيال اسلوب تعبير مبتكر تقدمه السوبرنوفنا لوصف التطبيقات بشكل سريع وعلى مستوى راقى من التعبير القريب من الانسان مقارنة بالاساليب السابق التعرف عليها فى لغات البرمجة الاخرى حيث من خلال البرمجة بوصف الخيال يمكن ان نحصل على نتائج تعادل استخدام جمل التحكم المتمثلة فى التفرع المشروط IF Statement والتكرار While Loop وايضا انشاء الكائنات Objects واسناد الخصائص اليها وايضا ادارة الاحداث Events Handling مثل الاحداث الناتجة عن التفاعل مع البيئة الرسومية GUI وكذلك تعريف احداث جديدة مثل تلك الناتجة عن وصول قيمة متغير الى قيمة معينة وكل ذلك على مستوى راقى من التعبير القريب الى الانسان وهذا يعطى لغة البرمجة سوبرنوفنا طابع مميز وهو متوقع من لغة برمجة تصنف ضمن لغات البرمجة من الجيل الخامس المختص باللغات الطبيعية وانظمة الذكاء الاصطناعى.

```
[1] What do you want ? (I Want ....)
    Example: I want Counter
[2] What do you remember? (The ...)
    Example: The label caption is Wow.
[3] What are the instructions? (... Instructions are)
    **Example**
    Manager. Instructions must done
    Manager. Instructions are
        I want window and Window title = wow.
    End of instructions
[4] What will happen when...? (Object Event. instructions are)
    **Example**
    I want window and the window name is win1.
    Win1 mouseclick. Instructions are
        You close window
    End of instructions
[5] What will happen if...? (Variable Value. instructions are)
    **Example**
    X 2. Instructions are
        i want window and window title = number 2.
    End of instructions
[x]. = (2).
```





## ما هي سمات البرمجة بوصف الخيال في لغة البرمجة سوبرنوبا ؟

تتسم البرمجة بوصف الخيال في لغة البرمجة سوبرنوبا بالنقاط الخمسة التالية:-

- التعبير عن الرغبة باستخدام I Want يليها اسم الشيء الذي نريده مثلا Window اذا كنا نريد نافذة او Counter اذا كنا نريد عداد
- الاشارة الى ما رغبتنا فيه على اعتبار انه اصبح موجود من خلال the الاختيارية يليها اسم الشيء الذي اردناه ثم نعبر اكثر بسرد خصائص هذا الشيء مثلا The window اشارة الى النافذة يليها title اشارة الى عنوانها يليها is hello. تحديد لهذا العنوان
- سرد التعليمات تحت مسمى ما مثلا Manager. Instructions are تشتير الى ان التعليمات التالية هي تعليمات المدير وتكون نهاية التعليمات اما بـ end of instructions او OK للاختصار حسب الرغبة.
- سرد تعليمات يتم تنفيذها عند حدوث شيء ما متوقع مثل Mouse click اي حدث النقر بالفارة ويتم ذلك بالاشارة الى اسم الكائن يليه اسم الحدث كعنوان للتعليمات مثلا win1 mouse click. Instructions are اشارة الى ان التعليمات القادمة ستنفذ عند الضغط بالفارة على النافذة التي تسمى win1
- سرد تعليمات يتم تنفيذها عندما تصل قيمة متغير الى قيمة معينة وذلك بذكر اسم المتغير ثم القيمة كعنوان للتعليمات مثلا 3. Instructions are x هي تعليمات تنفذ عندما تكون قيمة المتغير x تساوى 3.

في الاجزاء السابقة من الكتاب استعملنا ملامح البرمجة بوصف الخيال و تقريبا لا يخلو برنامج يتم كتابته بالسوبرنوبا من استخدام هذه الملامح ولكن في هذا الجزء من الكتاب نسلط الضوء على هذه الملامح بالعديد من الامثلة ان شاء الله رب العالمين.

### Fiction Description Samples

- [1] ClickLoop
- [2] ShutDown
- [3] Question
- [4] DownCounter
- [5] UpCounter
- [6] Counter
- [7] Counter2
- [8] Math
- [9] Math2
- [10] Logic
- [11] Logic2
- [12] CounterRange
- [13] DynamicObjects
- [14] Cases
- [15] CheckAnswer
- [16] Loops

نبدأ بمثال ClickLoop والذي يعرض لنا نافذة عنوانها اضغط على حتى تحصل على نافذة اخرى لون خلفيتها ابيض – وعند عمل ذلك نحصل على نافذة لون خلفيتها ابيض وعنوانها اضغط على لتحصل على نافذة لون خلفيتها احمر وعند عمل ذلك نحصل على نافذة لون خلفيتها احمر وعنوانها اضغط على ليتم انهاء التطبيق. بالمناسبة يمكن التعبير عن اللون الابيض ك {255,255,255} او نكتب white يمكن التعبير عن اللون الاحمر ك {255,0,0} او نكتب red حسب ما نرتاح اليه.

```
Set window auto show off


I want window and the window name is mywin.
The window title is click me to get a new white window.
The window width is 350. and the window height is 200.
window top = 50.
You Show Window

mywin mouseclick. instructions are
    I want window and the window name is mywin2.
    window bgcolor = {255,255,255}. and window top = 100.
    The window title is click me to get a new red window.
    The window width is 350. and the window height is 200.
    You Show Window
End of instructions


mywin2 mouseclick. instructions are
    I want window and the window bgcolor = {255,0,0}.
    window title = hello. and window top = 150.
    The window title is click me to close the application.
    The window name is mywin3.
    The window width is 350. and the window height is 200.
    You Show Window
End of instructions

mywin3 mouseclick. instructions are
    You close application
End of instructions
```

والتطبيق وقت التشغيل كالتالي



وهذه لقطة اخرى حيث تم انشاء العديد من النوافذ



من هذا المثال يتضح لنا ان الشفرة المصدرية التي اعتمدت على مفهوم البرمجة بوصف الخيال قامت بوصف مايقوم به التطبيق ومن ثم حصلنا على النتيجة بهذا الشكل المنطقي الذي نراه حيث اكتسبت النوافذ البيضاء خاصية انشاء نوافذ حمراء واكتسبت النوافذ الحمراء خاصية اغلاق التطبيق وذلك من خلال اسم النافذة الذي اصبح مشتركا بين جميع هذه النوافذ التي ملأت سطح المكتب.

**هل من المنطق ان تحمل اكثر من نافذة نفس الاسم ؟**

اعلم ان هذه القاعدة قد يتعجب لها الكثير من المبرمجين لانه في معظم لغات البرمجة الاخرى هذا شيء غير منطقي وغير مسموح به ويسبب خطأ لكن هنا في السوبرنوفامع البرمجة بوصف الخيال الامر يختلف لان الخيال يسمح لنا بذلك وانظر الى النتيجة بنفسك لقد تخيلنا اداء التطبيق ووصفنا هذا الاداء بمرونة واصبح الخيال واقع ولا داعي ان يكون الخيال مقيدا بالاساليب البرمجية المتعارف عليها في لغات لاتقوم على وصف الخيال فالاسلوب مختلف وبالتالي القواعد مختلفة نسبيا.

ننتقل الى مثال اخر وهو Shutdown والذي يشتمل على زر Close عند الضغط عليه نحصل على رسالة تأكيد وزرين لاتخاذ القرار اما بالانهاء او الاستمرار.

```
I want window and the window title is Welcome.

I want button and button caption equal Close.
The button top = 50. and button name = btn1.


btn1 mouseclick. instructions are
    I want label and label caption = Really!!!.
    The label top = 100.
    I want button and button caption = Yes.
    The button top = 150. and button name = btn2.
    I want button and button caption = No.
    The button top = 150. and button name = btn3.
    The button left = 150.

End of instructions

btn2 mouseclick. instructions are you close window ok

btn3 mouseclick. instructions are label caption is Great. ok
```

والتطبيق وقت التشغيل كالتالى



هنا تتضح لنا الديناميكية المتوفرة مع البرمجة بوصف الخيال حيث ان رسالة التاكيد وازرار اتخاذ القرار تم انشائها بشكل ديناميكي و فقط تم انشائها عند الضغط على زر .close

والسؤال الان

ماذا يحدث عند الضغط على زر Close ثم No ثم Close مرة اخرى؟

عند الضغط على زر Close اول مرة تظهر رسالة التاكيد وازرار اتخاذ القرار

ويكون نص رسالة التاكيد Really!!!

وعند الضغط على زر No يصبح نص الرسالة Great

وعند الضغط على زر Close مرة اخرى يحدث ماتخيلنا ان يقوم به الزر عند الضغط عليه حيث تظهر رسالة التاكيد مرة اخرى مع ازرار اتخاذ القرار بدون اى مشاكل حيث من جديد يتم انشاء الرسالة وازرار اتخاذ القرار بشكل ديناميكي ويتم اسناد الخصائص اليهما.

هل ما يحدث الان سليما تماما من الناحية البرمجية؟

## ليس سليما بشكل كامل

عند التامل لأول مرة ربما نظن ان السبب وراء عدم الصحة انه مع تكرار الضغط على زر Close سيتم تكرار انشاء الكائنات (العنوان وازرار اتخاذ القرار) وهذا اهدار للذاكرة Memory ومن ثم فان التطبيق الذى حصلنا عليه لايحسن استغلال موارد الحاسب جيدا.

فى الواقع من ناحية مفهوم البرمجة بوصف الخيال فان ما حدث هو ماتخيلناه وهو يعمل كما تخيلنا بشكل سليم وليست المشكلة الحقيقية هى ادارة الموارد بشكل سلبي كما هو الحال الان عند تكرار انشاء الكائنات بدون داعى وانما المشكلة فى ان زر Close مازال يعمل بعد الضغطة الاولى عليه وهذا ماكان يجب ان لانتخيله من الاساس عن طريق تخيل العكس (بان الزر لم يعد يعمل بعد الضغطة الاولى) .

## والان مع المثال Question والذي يسأل المستخدم عن الاسم ثم يظهر رسالة ترحيبية.

```
I want window and the window title is Welcome.  
Window width = 250. and window height = 350.
```

```
I want label and the label top = 50. and label width = 200.  
the label caption = What is your name ?.
```


```
I want textbox and textbox top = 100.  
textbox back color = {0,255,0}.
```

```
I want button and button caption equal Say Hello.  
The button top = 150. and button name = btn1.  
btn1 mouseclick. instructions are  
    textbox value to [name].  
    the window title from (Hello ) [name].  
end of instructions
```

```
I want button and button caption equal Close.  
the button top = 200. and button name = btn2.  
btn2 mouseclick. instructions are You Close window ok
```

## والتطبيق وقت التشغيل كالتالى

حيث قمت بادخال اسمى فى مربع النص المعرف داخل النافذة وضغطت الزر Say Hello فتغير عنوان النافذة للترحيب بى.




لدينا الان بعض الامثلة توضح كيفية عمل عداد يدوى حيث هناك رقم وعند الضغط على زر يتغير هذا الرقم اما بالزيادة فى حالة UP Counter او بالنقصان فى حالة Down Counter

مثلا Down Counter كالتالى

```
I want window and the window title is Down Counter.  
Window width = 250. and window height = 250.  
  
I want textbox and textbox top = 30. and textbox back color = {0,255,0}.  
  
I want button and button caption equal Count.  
The button top = 70. and button name = btn1.  
  
btn1 mouseclick. instructions are  
    textbox value to [num1]. and textbox value from [num1]-(1).  
End of instructions  
  
I want button and button caption equal Close.  
The button top = 130. and button name = btn2.  
btn2 mouseclick. instructions are You Close window ok
```

والتطبيق وقت التشغيل كالتالى  
حيث تم ادخال القيمة (-4) فى مربع النص المعرف داخل نافذة التطبيق ثم تم الضغط على الزر Count فتغيرت القيمة فى مربع النص لتصبح (-5) اى نقصت بمقدار 1.



## ويتم عمل UP Counter بطريقة مشابهة كالتالى

I want window and the window title is Up Counter.  
Window width = 250. and window height = 250.

I want textbox and textbox top = 30. and textbox back color = {0,255,0}.

I want button and button caption equal Count.  
The button top = 70. and button name = btn1.


btn1 mouseclick. Instructions are  
textbox value to [num1]. and textbox value from [num1]+(1).  
End of instructions

I want button and button caption equal Close. and the button top = 130.  
button name = btn2.

btn2 mouseclick. Instructions are You Close window ok

## والتطبيق وقت التشغيل كالتالى

حيث تم ادخال القيمة (1) فى مربع النص المعرف داخل نافذة التطبيق ثم تم الضغط على الزر Count فتغيرت القيمة فى مربع النص لتصبح (2) اى زادت بمقدار 1.





## والان مع التطبيق Counter الذي يدمج المثالين السابقين معا

### Application Files

- 1 - CounterApp.nova
- 2 - UpCounter.nova
- 3 - DownCounter.nova

### CounterApp.nova

"upcounter.nova" file instructions must done  
"downcounter.nova" file instructions must done

### UpCounter.nova

I want window and the window title is Up Counter.  
Window width = 250. and window height = 250.

I want textbox and textbox top = 30. and textbox back color = {0,255,0}.  
The textbox name is up\_text1.

I want button and button caption equal Count.  
The button top = 70. and button name = up\_btn1.

up\_btn1 mouseclick. Instructions are  
    up\_text1. textbox selected  
    Textbox value to [num1]. and textbox value from [num1]+(1).  
End of instructions

I want button and button caption equal Close. and the button top = 130.  
Button name = up\_btn2.

up\_btn2 mouseclick. Instructions are You Close window ok

### DownCounter.nova

I want window and the window title is Down Counter.  
Window left = 350. Window width = 250. and window height = 250.

I want textbox and textbox top = 30. and textbox back color = {0,255,0}.  
The textbox name is down\_Text1.

I want button and button caption equal Count.  
The button top = 70. and button name = down\_btn1.


down\_btn1 mouseclick. instructions are  
    down\_text1. textbox selected  
    textbox value to [num1]. and textbox value from [num1]-(1).  
End of instructions

I want button and button caption equal Close.  
The button top = 130. and button name = down\_btn2.

down\_btn2 mouseclick. Instructions are You Close window ok

نلاحظ من الشفيرة المصدرية انها عبارة عن ثلاثة ملفات حيث هناك ملف رئيسي وملف لـ UP Counter وملف اخر لـ Down Counter وتقسيم التطبيق لاكثر من ملف يعطى لمسة من التنظيم مفيدة جدا عند الانطلاق فى عمل تطبيقات ضخمة حتى تكون الشفيرة المصدرية لهذه التطبيقات اكثر تنظيما وايضا عندما يتم التطوير فى نفس التطبيق من قبل اكثر من شخص حيث يقوم كل فرد ببناء جزء معين من التطبيق فى ملف وايضا عندما يتم استخدام مكتبات قمنا بعملها او قام اخرون بعملها باستخدام السوبرنوفافا.

والتطبيق وقت التشغيل كالتالى



والسؤال الان

**ماهى النافذة الرئيسية للتطبيق التى عند اغلاقها يتم اغلاق التطبيق ؟**

هذا ما اعتدنا عليه فى معظم لغات البرمجة الاخرى حيث عند بناء التطبيقات الرسومية GUI التى تقوم على استعمال النوافذ Windows يكون هناك نافذة رئيسية للتطبيق والتى عند اغلاقها يتم اغلاق التطبيق وتكون هى نقطة البداية والانطلاق.

مع البرمجة بوصف الخيال الوضع يختلف حيث ان وجود النافذة الرئيسية بشكل تلقائى يقيد الخيال بدون ضرورة ويمنعه من الانطلاق كما نريد حيث لا يوجد ما يمنع ان يستمر التطبيق فى العمل وان ينتهى فقط عند اغلاق كلا النافذتين الممثلتين للتطبيق الا يقبل الخيال ذلك !!

بلى يقبل وهذا ما يحدث هنا بالفعل حيث ان اغلاق اى نافذة لا يودى الى انتهاء التطبيق بشكل تلقائى الا اذا كانت هذه النافذة هى النافذة الاخيرة على قيد الحياة

والان نرى تطورا للمثال السابق وهو Counter2

فى هذا المثال الجديد تم اتباع الاسلوب الشائع فى ان يكون للتطبيق نافذة رئيسية والتي عند اغلاقها يتم اغلاق التطبيق باستخدام الامر `You Close Application` والذي يمكن ان يكتب ايضا على الصورة `Please Close Application` حيث ان كل من الكلمتين `You` و `Please` قابلتين للاستبدال من قبل مترجم اللغة.

ومتابعة لحفلة الاستبدالات فان اللون الابيض `{255,255,255}` يمكن ان يستبدل بكلمة `White` وايضا الامر `file instructions must done` يمكن ان يستبدل بالامر المختصر `do file` يليه اسم الملف من النوع `Nova` وكذلك الامر `end of instructions` يمكن ان يستبدل بالامر `Ok` وذلك للاختصار والتسهيل.

#### Application Files

- 1 - Counterapp2.nova
- 2 - UpCounter.nova
- 3 - DownCounter.nova

#### Counterapp2.nova

```
"downcounter.nova" file instructions must done
"upcounter.nova" file instructions must done
```

```
The application title is "Counter (2)"
```

```
I want window
The window title is "Counter2"
The window backcolor is {255,255,255}.
The window name is the main window.
```

```
I want menubar
I want menu popup and menu popup caption is "File"
  I want menu item and the menu item caption is "Up Counter"
  The menu item name is myitem1.
  myitem1 mouseclick. instructions are
    myup. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Down Counter"
  The menu item name is myitem2.
  myitem2 mouseclick. instructions are
    mydown. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Exit"
  The menu item name is myitem3.
  myitem3 mouseclick. instructions are
    You close application
  End of instructions
```

```
I want statusbar and i want status item and the status item caption is Ready.
```

```
You maximize window
```

## وفيما يلي تابع الشفيرة المصدرية

```
UpCounter.nova
```

```
-----
```

```
myup. instructions are
```

```
i want window and the window title is Up Counter.  
Window width = 250. and window height = 250. and window top = 100.  
The window parent is the main window.
```

```
i want textbox and textbox top = 30. and textbox back color = {0,255,0}.  
the textbox name is up_text1.
```

```
i want button and button caption equal Count.  
The button top = 70. and button name = up_btn1.
```

```
up_btn1 mouseclick. instructions are  
    up_text1. textbox selected  
    textbox value to [num1]. and textbox value from [num1]+(1).  
end of instructions
```

```
i want button and button caption equal Close.  
the button top = 130. and button name = up_btn2.  
the up_btn2 mouseclick. instructions are You Close window ok
```

```
ok
```

```
DownCounter.nova
```

```
-----
```

```
mydown. instructions are
```

```
i want window and the window title is Down Counter.  
window top = 100. and window left = 350.  
Window width = 250. and window height = 250.
```

```
The window parent is the main window.
```

```
i want textbox and textbox top = 30. and textbox back color = {0,255,0}.  
the textbox name is down_Text1.
```

```
i want button and button caption equal Count.  
The button top = 70. and button name = down_btn1.
```

```
down_btn1 mouseclick. instructions are  
    down_text1. textbox selected  
    textbox value to [num1]. and textbox value from [num1]-(1).  
end of instructions
```


```
i want button and button caption equal Close.  
the button top = 130. and button name = down_btn2.  
the down_btn2 mouseclick. instructions are You Close window ok
```

```
ok
```

ونلاحظ ان الشفيرة المصدرية داخل كل ملف وضعت داخل اجراء ليتم استدعائه من النافذة الرئيسية عند الطلب من قبل المستخدم.

والتطبيق وقت العمل كالتالى

حيث تم فتح اكثر من Up Counter واكثر من Down Counter فى نفس الوقت.



ونلاحظ المرونة المتمثلة فى امكانية استدعاء الـ Up Counter والـ Down Counter اكثر من مرة وفى كل مرة نحصل على نافذة جديدة مستقلة لاداء المهمة.

وجميع هذه النوافذ التى يتم انشائها تصبح داخل اطار النافذة الرئيسية وذلك تاثير الامر The window parent is الذى يتم استعماله مع النوافذ لوضع نافذة داخل نافذة اخرى

ومن هذا المثال تظهر روعة البرمجة بوصف الخيال حيث حصلنا على تطبيق مرن ومنظم بسهولة وبدون الدخول فى مفاهيم برمجية معقدة.

بنفس الاسلوب الذي تم به تقديم Counter و Counter2 لدينا اربعة امثلة وهي Math و Math2 لاجراء العمليات الحسابية البسيطة (الجمع والطرح والضرب والقسمة) ولدينا Logic و Logic2 لاجراء العمليات المنطقية AND و OR و NOT وفيما يلي الشفيرة المصدرية وصور للتطبيقات وقت التشغيل.

#### Application Files

- 1 - MathApp.nova
- 2 - SumApp.nova
- 3 - MinApp.nova
- 4 - MulApp.nova
- 5 - DivApp.nova

#### MathApp.nova

```
"sumapp.nova" file instructions must done
"minapp.nova" file instructions must done
"mulapp.nova" file instructions must done
"divapp.nova" file instructions must done
```

#### SumApp.nova

Set Window Auto Show Off

```
I want window and the window title is Sum.
Window top = 20. and window left = 50.
Window width = 250. and Window height = 250.
```

```
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
```

```
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
```

```
I want textbox and the textbox name = sum_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
```

```
I want textbox and the textbox name = sum_txt2.
Textbox top = 100. and textbox left = 100. and textbox width = 50.
```

```
I want button and button caption = do operation.
Button top = 150. and button name = sum_btn1.
sum_btn1 mouseclick. instructions are
    sum_txt1. textbox selected and store textbox value to [num1].
    sum_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]+[num2].
End of instructions
```

```
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = Sum_btnClose.
sum_btnclose mouseclick. instructions are you close window ok
```

#### You Show Window

MinApp.nova

-----

```

set window auto show off
I want window and the window title is Min.
Window top = 20. and window left = 350.
Window width = 250. and Window height = 250.
Window backcolor = {255,0,0}.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = min_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = min_txt2.
Textbox top = 100. and textbox left = 100.and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = min_btn1.
min_btn1 mouseclick. instructions are
    min_txt1. textbox selected and store textbox value to [num1].
    min_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]-[num2].
End of instructions
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = min_btnClose.
min_btnclose mouseclick. instructions are you close window ok
You show window

```

MulApp.nova

-----

```

set window auto show off
I want window and the window title is Mul.
Window top = 300. and window left = 50.
Window width = 250. and Window height = 250.
Window Back Color = {0,255,0}.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = mul_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = mul_txt2.
Textbox top = 100. and textbox left = 100.and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = mul_btn1.
mul_btn1 mouseclick. instructions are
    mul_txt1. textbox selected and store textbox value to [num1].
    mul_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]x[num2].
End of instructions
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = mul_btnClose.
mul_btnclose mouseclick. instructions are you close window ok
You show window

```

## تابع الشفيرة المصدرية

```
DivApp.nova
```

```
-----
```

```
Set Window Auto Show off
```

```
I want window and the window title is Div.  
Window top = 300. and window left = 350.  
Window width = 250. and Window height = 250.  
Window Back Color = {0,0,255}.
```

```
I want label and the label caption is Number (1).  
Label top = 50. and label width = 80.  
I want label and the label caption is Number (2).  
Label top = 100. and label width = 80.
```

```
I want textbox and the textbox name = div_txt1.  
Textbox top = 50. and textbox left = 100. and textbox width = 50.
```

```
I want textbox and the textbox name = div_txt2.  
Textbox top = 100. and textbox left = 100. and textbox width = 50.
```

```
I want button and button caption = do operation.
```

```
Button top = 150. and button name = btn1.
```

```
btn1 mouseclick. instructions are
```

```
    div_txt1. textbox selected and store textbox value to [num1].
```

```
    div_txt2. textbox selected and store textbox value to [num2].
```

```
    The window title from [num1]/[num2].
```

```
End of instructions
```


```
I want button and button caption = Close.
```

```
Button top = 150. and button left = 130. and button name = btnClose.
```

```
btnClose mouseclick. instructions are you close window ok
```

```
You Show Window
```

## والتطبيق وقت التشغيل كالتالى






ومن خلال التجربة بادخال قيم و اظهار الناتج نتأكد من صحة التطبيق كما يلي:-

بالنسبة لنافذة الجمع Sum فقد تم ادخال الرقم الاول (5) وتم ادخال الرقم الثاني (4) ثم تم الضغط على زر do operation ليتغير عنوان نافذة التطبيق ليعرض الناتج 9

اما بالنسبة لنافذة الطرح Min تم ادخال الرقم الاول (12) وتم ادخال الرقم الثاني (3) ثم تم الضغط على زر do operation ليتغير عنوان نافذة التطبيق ليعرض الناتج 9

و بالنسبة لنافذة الضرب Mul تم ادخال الرقم الاول (9) وتم ادخال الرقم الثاني (1) ثم تم الضغط على زر do operation ليتغير عنوان نافذة التطبيق ليعرض الناتج 9

اما بالنسبة لنافذة القسمة Div تم ادخال الرقم الاول (81) وتم ادخال الرقم الثاني (9) ثم تم الضغط على زر do operation ليتغير عنوان نافذة التطبيق ليعرض الناتج 9



والآن مع Math2 الذي يعتبر تطوير للمثال السابق ليحصل التطبيق على طابع فني ارقى من خلال نافذة رئيسية للبرنامج مع امكانية الحصول على العديد من النوافذ التي تعمل في وقت واحد بدون مشاكل.

والشفيرة المصدرية للتطبيق كالتالى

#### Application Files

- 1 - MathApp2.nova
- 2 - SumApp.nova
- 3 - MinApp.nova
- 4 - MulApp.nova
- 5 - DivApp.nova

#### MathApp2.nova

```
"sumapp.nova" file instructions must done
"minapp.nova" file instructions must done
"mulapp.nova" file instructions must done
"divapp.nova" file instructions must done
The application title is "Math (2)"
I want window
The window title is "Math2"
The window bgcolor is {255,255,255}.
The window name is the main window.
I want menubar
I want menu popup and menu popup caption is "File"
  I want menu item and the menu item caption is "Sum (+)"
  The menu item name is myitem1.
  myitem1 mouseclick. instructions are
    mysum. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Min (-)"
  The menu item name is myitem2.
  myitem2 mouseclick. instructions are
    mymin. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Mul (x)"
  The menu item name is myitem3.
  myitem3 mouseclick. instructions are
    mymul. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Div (/)"
  The menu item name is myitem4.
  myitem4 mouseclick. instructions are
    mydiv. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Exit"
  The menu item name is myitem5.
  myitem5 mouseclick. instructions are
    You close application
  End of instructions

I want statusbar and i want status item and the status item caption is Ready.

You maximize window
```

```

SumApp.nova
-----
mysum. instructions are
Set Window Auto Show Off
I want window and the window title is Sum.
Window top = 20. and window left = 50.
Window width = 250. and Window height = 250.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = sum_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = sum_txt2.
Textbox top = 100. and textbox left = 100. and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = sum_btn1.
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = Sum_btnClose.
The window parent is the main window.
You Show Window
end of instructions
sum_btn1 mouseclick. instructions are
    sum_txt1. textbox selected and store textbox value to [num1].
    sum_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]+[num2].
End of instructions
sum_btnclose mouseclick. instructions are you close window ok

MinApp.nova
-----
mymin. instructions are
set window auto show off
I want window and the window title is Min.
Window top = 20. and window left = 350.
Window width = 250. and Window height = 250.
Window bgcolor = {255,0,0}.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = min_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = min_txt2.
Textbox top = 100. and textbox left = 100. and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = min_btn1.
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = min_btnClose.
The window parent is the main window.
you show window
end of instructions
min_btn1 mouseclick. instructions are
    min_txt1. textbox selected and store textbox value to [num1].
    min_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]-[num2].
End of instructions

min_btnclose mouseclick. instructions are you close window ok

```


```

MulApp.nova
-----
mymul. instructions are
set window auto show off
I want window and the window title is Mul.
Window top = 300. and window left = 50.
Window width = 250. and Window height = 250.
Window Back Color = {0,255,0}.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = mul_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = mul_txt2.
Textbox top = 100. and textbox left = 100.and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = mul_btn1.
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = mul_btnClose.
The window parent is the main window.
you show window
end of instructions
mul_btn1 mouseclick. instructions are
    mul_txt1. textbox selected and store textbox value to [num1].
    mul_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]x[num2].
End of instructions
mul_btnclose mouseclick. instructions are you close window ok

DivApp.nova
-----
mydiv. instructions are
Set Window Auto Show off
I want window and the window title is Div.
Window top = 300. and window left = 350.
Window width = 250. and Window height = 250.
Window Back Color = {0,0,255}.
I want label and the label caption is Number (1).
Label top = 50. and label width = 80.
I want label and the label caption is Number (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = div_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = div_txt2.
Textbox top = 100. and textbox left = 100.and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = btn1.
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = btnClose.
The window parent is the main window.
You Show Window
end of instructions
btn1 mouseclick. instructions are
    div_txt1. textbox selected and store textbox value to [num1].
    div_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1]/[num2].
End of instructions
btnclose mouseclick. instructions are you close window ok

```

## والتطبيق وقت التشغيل كالتالى



والان لدينا المثال Logic والذي يوضح اجراء العمليات المنطقية الاساسية وهى AND و OR و NOT والشيفرة المصدرية لهذا المثال كالتالى

### Application Files

1 - LogicApp.nova 2 - AndApp.nova 3 - OrApp.nova 4 - NotApp.nova

#### LogicApp.nova

```
"andapp.nova" file instructions must done
"orapp.nova" file instructions must done
"notapp.nova" file instructions must done
```

#### AndApp.nova

```
Set Window Auto Show Off
I want window and the window title is And.
Window top = 20. and window left = 350.
Window width = 250. and Window height = 250.
I want label and the label caption is Value (1).
Label top = 50. and label width = 80.
I want label and the label caption is Value (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = log_and_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = log_and_txt2.
Textbox top = 100. and textbox left = 100.and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = log_and_btn1.
log_and_btn1 mouseclick. instructions are
    log_and_txt1. textbox selected and store textbox value to [num1].
    log_and_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1] and [num2].
End of instructions
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = log_and_btnClose.
log_and_btnclose mouseclick. instructions are you close window ok
You Show Window
```

OrApp.nova

-----

Set Window Auto Show Off

I want window and the window title is OR.

Window top = 20. and window left = 50.

Window width = 250. and Window height = 250.

I want label and the label caption is Value (1).

Label top = 50. and label width = 80.

I want label and the label caption is Value (2).

Label top = 100. and label width = 80.

I want textbox and the textbox name = OR\_txt1.

Textbox top = 50. and textbox left = 100. and textbox width = 50.

I want textbox and the textbox name = OR\_txt2.

Textbox top = 100. and textbox left = 100. and textbox width = 50.

I want button and button caption = do operation.

Button top = 150. and button name = OR\_btn1.

OR\_btn1 mouseclick. instructions are

    OR\_txt1. textbox selected and store textbox value to [num1].

    OR\_txt2. textbox selected and store textbox value to [num2].

    The window title from [num1] OR [num2].

End of instructions

I want button and button caption = Close.

Button top = 150. and button left = 130. and button name = OR\_btnClose.

OR\_btnclose mouseclick. instructions are you close window ok

You Show Window

NotApp.nova

-----

Set Window Auto Show Off

I want window and the window title is Not.

Window top = 300. and window left = 50.

Window width = 250. and Window height = 250.

I want label and the label caption is Value.

Label top = 50. and label width = 80.

I want textbox and the textbox name = Not\_txt1.

Textbox top = 50. and textbox left = 100. and textbox width = 50.

I want button and button caption = do operation.

Button top = 150. and button name = Not\_btn1.

Not\_btn1 mouseclick. instructions are

    Not\_txt1. textbox selected and store textbox value to [num1].

    The window title from not [num1].

End of instructions


I want button and button caption = Close.

Button top = 150. and button left = 130. and button name = Not\_btnClose.

Not\_btnclose mouseclick. instructions are you close window ok

You Show Window

والتطبيق وقت التشغيل كالتالى



حيث نجد ثلاثة نوافذ نافذة لعملية AND و نافذة لعملية OR و نافذة لعملية NOT والتطبيق ينتهى عند اغلاق جميع النوافذ حيث لا يوجد نافذة رئيسية للتطبيق

وللتذكرة الـ AND تعطى خرج True فى حالة كون جميع قيم الدخل True

بينما الـ OR تعطى خرج True فى حالة كون اى قيمة من قيم الدخل True

اما الـ NOT فهى تعطى خرج عكس الدخل اى تنفى الدخل فاذا كان الدخل True فان الناتج هو False واذا كان الدخل False فان الناتج هو True

والآن نتابع لرؤية المثال السابق بعد التطوير كما هو متوقع لدينا الآن Logic2

والشفيرة المصدرية كالتالى

#### Application Files

---

- 1 - LogicApp2.nova
- 2 - AndApp.nova
- 3 - OrApp.nova
- 4 - NotApp.nova

#### LogicApp2.nova

---

```
"andapp.nova" file instructions must done
"orapp.nova" file instructions must done
"notapp.nova" file instructions must done
```

```
The application title is "Logic (2)"
```

```
I want window
The window title is "Logic2"
The window backcolor is {255,255,255}.
The window name is the main window.
```

```
I want menubar
I want menu popup and menu popup caption is "File"
  I want menu item and the menu item caption is "And"
  The menu item name is myitem1.
  myitem1 mouseclick. instructions are
    myand. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Or"
  The menu item name is myitem2.
  myitem2 mouseclick. instructions are
    myor. instructions must done
  End of instructions
  I want menu item and the menu item caption is "Not"
  The menu item name is myitem3.
  myitem3 mouseclick. instructions are
    mynot. instructions must done
  End of instructions
```

```
  I want menu item and the menu item caption is "Exit"
  The menu item name is myitem4.
  myitem4 mouseclick. instructions are
    You close application
  End of instructions
```

```
I want statusbar and i want status item and the status item caption is Ready.
```



```

AndApp.nova
-----
myand. instructions are
Set Window Auto Show Off
I want window and the window title is And.
Window top = 20. and window left = 350.
Window width = 250. and Window height = 250.
I want label and the label caption is Value (1).
Label top = 50. and label width = 80.
I want label and the label caption is Value (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = log_and_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = log_and_txt2.
Textbox top = 100. and textbox left = 100. and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = log_and_btn1.
log_and_btn1 mouseclick. instructions are
    log_and_txt1. textbox selected and store textbox value to [num1].
    log_and_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1] and [num2].
End of instructions
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = log_and_btnClose.
log_and_btnclose mouseclick. instructions are you close window ok
The window parent is the main window.
You Show Window
end of instructions

OrApp.nova
-----
myor. instructions are
Set Window Auto Show Off
I want window and the window title is OR.
Window top = 20. and window left = 50.
Window width = 250. and Window height = 250.
I want label and the label caption is Value (1).
Label top = 50. and label width = 80.
I want label and the label caption is Value (2).
Label top = 100. and label width = 80.
I want textbox and the textbox name = OR_txt1.
Textbox top = 50. and textbox left = 100. and textbox width = 50.
I want textbox and the textbox name = OR_txt2.
Textbox top = 100. and textbox left = 100. and textbox width = 50.
I want button and button caption = do operation.
Button top = 150. and button name = OR_btn1.
OR_btn1 mouseclick. instructions are
    OR_txt1. textbox selected and store textbox value to [num1].
    OR_txt2. textbox selected and store textbox value to [num2].
    The window title from [num1] OR [num2].
End of instructions
I want button and button caption = Close.
Button top = 150. and button left = 130. and button name = OR_btnClose.
OR_btnclose mouseclick. instructions are you close window ok
The window parent is the main window.
You Show Window
end of instructions


```

## تابع الشفرة المصدرية

NotApp.nova

```
-----  
  
mynot. instructions are  
Set Window Auto Show Off  
I want window and the window title is Not.  
Window top = 270. and window left = 50.  
Window width = 250. and Window height = 200.  
I want label and the label caption is Value.  
Label top = 50. and label width = 80.  
I want textbox and the textbox name = Not_txt1.  
Textbox top = 50. and textbox left = 100. and textbox width = 50.  
I want button and button caption = do operation.  
Button top = 100. and button name = Not_btn1.  
Not_btn1 mouseclick. instructions are  
    Not_txt1. textbox selected and store textbox value to [num1].  
    The window title from not [num1].  
End of instructions  
I want button and button caption = Close.  
Button top = 100. and button left = 130. and button name = Not_btnClose.  
Not_btnclose mouseclick. instructions are you close window ok  
The window parent is the main window.  
You Show Window  
end of instructions
```

## والتطبيق وقت التشغيل كالتالى



والان ننطلق لنرى مثالين حول استخدام العداد Counter الذي يعد من رقم الى رقم اخر من خلال خطوة (رقم يضاف او يتم طرحه) محددة ويتيح لنا العداد تنفيذ تعليمات معينة عندما تتغير قيمته كما يمكن لنا استخدامه اكثر من مرة وهذا يميز العداد عن جمل التكرار مثل جملة نفذ طالما Do While.

والشفيرة المصدرية للمثال الاول كالتالى

```
:note: "application settings"
set window auto show off
set label transparent on
The application title is "Counter Range"

:note: "application window"
I want window and the window title is "Counter Range"
window width = 370. and window height = 360.
the window back color = {255,255,255}.

:note: "window controls"
I want label and label caption is "From"
I want textbox and textbox left is 50. and the textbox name is text1.
The textbox value is 1.
I want label and label caption is "To" and the label top is 50.
I want textbox and textbox left is 50. and the textbox name is text2.
The textbox value is 10.
The textbox top is 50.
I want listbox and listbox top is 100.
I want button and button caption is "Count Forward"
The button top is 100. and the button left is 230.
The button name is btn1.

:note: "application task"
btn1 mouseclick. instructions are

    I want counter

    text1. textbox selected and textbox value to [num1].
    text2. textbox selected and textbox value to [num2].

    Get the counter start from [num1].
    Get the counter end from [num2].
    The counter step is 1.
    The counter name is mycount1.

    You use counter

    mycount1 change. instructions are


        Counter value to [myvalue].
        Listbox must add from [myvalue].

    End of instructions

End of instructions

You Show Window
```

والتطبيق وقت التشغيل كالتالى



ونلاحظ ان العد تصاعدى من 1 الى 10

ولكى يكون العد تنازلى لدينا مثال اخر يوضح ذلك

حيث فى كل مرة يتم انقاص رقم 1 من القيمة الابتدائية وذلك اثر الامر

The counter step is -1.

بينما نجد ان القيمة الابتدائية هى (10) والتي سوف نقوم بادخالها اثناء وقت التشغيل ان شاء الله رب العالمين فى مربع نص مربع داخل نافذة التطبيق ومن ثم يتم نقلها الى المتغير [num1] الذى يستخدم لتحديد نقطة البداية للعداد وذلك اثر الامر

Get the counter start from [num1].

اما القيمة التى يقف عندها العداد فانها (1) ونقوم بادخالها ان شاء الله فى مربع النص المعروف داخل نافذة التطبيق ومن ثم تنقل الى المتغير [num2] الذى يستخدم لتحديد نقطة النهاية للعداد وذلك اثر الامر

Get the counter end from [num2].

ومن مميزات استخدام العداد انه بإمكاننا ان نستعمله فى اى وقت عند الحاجة الى ذلك من خلال الامر

You use counter


```

:note: "application settings"
set window auto show off
set label transparent on
The application title is "Counter Range 2"
:note: "application window"
I want window and the window title is "Counter Range"
window width = 370. and window height = 360.
the window back color = {255,255,255}.
:note: "window controls"
i want label and label caption is "From"
i want textbox and textbox left is 50. and the textbox name is text1.
the textbox value is 10.
i want label and label caption is "To" and the label top is 50.
i want textbox and textbox left is 50. and the textbox name is text2.
the textbox value is 1.
the textbox top is 50.
i want listbox and listbox top is 100.
i want button and button caption is "Count backward"
the button top is 100. and the button left is 230.
the button name is btn1.
:note: "application task"
btn1 mouseclick. instructions are
    i want counter
    text1. textbox selected and textbox value to [num1].
    text2. textbox selected and textbox value to [num2].
    get the counter start from [num1].
    get the counter end from [num2].
    the counter step is -1.
    the counter name is mycount1.
    you use counter
    mycount1 change. instructions are
        counter value to [myvalue].
        listbox must add from [myvalue].
    end of instructions
end of instructions

```

You Show Window

والتطبيق وقت التشغيل كالتالى



والآن مع المثال DynamicObjects والذي يوضح كيفية انشاء الكائنات بشكل ديناميكي مع تحديد عدد هذه الكائنات من قبل المستخدم

والشفيرة المصدرية كالتالى

```
:note: "application settings"
Set window auto show off
Set label transparent on
The application title is "Dynamic Objects"

:note: "application window"
I want window and the window title is "Dynamic Objects"
Window width = 450. and window height = 500.
The window back color = {255,255,255}.

:note: "window controls"
I want label and label caption is "Enter Number"
I want textbox and textbox left is 100. and the textbox name is text1.

I want button and button caption is "Create"
The button left is 320. and the button name is btn1.

:note: "application task"
btn1 mouseclick. instructions are

    text1. textbox selected and textbox value to [num1].
    I want counter
    The counter start is 1.
    Get the counter end from [num1].
    The counter step is 1.
    The counter name is mycount1.
    [theleft]. = (30). and you use counter
    [theleft]. = (150). and you use counter
    [theleft]. = (270). and you use counter

    mycount1 change. instructions are

        counter value to [myvalue].
        [mytop]. = [myvalue] x (50).

        I want button
        The button caption from [myvalue].
        The button top from [mytop].
        The button left from [theleft].
        The button backcolor is {255,255,255}.
        The button name is dynamicbutton.

        dynamicbutton mouseclick. instructions are
            the button backcolor is {255,255,0}.
        End of instructions


    End of instructions

End of instructions
```

**You Show Window**

فى هذا المثال نلاحظ انه لانشاء الازرار تم استخدام عداد Counter يقوم بهذه المهمة وقد تم استخدام هذا العداد ثلاثة مرات من اجل انشاء الازرار مشكلة على ثلاثة اعمدة.

والتطبيق وقت التشغيل كالتالى



حيث تم ادخال الرقم (7) فى مربع النص المعرف داخل نافذة التطبيق

ثم تم الضغط على الزر Create

فتم انشاء (21) زر مقسمين الى ثلاثة اعمدة يشتمل كل عمود منها على (7) ازرار

وعند الضغط على اى زر من هذه الازرار يتغير لونه خلفيته مشيرا الى ان هذا الزر قد تم الضغط عليه من قبل المستخدم.

والآن نتابع لنرى مثالين على تعريف احداث معرفة على اساس قيمة لمتغير بمعنى ان التعليمات يتم تنفيذها عندما تصبح قيمة المتغير قيمة معينة يتم تحديدها

والشفيرة المصدرية للمثال الاول كالتالى

```
:note: "application settings"
set window auto show off
set label transparent on
The application title is "Case Sample"

:note: "application window"
I want window and the window title is "Counter Sensitive to number 3"
window width = 370. and window height = 360.
the window back color = {255,255,255}.

:note: "window controls"
i want label and label caption is "From"
i want textbox and textbox left is 50. and the textbox name is text1.
the textbox value is 1.
i want label and label caption is "To" and the label top is 50.
i want textbox and textbox left is 50. and the textbox name is text2.
the textbox value is 10.
the textbox top is 50.
i want listbox and listbox top is 100.
i want button and button caption is "Count Forward"
the button top is 100. and the button left is 230.
the button name is btn1.

:note: "application task"
btn1 mouseclick. instructions are
  i want counter
  text1. textbox selected and textbox value to [num1].
  text2. textbox selected and textbox value to [num2].
  get the counter start from [num1].
  get the counter end from [num2].
  the counter step is 1.
  the counter name is mycount1.
  you use counter
  mycount1 change. instructions are
    counter value to [myvalue].
    listbox must add from [myvalue].
    [mytest]. = [myvalue] isequal (3).
    mytest true. instructions are
      listbox must add item "this is number three"
    end of instructions
  end of instructions
end of instructions

You Show Window
```

وفى هذا المثال تم تعريف حدث Event يشتمل على تعليمات يتم استدعائها عندما تكون قيمة المتغير mytest تساوى true. وتم استخدام isequal لكى تعطى ناتج true اذا كانت قيمة المتغير [myvalue] تساوى (3)



والتطبيق وقت التشغيل كالتالى

والتطبيق يستخدم عداد Counter يستقبل نقطة بدايته ونقطة نهايته من المستخدم ويتم العد من نقطة البداية الى نقطة النهاية بخطوة تضيف رقم (1) الى الرقم الحالى لدى العداد

وقد تم ادخال الرقم (1) كنقطة بداية


بينما تم ادخال الرقم (10) كنقطة نهاية

وبالتالى يقوم العداد بالعد من الرقم 1 ثم 2 ثم 3 ..... إلى ان يصل الى الرقم 10.

وفى كل مرة يتم استدعاء تعليمات الحدث change لان قيمة العداد تغيرت.

وفى هذه التعليمات يتم اختبار قيمة العداد هل هى تساوى الرقم (3) ام لا ؟ ويتم تسجيل نتيجة هذا الاختبار فى متغير.

وقد تم تعريف حدث يشتمل تعليمات يتم تنفيذها عندما تكون قيمة هذا المتغير true اى عندما تكون نتيجة الاختبار صحيحة true .



## والآن نتابع لنرى المثال الثانى على احداث الحالة State Events التى تقوم على قيم المتغيرات والشفيرة المصدرية لهذا المثال كالتالى

```
:note: "application settings"
set window auto show off
set label transparent on
The application title is "Cases Sample"

:note: "application window"
I want window and the window title is "Cases"
window width = 370. and window height = 360.
the window back color = {0,0,100}.

:note: "window controls"
i want label and label caption is "From"
the label fore color is {255,255,255}.
i want textbox and textbox left is 50. and the textbox name is text1.
the textbox value is 1.
i want label and label caption is "To" and the label top is 50.
the label fore color is {255,255,255}.
i want textbox and textbox left is 50. and the textbox name is text2.
the textbox value is 5.
the textbox top is 50.
i want listbox and listbox top is 100.
i want button and button caption is "Count Forward"
the button top is 100. and the button left is 230.
the button name is btn1.

:note: "application task"
btn1 mouseclick. instructions are

    i want counter

    text1. textbox selected and textbox value to [num1].
    text2. textbox selected and textbox value to [num2].

    get the counter start from [num1].
    get the counter end from [num2].
    the counter step is 1.
    the counter name is mycount1.

    you use counter

    mycount1 change. instructions are


        counter value to [myvalue]. and [mytest]. = [myvalue].
        :note: Cases.
        mytest 1. instructions are [myvalue]. = (One ) [myvalue]. ok
        mytest 2. instructions are [myvalue]. = (Two ) [myvalue]. ok
        mytest 3. instructions are [myvalue]. = (Three ) [myvalue]. ok
        mytest 4. instructions are [myvalue]. = (Four ) [myvalue]. ok
        mytest 5. instructions are [myvalue]. = (Five ) [myvalue]. ok
        listbox must add from [myvalue].

    End of instructions

End of instructions

You Show Window
```

والتطبيق وقت التشغيل كالتالى



والتطبيق يستخدم عداد Counter يستقبل نقطة بدايته ونقطة نهايته من المستخدم ويتم العد من نقطة البداية الى نقطة النهاية بخطوة تضيف رقم (1) الى الرقم الحالى لدى العداد

وقد تم ادخال الرقم (1) كنقطة بداية

بينما تم ادخال الرقم (5) كنقطة نهاية

وبالتالى يقوم العداد بالعد من الرقم 1 ثم 2 ثم 3 ..... إلى ان يصل الى الرقم 5.

وفى كل مرة يتم استدعاء تعليمات الحدث change لان قيمة العداد تغيرت.

فى هذه التعليمات يتم اسناد قيمة العداد الى المتغير mytest الذى تم تعريف خمسة احداث تقوم على خمسة قيم مختلفة لهذا المتغير وهى القيم من 1 الى 5

وبالتالى يتم تشغيل هذه الاحداث الخمسة المبنية على حالة المتغير mytest وفى كل مرة تتغير قيمة العداد يتم تنفيذ تعليمات الحدث المبنى على هذه القيمة ونحصل على النتيجة التى شاهدناها اثناء تجربة التطبيق.

والان نتابع لنرى المثال CheckAnswer والذي يعطى سؤال للمستخدم وينتظر الاجابة ثم يقيم هذه الاجابة هل هي صحيحة ام لا باستخدام احداث الحالة State Events المبنية على حالة المتغير.

والشفيرة المصدرية كالتالى

```
The application title is "Check Answer"
Set window auto show off

I want window and the window title is "Test"
Window width = 730. and window height = 530.

I want label
The label width is 700. and label height is 120.
The label font size is 20.
The label back color = {255,255,255}.
The label caption is
"
    Who is the creator of
    Supernova Programming Language ?
"

I want listbox and listbox top is 150.
Listbox width is 700. and listbox height is 250.
The listbox must add item "Gudio"
The listbox must add item "Mahmoud"
The listbox must add item "Matz"
The listbox font size is 15.

I want button and the button top is 450.
The button left is 240. and button caption = "Check Answer"
The button name is btn1.
The button back color is {255,255,255}.

I want button and the button top is 450.
The button left is 360. and button caption = "End Application"
The button name is btnclose.
The button fore color is {255,255,255}.
The button back color is {255,0,0}.
btnclose mouseclick. instructions are you close application ok

You show window

btn1 mouseclick. instructions are
    The listbox value to [x].
    [mytest]. = [x] isequal (2).
    mytest true. instructions are
        The label caption is
        "
            True Answer
        " ok
    mytest false. instructions are
        The label caption is
        "
            False Answer
        " ok


End of instructions
```

والتطبيق وقت التشغيل كالتالى

حيث يتم طرح سؤال عن من ماهو مبتكر لغة البرمجة سوبرنوبا ؟  
Who is the creator of Supernova Programming Language ?

ويتم اعطاء المستخدم ثلاثة اختيارات وهى  
Gudio (مبتكر لغة البايثون Python)  
Mahmoud (مبتكر لغة السوبرنوبا Supernova)  
Matz (مبتكر لغة الروبى Ruby)

وعند الضغط على زر Check Answer يتم فحص الاجابة التى قام باختيارها المستخدم ليتم تحديد هل هى اجابة صحيحة True Answer ام اجابة خاطئة False Answer وتظهر هذه النتيجة فى العنوان المعرف داخل نافذة التطبيق والذى يعرض السؤال.



ولانهاء التطبيق يتم الضغط على الزر End Application ونلاحظ ان لون خلفية هذا الزر احمر وهذا من الاساليب الارشادية فى تصميم واجهة المستخدم حيث يكون مظهر الزر له احياء بوظيفته.

والان لدينا ثلاثة امثلة لاداء وظيفة الحلقات Loops والشفيرة المصدرية للمثال الاول كالتالى


```
The application title is "Infinite Loop"  
I want window contains listbox and the window title is "Infinite Loop"  
the window name is mywin.
```

```
I want button and button caption is "Close"  
the button name is "myclose" and the button left is 300.  
myclose mouseclick. instructions are  
    you close window  
End of instructions
```

```
[y]. = (0).  
[x]. = (0).
```

```
[test]. = [y] isequal (1).  
test false. instructions are  
    [x]. = [x]+(1).  
    listbox must add from [x].  
    do windows events  
    [test]. = [y] isequal (1).  
End of instructions
```

والتطبيق وقت التشغيل كالتالى  
حيث نلاحظ ان لدينا حلقة لانهاية Infinite Loop اى لا تتوقف طالما ان التطبيق  
مازال يعمل وتنتهى فقط مع انهاء التطبيق بالضغط على الزر Close



## والشفيرة المصدرية للمثال الثاني كالتالي

```
The application title is "Loop Under Control"
I want window contains listbox and the window title is "Loop Under Control"
The window name is mywin.


I want button and button caption is "Close"
The button name is "myclose" and the button left is 300.
myclose mouseclick. instructions are
    you close window
End of instructions

I want button and button caption is "Stop"
The button name is "mystop" and the button left is 300.
The button top is 80.
mystop mouseclick. instructions are
    [y]. = (10000).
End of instructions

[y]. = (0).

[test]. = [y] isequal (10000).
test false. instructions are
    [y]. = [y]+(1).
    listbox must add from [y].
    do windows events
    [test]. = [y] isequal (10000).
End of instructions
```

والتطبيق وقت التشغيل كالتالي حيث يمكن ايقاف الحلقة بالضغط على الزر Stop



## والشفيرة المصدرية للمثال الثالث كالتالى

```
The application title is "Application will end after some processing"
```

```
[y]. = (0).
```

```
myloop. instructions must done  
myloop. instructions are  
do windows events  
[y]. = [y]+(1).  
myloop. instructions must done  
end of instructions
```

```
y 100000. instructions are You close application ok
```

## والتطبيق وقت التشغيل كالتالى



حيث يعمل التطبيق بدون اظهار نافذة – فقط يظهر فى شريط المهام لاداء مهمة معينة وعند الانتهاء منها يتم انهاء التطبيق.

وبهذا نكون قد وصلنا الى نهاية جزء (البرمجة بوصف الخيال) واتمنى ان اكون قد نجحت فى عرضه باسلوب بسيط وشيق لما له من اهمية كبير فى عالم البرمجة بلغة السوبرنوبا لان البرمجة بوصف الخيال هى عصب الحياة فى اللغة.





## واجهة المستخدم الرسومية GUI

تعمل تطبيقات السوبرنوبا بواجهة مستخدم رسومية تتكون من النوافذ Windows وعناصر التحكم Controls التي يتفاعل مع المستخدم ومن ثم تتحقق أحداث Events تكون بدورها سبب في تنفيذ تعليمات ما حتى يتم استكمال العمل في الاتجاه المطلوب.

وفي الواقع لقد استخدمنا واجهة المستخدم الرسومية في جميع التطبيقات التي تم التعرض لها في الكتاب ونحن هنا في هذا الجزء من الكتاب نتعمق أكثر في استعمال واجهة المستخدم الرسومية لاكتساب المزيد من المهارة في التعامل مع واجهة المستخدم الرسومية بلغة السوبرنوبا التي تتسم بسهولة تعليماتها المستخدمة لإنشاء الكائنات والاستجابة للأحداث المختلفة وذلك بتحديد تعليمات يتم تنفيذها عند تحقق الحدث مستفيدين بأسلوب البرمجة بوصف الخيال.

في هذا الجزء من الكتاب سوف نتعرف على معظم التعليمات المستخدمة للتعامل مع النوافذ وعناصر التحكم المختلفة مع الاطلاع على العديد من الامثلة البسيطة التي تسهل الاستيعاب ان شاء الله رب العالمين.



## التعليمات التالية هي المتوفرة للتعامل مع النوافذ Windows

### Windows

-----

```
* Application title is ....data as value....
* You close application
* Do windows events
* Set window auto show on
* Set window auto show off
* Set window resize on
* Set window resize off
* Set window modal on
* Set window modal off
* I want window
* Window title is ....data as value....
* Window direction is right to left
* Window back color is ....data as value....
* Window top is ....data as value....
* Window left is ....data as value....
* Window width is ....data as value....
* Window height is ....data as value....
* You maximize window
* You minimize window
* You centre window
* You show window
* You hide window
* You close window
* Window name is ....data as value....
* ....data as value.... window selected
* Window parent is ....data as value....
* Window title from ....data as expression....
* Window title to ....data as expression....
* Window Handle to ....data as expression....
```

### Window Events

-----

```
move
gotfocus
lostfocus
mouseclick
mousemove
resize
maximize
minimize
release
```

**الامر Application title is**  
يستخدم لتحديد عنوان التطبيق

**الامر You close application**  
يستخدم لاقلاق التطبيق بشكل كامل

### الامر Do windows events

يستخدم لتنفيذ احداث الويندوز المختلفة حتى لا يتعطل التطبيق عن العمل عند الدخول فى حلقة Loop تستغرق وقتا طويلا نسبيا

### الامر Set window auto show off

يستخدم لمنع الاظهار التلقائى للنوافذ حتى تظهر النافذة فى الوقت المناسب عند اكتمال تحديد خصائصها وازضافة عناصر التحكم المختلفة اليها وذلك حتى لا يكون هناك رعشة Flicker تظهر حين تغير خواص النافذة وهى تظهر امام المستخدم

### الامر Set window auto show on

يستخدم لتفعيل الاظهار التلقائى للنافذة

### الامر Set window resize on

يتيح للمستخدم امكانية اعادة تحجيم النافذة

### الامر Set window resize off

يمنع امكانية اعادة تحجيم النافذة حتى تظل النافذة طوال الوقت بحجم معين

### الامر Set window modal on

يمنع التنقل الى نوافذ التطبيق الاخرى حتى يتم اغلاق النافذة

### الامر Set window modal off

يتيح لنا امكانية التنقل الى نوافذ التطبيق الاخرى اثناء عمل النافذة

### الامر I want window

يستخدم لانشاء نافذة جديدة ويمكن استبدال I want بكلمة Contains او بكلمة Create فكل منهما مقبولان محل I want من قبل مترجم اللغة

### الامر Window title is

يستخدم لتحديد عنوان النافذة عن طريق بيانات تدخل كقيمة Data as value

### الامر Window direction is right to left

يستخدم حتى يكون اتجاه النافذة بداية من اليمين لليساار

### الامر Window back color is

يستخدم لتحديد لون الخلفية فى النافذة

الامر Window top is  
يستخدم لتحديد نقطة البداية العلوية للنافذة

الامر Window left is  
يستخدم لتحديد نقطة البداية اليسرى للنافذة

الامر Window width is  
يستخدم لتحديد عرض النافذة

الامر Window height is  
يستخدم لتحديد ارتفاع النافذة

الامر You maximize window  
يستخدم لتكبير النافذة ويمكن استبدال كلمة You بـ Please

الامر You minimize window  
يستخدم لتصغير النافذة

الامر You centre window  
يستخدم لتوسيط النافذة

الامر You show window  
يستخدم لعرض النافذة وهو ضروري في حالة منع الاظهار التلقائي للنافذة

الامر You hide window  
يستخدم لاختفاء النافذة

الامر You close window  
يستخدم لاقلاق النافذة وفي حالة كونها اخر نافذة موجودة فانه يتم اغلاق التطبيق بشكل كامل.

الامر Window name is  
يستخدم لتحديد اسم النافذة وهذا ضروري لتعريف تعليمات تستجيب لاحداث النافذة المختلفة

الامر Window parent is  
يستخدم لتحديد النافذة الام التي يجب ان تحتوى النافذة

## الامر Window title from

يستخدم لتحديد عنوان النافذة من بيانات كتعبير اي Data as expression وهذا يستخدم عندما نرغب فى تحديد عنوان النافذة من خلال متغير.

## الامر Window handle to

يستخدم لمعرفة مقبض النافذة (رقم خاص بها) المسجل لدى نظام التشغيل ولهذا استخدامات متقدمة عند استخدام دوال خارجية فى مكتبات ربط ديناميكية حيث تستخدم هذه الدوال هذا المقبض للوصول الى النافذة والتعامل معها للتعديل والتحكم فى خصائصها.

## الحدث move

يتحقق عند تحرك النافذة

## الحدث gotfocus

يتحقق عندما تمتلك النافذة البؤرة

## الحدث lostfocus

يتحقق عندما تفقد النافذة البؤرة

## الحدث mouse click

يتحقق عند الضغط بالفارة على سطح النافذة

## الحدث mouse move

يتحقق عندما تتحرك الفارة على سطح النافذة

## الحدث resize

يتحقق عندما يتم اعادة تحجيم النافذة

## الحدث maximize

يتحقق عندما يتم تكبير النافذة

## الحدث minimize

يتحقق عندما يتم تصغير النافذة

## الحدث release

يتحقق عندما يتم اغلاق النافذة

## التعليمات التالية هي المتوفرة للتعامل مع العنوان Label

```
Label
-----

* I want label
* Label caption is ....data as value....
* Label top is ....data as value....
* Label left is ....data as value....
* Label width is ....data as value....
* Label height is ....data as value....
* Label name is ....data as value....
* Label font name is ....data as value....
* Label font size is ....data as value....
* Label fore color is ....data as value....
* Label back color is ....data as value....
* Label font bold is ....data as value....
* Label font italic is ....data as value....
* Label font underline is ....data as value....
* Label font strikethrough is ....data as value....
* Set label transparent on
* Set label transparent off
* ....data as value.... label selected
* Label caption from ....data as expression....
* Label caption to ....data as expression....
* Label top from ....data as expression....
* Label top to ....data as expression....

Label Events
-----

        mouseclick
```

**I want label** الامر  
يستخدم لانشاء عنوان

**Label caption is** الامر  
يستخدم لتحديد نص عنوان

**Label top is** الامر  
يستخدم لتحديد نقطة البداية العلوية للعنوان على سطح النافذة

**Label left is** الامر  
يستخدم لتحديد نقطة البداية اليسرى للعنوان على سطح النافذة

**Label width is** الامر  
يستخدم لتحديد عرض العنوان على سطح النافذة

Label height is الامر  
يستخدم لتحديد ارتفاع العنوان على سطح النافذة

Label name is الامر  
يستخدم لتحديد اسم للعنوان وهو ضرورى لتعريف تعليمات تنفذ عند تحقق حدث مرتبط بالعنوان – وايضا لتحديد العنوان للتعديل فى خواصه

Label font name is الامر  
يستخدم لتحديد اسم الخط المستخدم فى عرض نص العنوان

Label font size is الامر  
يستخدم لتحديد حجم الخط المستخدم فى عرض نص العنوان

Label fore color is الامر  
يستخدم لتحديد لون الكتابة المستخدم فى عرض العنوان

Label back color is الامر  
يستخدم فى تحديد لون الخلفية المستخدم فى عرض العنوان

Label font bold is الامر  
يستخدم لتحديد هل الخط المستخدم فى عرض العنوان سميك ام لا ويتم ذلك من خلال قيمة نقوم بتحديدها اما True او False

Label font italic is الامر  
يستخدم لتحديد هل الخط المستخدم فى عرض العنوان مائل ام لا

Label font under line is الامر  
يستخدم لتحديد هل الكتابة سيكون تحتها خط ام لا

Label font strick out is الامر  
يستخدم لتحديد هل النص سيكون بارز ام لا

Set label transparent on الامر  
يستخدم فى حالة الرغبة فى ان يكون العنوان شفاف (لايوجد لون للخلفية)

Set label transparent off الامر  
يستخدم لالغاء شفافية العنوان (اي يتم تفعيل لون الخلفية)

## الامر Label selected

يستخدم لتحديد عنوان للتعامل معه والتعديل في خصائصه

## الامر Label caption from

يستخدم لتحديد النص من بيانات كتعبير Data as expression

## الامر Label caption to

يستخدم في حفظ نص العنوان في متغير

## الامر Label top from

يستخدم لتحديد البداية العلوية للعنوان من بيانات كتعبير

## الامر Label top to

يستخدم لحفظ البداية العلوية للعنوان في متغير

## الحدث mouse click

يتحقق عند النقر بالفأرة على العنوان

والتعليمات التالية هي المتوفرة للتعامل مع مربع النص Textbox

### Textbox

-----

- \* I want textbox
- \* Textbox value is ....data as value....
- \* Textbox top is ....data as value....
- \* Textbox left is ....data as value....
- \* Textbox width is ....data as value....
- \* Textbox height is ....data as value....
- \* Textbox name is ....data as value....
- \* Textbox font name is ....data as value....
- \* Textbox font size is ....data as value....
- \* Textbox fore color is ....data as value....
- \* Textbox back color is ....data as value....
- \* Textbox font bold is ....data as value....
- \* Textbox font italic is ....data as value....
- \* Textbox font underline is ....data as value....
- \* Textbox font strikethrough is ....data as value....
- \* ....data as value.... Textbox selected
- \* Textbox value from ....data as expression....
- \* Textbox value to ....data as expression....

### Textbox Events

-----

gotfocus  
lostfocus  
change  
enter



الامر I want textbox  
يستخدم لإنشاء مربع نص

الامر Textbox value is  
يستخدم لتحديد القيمة الموجودة في مربع النص

الامر Textbox top is  
يستخدم لتحديد نقطة البداية العلوية لمربع النص على سطح النافذة

الامر Textbox left is  
يستخدم لتحديد نقطة البداية اليسرى لمربع النص على سطح النافذة

الامر Textbox width is  
يستخدم لتحديد عرض مربع النص

الامر Textbox height is  
يستخدم لتحديد ارتفاع مربع النص

الامر Textbox name is  
يستخدم لتحديد اسم مربع النص

الامر Textbox font name is  
يستخدم لتحديد اسم الخط

الامر Textbox font size is  
يستخدم لتحديد حجم الخط

الامر Textbox fore color is  
يستخدم لتحديد لون الكتابة

الامر Textbox back color is  
يستخدم لتحديد لون الخلفية

الامر Textbox font bold is  
يستخدم لتحديد هل الخط سميك ام لا

الامر **Textbox font italic is**  
يستخدم لتحديد هل الخط مائل ام لا

الامر **Textbox font unde line is**  
يستخدم لتحديد هل هناك خط تحت النص ما لا

الامر **Textbox font strick out is**  
يستخدم لتحديد هل الخط بارز ام لا

الامر **Textbox selected**  
يستخدم لتحديد مربع نص للتعامل معه والتعديل فى خصائصه

الامر **Textbox value from**  
يستخدم لتحديد القيمة عن طريق بيانات كتعبير **Data as expression**

الامر **Textbox value to**  
يستخدم لحفظ القيمة فى متغير

الحدث **Got Focus**  
يتحقق عندما يمتلك مربع النص البؤرة

الحدث **Lost focus**  
يتحقق عندما يفقد مربع النص البؤرة

الحدث **Change**  
يتحقق عندما يتغير محتوى مربع النص (القيمة المكتوبة فيه) وهذا الحدث شائع الاستخدام لتطبيق البحث التفاعلى **Interactive Search** اى البحث التلقائى الذى يتم اثناء ادخال المستخدم للنص وقبل الانتهاء من الادخال لان اظهار نتيجة البحث مبكرا قد يكون كافيا للوصول لما نبحث عنه وبالتالى يحدث توفير للوقت وهذه سمة من سمات واجهة المستخدم المتطورة تحت مسمى واجهة المستخدم التفاعلية وهو اسلوب مطبق على نطاق واسع حاليا.

الحدث **Enter**  
يتحقق عندما يتم الضغط على مفتاح الادخال

## والتعليمات التالية هي المتوفرة للتعامل مع الزر Button

Button

-----

- \* I want button
- \* Button caption is ....data as value....
- \* Button top is ....data as value....
- \* Button left is ....data as value....
- \* Button width is ....data as value....
- \* Button height is ....data as value....
- \* Button name is ....data as value....
- \* Button font name is ....data as value....
- \* Button font size is ....data as value....
- \* Button fore color is ....data as value....
- \* Button back color is ....data as value....
- \* Button font bold is ....data as value....
- \* Button font italic is ....data as value....
- \* Button font underline is ....data as value....
- \* Button font strickout is ....data as value....
- \* Button picture is ....data as value....
- \* ....data as value.... Button selected
- \* Button caption from ....data as expression....
- \* Button caption to ....data as expression....
- \* Button top from ....data as expression....
- \* Button top to ....data as expression....
- \* Button left from ....data as expression....
- \* Button left to ....data as expression....

Button Events

-----

mouseclick

### الامر I want button

يستخدم لإنشاء زر جديد داخل النافذة المعرفة

### الامر Button caption is

يستخدم لتحديد نص الزر

### الامر Button top is

يستخدم لتحديد نقطة البداية العلوية للزر على سطح النافذة

### الامر Button left is

يستخدم لتحديد نقطة البداية اليسرى للزر على سطح النافذة

### الامر Button width is

يستخدم لتحديد عرض الزر

الامر Button height is  
يستخدم لتحديد ارتفاع النافذة

الامر Button name is  
يستخدم لتحديد اسم الزر لتحديده والتعامل معه فيما بعد

الامر Button font name is  
يستخدم لتحديد اسم الخط

الامر Button font size is  
يستخدم لتحديد حجم الخط

الامر Button fore color is  
يستخدم لتحديد لون الكتابة

الامر Button back color is  
يستخدم لتحديد لون الخلفية

الامر Button font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Button font italic is  
يستخدم لتحديد هل الخط مائل ام لا

الامر Button font under line is  
يستخدم لتحديد هل يوجد خط تحت نص الزر ام لا

الامر Button font strick out is  
يستخدم لتحديد هل الخط بارز ام لا

الامر Button picture is  
يستخدم لتحديد صورة تظهر داخل الزر وهذا امر حيوى حيث يعطى مظهر جمالى فى التطبيقات.

الامر Button selected  
يستخدم لتحديد زر من خلال اسمه حتى نتعامل معه و نعدل فى خصائصه.

## الامر Button caption from

يستخدم لتحديد نص زر من بيانات كتعبير Data as expression

## الامر Button caption to

يستخدم لحفظ نص زر في متغير

## الامر Button top from

يستخدم لتحديد نقطة البداية العلوية للزر داخل سطح النافذة من بيانات كتعبير

## الامر Button top to

يستخدم لحفظ نقطة البداية العلوية للزر داخل سطح النافذة في متغير

## الامر Button left from

يستخدم لتحديد نقطة البداية اليسرى للزر داخل سطح النافذة من بيانات كتعبير

## الامر Button left to

يستخدم لحفظ نقطة البداية اليسرى للزر داخل سطح النافذة في متغير

## الحدث Mouse click

يتحقق عندما يتم الضغط بالفأرة على الزر اثناء وقت تشغيل البرنامج

## التعليمات التالية للتعامل مع مربع التحرير Editbox

Editbox

-----

```
* I want editbox
* Editbox value is ....data as value....
* Editbox top is ....data as value....
* Editbox left is ....data as value....
* Editbox width is ....data as value....
* Editbox height is ....data as value....
* Editbox name is ....data as value....
* Editbox font name is ....data as value....
* Editbox font size is ....data as value....
* Editbox fore color is ....data as value....
* Editbox back color is ....data as value....
* Editbox font bold is ....data as value....
* Editbox font italic is ....data as value....
* Editbox font underline is ....data as value....
* Editbox font strickout is ....data as value....
* ....data as value.... Editbox selected
* Editbox value from ....data as expression....
```

Editbox Events

-----

gotfoucs

lostfoucs

change

الامر I want editbox  
يستخدم لإنشاء مربع تحرير

الامر Editbox value is  
يستخدم لتحديد القيمة الظاهرة في مربع التحرير

الامر Editbox top is  
يستخدم لتحديد نقطة البداية العلوية لمربع التحرير على سطح النافذة

الامر Editbox left is  
يستخدم لتحديد نقطة البداية اليسرى لمربع التحرير على سطح النافذة

الامر Editbox width is  
يستخدم لتحديد عرض مربع التحرير

الامر Editbox height is  
يستخدم لتحديد ارتفاع مربع التحرير

الامر Editbox name is  
يستخدم لتحديد اسم مربع التحرير لتحديده والتعامل معه فيما بعد

الامر Editbox font name is  
يستخدم لتحديد اسم الخط

الامر Editbox font size is  
يستخدم لتحديد حجم الخط

الامر Editbox fore color is  
يستخدم لتحديد لون الخط

الامر Editbox back color is  
يستخدم لتحديد لون الخلفية

الامر Editbox font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

### الامر Editbox font italic is

يستخدم لتحديد هل الخط مائل ام لا  
ويتم ذلك من خلال قيمة اما True او False

### الامر Editbox font under line is

يستخدم لتحديد هل يكون هناك خط تحت النص ام لا  
ويتم ذلك من خلال قيمة اما True او False

### الامر Editbox font strick out is

يستخدم لتحديد هل الخط بارز ام لا  
ويتم ذلك من خلال قيمة اما True او False

### الامر Editbox selected

يستخدم لتحديد مربع تحرير عن طريق اسمه وبعد تحديده يمكن التعامل معه  
مباشرة والتعديل في خصائصه

### الامر Editbox value from

يستخدم لتحديد القيمة الظاهرة في مربع التحرير عن طريق بيانات كتعبير  
Data as expression وغالبا تكون متغير

### الحدث Got focus

يتحقق عندما يمتلك مربع التحرير البؤرة

### الحدث Lost focus

يتحقق عندما يفقد مربع التحرير البؤرة

### الحدث change

يتحقق عندما تتغير القيمة الظاهرة في مربع التحرير ويمكن الاستفادة من هذه  
الامكانية في تطوير تطبيقات تفاعلية والتي تقوم بالتفاعل مع المستخدم عن  
طريق اجراء عمليات البحث بشكل تلقائي بمجرد ان يقوم المستخدم بعمل  
تعديلات على النص الظاهر في مربع التحرير وبالتالي تكون هناك سرعة  
في الوصول الى النتائج التي بدورها توجه المستخدم للخطوات التالية التي  
يرغب فيها.

## والتعليقات المتوفرة للتعامل مع مربع القائمة كالتالى

### Listbox

-----

- \* I want listbox
- \* Listbox value is ....data as value....
- \* Listbox top is ....data as value....
- \* Listbox left is ....data as value....
- \* Listbox width is ....data as value....
- \* Listbox height is ....data as value....
- \* Listbox name is ....data as value....
- \* Listbox font name is ....data as value....
- \* Listbox font size is ....data as value....
- \* Listbox fore color is ....data as value....
- \* Listbox back color is ....data as value....
- \* Listbox font bold is ....data as value....
- \* Listbox font italic is ....data as value....
- \* Listbox font underline is ....data as value....
- \* Listbox font strickout is ....data as value....
- \* ....data as value.... Listbox selected
- \* Listbox must add item ....data as value....
- \* Listbox must add from ....data as expression....
- \* Listbox value from ....data as expression....
- \* Listbox value to ....data as expression....

### Listbox Events

-----

gotfoucs  
lostfocus  
change

### الامر I want listbox

يستخدم لإنشاء مربع قائمة جديد داخل النافذة

### الامر Listbox value is

يستخدم لتحديد العنصر النشط فى القائمة عن طريق رقمه

### الامر Listbox top is

يستخدم لتحديد نقطة البداية العلوية لمربع القائمة على سطح النافذة

### الامر Listbox left is

يستخدم لتحديد نقطة البداية اليسرى لمربع القائمة على سطح النافذة

### الامر Listbox width is

يستخدم لتحديد عرض مربع القائمة



الامر Listbox height is  
يستخدم لتحديد ارتفاع مربع القائمة على سطح النافذة

الامر Listbox name is  
يستخدم لتحديد اسم مربع القائمة لاستخدامه في تحديده فيما بعد والتعامل معه

الامر Listbox font name is  
يستخدم في تحديد اسم الخط

الامر Listbox font size is  
يستخدم في تحديد حجم الخط

الامر Listbox fore color is  
يستخدم لتحديد لون الكتابة

الامر Listbox back color is  
يستخدم لتحديد لون الخلفية

الامر Listbox font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Listbox font italic is  
يستخدم لتحديد هل الخط مائل ام لا

الامر Listbox font under line is  
يستخدم لتحديد هل يتم وضع خط تحت النص ام لا

الامر Listbox font strick out  
يستخدم لتحديد هل الخط بارز ام لا

الامر Listbox selected  
يستخدم لتحديد مربع القائمة عن طريق الاسم للتعامل معه

الامر Listbox must add item  
يستخدم لاضافة عنصر الى مربع القائمة من خلال بيانات كقيمة  
Data as value

### الامر Listbox must add from

يستخدم لاضافة عنصر الى مربع القائمة من خلال بيانات كتعبير Data as expression وغالبا مايكون متغير

### الامر Listbox value from

يستخدم لتحديد قيمة مربع القائمة (رقم العنصر النشط فيها) من خلال بيانات كتعبير Data as expression وغالبا ماتكون متغير

### الامر Listbox value to

يستخدم لحفظ قيمة مربع القائمة (رقم العنصر النشط فيها) فى متغير

### الحدث Got Focus

يتحقق عندما يمتلك مربع القائمة البؤرة اثناء وقت التشغيل

### الحدث Lost Focus

يتحقق عندما يفقد مربع القائمة البؤرة اثناء وقت التشغيل

### الحدث Change

يتحقق عندما يتم تعديل قيمة مربع القائمة (العنصر النشط)

والتعليمات التالية للتعامل مع الكومبو / صندوق القائمة المنسدلة Combobox

#### Combobox

-----

- \* I want combobox
- \* Combobox value is ....data as value....
- \* Combobox top is ....data as value....
- \* Combobox left is ....data as value....
- \* Combobox width is ....data as value....
- \* Combobox height is ....data as value....
- \* Combobox name is ....data as value....
- \* Combobox fontname is ....data as value....
- \* Combobox fontsize is ....data as value....
- \* Combobox forecolor is ....data as value....
- \* Combobox backcolor is ....data as value....
- \* Combobox fontbold is ....data as value....
- \* Combobox fontitalic is ....data as value....
- \* Combobox fontunderline is ....data as value....
- \* Combobox fontstrickout is ....data as value....
- \* ....data as value.... Combobox selected
- \* Combobox must add item ....data as value....

#### Combobox Events

-----

gotfocus                      lostfocus                      change

الامر I want combobox  
يقوم بإنشاء صندوق قائمة منسدلة جديد

الامر Combobox value is  
يستخدم لتحديد القيمة (رقم العنصر النشط)

الامر Combobox top is  
يستخدم لتحديد نقطة البداية العلوية للكائن على سطح النافذة

الامر Combobox left is  
يستخدم لتحديد نقطة البداية اليسرى للكائن على سطح النافذة

الامر Combobox width is  
يستخدم لتحديد العرض

الامر Combobox height is  
يستخدم لتحديد الارتفاع

الامر Combobox name is  
يستخدم لتحديد الاسم لاستخدامه في التعامل مع الكائن فيما بعد

الامر Combobox font name is  
يستخدم لتحديد اسم الخط

الامر Combobox font size is  
يستخدم لتحديد حجم الخط

الامر Combobox fore color is  
يستخدم لتحديد لون الكتابة

الامر Combobox back color is  
يستخدم لتحديد لون الخلفية

الامر Combobox font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Combobox font italic is  
يستخدم لتحديد هل الخط مائل ام لا

الامر Combobox font under line is  
يستخدم لتحديد هل هناك خط تحت النص ام لا

الامر Combobox font strick out is  
يستخدم لتحديد هل الخط بارز ام لا

الامر Combobox selected  
يستخدم لتحديد الكائن عن طريق اسمه

الامر Combobox must add item  
يستخدم لاضافة عنصر من خلال بيانات كقيمة

الحدث Got focus  
يتحقق عندما يمتلك الكائن البؤرة

الحدث Lost focus  
يتحقق عندما يفقد الكائن البؤرة

الحدث change  
يتحقق عندما تتغير القيمة (العنصر النشط)

والتعليقات التالية هي المتوفرة للتعامل مع مربع الفحص Checkbox

Checkbox

-----

```
* I want checkbox
* Checkbox value is ....data as value....
* Checkbox top is ....data as value....
* Checkbox left is ....data as value....
* Checkbox width is ....data as value....
* Checkbox height is ....data as value....
* Checkbox name is ....data as value....
* Checkbox font name is ....data as value....
* Checkbox font size is ....data as value....
* Checkbox fore color is ....data as value....
* Checkbox back color is ....data as value....
* Checkbox font bold is ....data as value....
* Checkbox font italic is ....data as value....
* Checkbox font underline is ....data as value....
* Checkbox font strickout is ....data as value....
* ....data as value.... Checkbox selected
* Checkbox caption is ....data as value....
```

تابع التعليمات المتوفرة للتعامل مع مربع الفحص

Checkbox Events

gotfocus  
lostfocus  
change

**الامر I want checkbox**

يستخدم لإنشاء مربع فحص جديد

**الامر Checkbox value is**

يستخدم لتحديد قيمة مربع الفحص هل هي True ام False

**الامر Checkbox top is**

يستخدم لتحديد نقطة البداية العلوية لمربع الفحص على سطح النافذة

**الامر Checkbox left is**

يستخدم لتحديد نقطة البداية اليسرى لمربع الفحص على سطح النافذة

**الامر Checkbox width is**

يستخدم لتحديد عرض مربع الفحص على سطح النافذة

**الامر Checkbox height is**

يستخدم لتحديد ارتفاع مربع الفحص على سطح النافذة

**الامر Checkbox name is**

يستخدم لتحديد اسم مربع الفحص ليستخدم لتحديده والتعامل معه فيما بعد

**الامر Checkbox font name is**

يستخدم لتحديد اسم الخط

**الامر Checkbox font size is**

يستخدم لتحديد حجم الخط

**الامر Checkbox fore color is**

يستخدم لتحديد لون الكتابة

الامر **Checkbox back color is**  
يستخدم لتحديد لون الخلفية

الامر **Checkbox font bold is**  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر **Checkbox font italic is**  
يستخدم لتحديد هل الخط المائل ام لا

الامر **Checkbox font under line is**  
يستخدم لتحديد هل هناك خط تحت النص ام لا

الامر **Checkbox strick out is**  
يستخدم لتحديد هل الخط بارز ام لا

الامر **Checkbox selected**  
يستخدم لتحديد مربع فحص من خلال اسمه للتعامل معه  
والتعديل في خصائصه

الامر **Checkbox caption is**  
يستخدم لتحديد نص مربع الفحص

الحدث **Got focus**  
يتحقق عندما يمتلك مربع الفحص البؤرة

الحدث **Lost focus**  
يتحقق عندما يفقد مربع الفحص البؤرة

الحدث **change**  
يتحقق عندما تتغير قيمة مربع الفحص ( True او False ) وهذه الامكانية هامة للتطبيقات التفاعلية حيث يمكن ان يتخذ التطبيق اجراء معيناً تبعاً لتغير حالة مربع الفحص فقد يكون هذا الاجراء مثلاً اظهار جزء من محتويات النافذة او اخفائه او التعديل في الخيارات المتاحة.

## التعليمات التالية هي المتوفرة للتعامل مع شريط التقدم Progress bar

Progressbar

```
-----  
  
* I want progressbar  
* Progressbar value is ....data as value....  
* Progressbar top is ....data as value....  
* Progressbar left is ....data as value....  
* Progressbar width is ....data as value....  
* Progressbar height is ....data as value....  
* Progressbar name is ....data as value....  
* ....data as value.... Progressbar selected  
* Progressbar range minimum is ....data as value....  
* Progressbar range maximum is ....data as value....  
* Progressbar fore color is ....data as value....  
* Progressbar back color is ....data as value....
```

**I want progress bar** الامر  
يستخدم لإنشاء شريط تقدم جديد

**Progressbar value is** الامر  
يستخدم لتحديد مقدار التقدم (قيمة رقمية)

**Progressbar top is** الامر  
يستخدم لتحديد نقطة البداية العلوية لشريط التقدم على سطح النافذة.

**Progressbar left is** الامر  
يستخدم لتحديد نقطة البداية اليسرى لشريط التقدم على سطح النافذة

**.Progressbar width is** الامر  
يستخدم لتحديد عرض شريط التقدم على سطح النافذة

**Progressbar height is** الامر  
يستخدم لتحديد ارتفاع شريط التقدم على سطح النافذة

**Progressbar name is** الامر  
يستخدم لتحديد اسم شريط التقدم وتحديدته والتعامل معه فيما بعد

**Progressbar selected** الامر  
يستخدم لتحديد شريط تقدم من خلال اسمه

**Progressbar range minimum is** الامر  
يستخدم لتحديد الحد الادنى للنطاق المستخدم من قبل شريط التقدم

**Progressbar range maximim is** الامر  
يستخدم لتحديد الحد الاقصى للنطاق المستخدم من قبل شريط التقدم

**Progressbar fore color is** الامر  
يستخدم لتحديد لون الكتابة

**Progressbar back color is** الامر  
يستخدم لتحديد لون الخلفية

التعليمات التالية هي المتوفرة للتعامل مع السوستة Slider

```
Slider
```

```
-----
```

```
* I want slider  
* Slider value is ....data as value....  
* Slider top is ....data as value....  
* Slider left is ....data as value....  
* Slider width is ....data as value....  
* Slider height is ....data as value....  
* Slider name is ....data as value....  
* ....data as value.... slider selected  
* Slider range minimum is ....data as value....  
* Slider range maximum is ....data as value....
```

```
Slider Events
```

```
-----
```

```
change  
scroll
```

**I want slider** الامر  
يستخدم فى انشاء كائن جديد

**Slider value is** الامر  
يستخدم لتحديد القيمة (قيمة رقمية)



**Slider top is** الامر  
يستخدم لتحديد نقطة البداية العلوية للكائن على سطح النافذة

**Slider left is** الامر  
يستخدم لتحديد نقطة البداية اليسر للكائن على سطح النافذة

**Slider width is** الامر  
يستخدم لتحديد العرض

**Slider height is** الامر  
يستخدم لتحديد الارتفاع

**Slider name is** الامر  
يستخدم لتحديد اسم الكائن لتحديده والتعامل معه فيما بعد

**Slider selected** الامر  
يستخدم لتحديد الكائن من خلال اسمه للتعامل معه

**Slider range minimum is** الامر  
يستخدم لتحديد الحد الأدنى للنطاق المستخدم من قبل الكائن

**Slider range maximum is** الامر  
يستخدم لتحديد الحد الأقصى للنطاق المستخدم من قبل الكائن

**Change** الحدث  
يتحقق عندما تتغير قيمة الكائن

**Scroll** الحدث  
يتحقق عندما يتم عمل سحب بالفارة لتغيير قيمة الكائن

## التعليمات التالية هي المتوفرة للتعامل مع الصورة Picture

```
* I want picture
* Picture file is ....data as value....
* Picture top is ....data as value....
* Picture left is ....data as value....
* Picture width is ....data as value....
* Picture height is ....data as value....
* Picture name is ....data as value....
* ....data as value.... Picture selected
```

Picture Events

-----  
mouseclick

**I want picture** الامر  
يستخدم لإنشاء كائن جديد

**Picture file is** الامر  
يستخدم لتحديد ملف الصورة

**Picture top is** الامر  
يستخدم لتحديد نقطة البداية العلوية للصورة على سطح النافذة

**Picture left is** الامر  
يستخدم لتحديد نقطة البداية اليسرى للصورة على سطح النافذة

**Picture width is** الامر  
يستخدم لتحديد عرض الصورة

**Picture height is** الامر  
يستخدم لتحديد ارتفاع الصورة

**Picture name is** الامر  
يستخدم لتحديد اسم كائن الصورة لتحديد والتعامل معها فيما بعد

**Picture selected** الامر  
يستخدم لتحديد الصورة للتعامل معها والتعديل في خصائصها

**Mouse Click** الحدث  
يتحقق عند النقر بالفأرة على الصورة

## Date Picker التعليمات التالية هي المتوفرة للتعامل مع ملتقط التاريخ

### DatePicker

```
-----  
* I want datepicker  
* Datepicker value is ....data as value....  
* Datepicker top is ....data as value....  
* Datepicker left is ....data as value....  
* Datepicker width is ....data as value....  
* Datepicker height is ....data as value....  
* Datepicker name is ....data as value....  
* ....data as value.... datepicker selected  
* Datepicker fore color is ....data as value....  
* Datepicker back color is ....data as value....  
* Datepicker font name is ....data as value....  
* Datepicker font size is ....data as value....  
* Datepicker font bold is ....data as value....  
* Datepicker font italic is ....data as value....  
* Datepicker font underline is ....data as value....  
* Datepicker font strickout is ....data as value....
```

### Datepicker Events

```
-----  
gotfocus  
lostfocus  
change  
enter
```

### الامر I want datepicker

يستخدم لإنشاء كائن جديد لتحديد التاريخ

### الامر Date picker value is

يستخدم لتحديد القيمة (التاريخ)

### الامر Datepicker top is

يستخدم لتحديد نقطة البداية العلوية

### الامر Datepicker left is

يستخدم لتحديد نقطة البداية اليسرى

### الامر Datepicker width is

يستخدم لتحديد العرض

الامر Datepicker height is  
يستخدم لتحديد الارتفاع

الامر Datepicker name is  
يستخدم لتحديد اسم الكائن ليستخدم فيما بعد لتحديده والتعامل معه

الامر Datepicker selected  
يستخدم لتحديد كائن عن طريق اسمه للتعامل معه والتعديل في خواصه

الامر Datepicker fore color is  
يستخدم لتحديد لون الكتابة

الامر Datepicker back color is  
يستخدم لتحديد لون الخلفية

الامر Datepicker font name is  
يستخدم لتحديد اسم الخط

الامر Datepicker font size is  
يستخدم لتحديد حجم الخط

الامر Datepicker font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Datepicker font italic is  
يستخدم لتحديد هل الخط مائل ام لا

الامر Datepicker font under line is  
يستخدم لتحديد هل هناك خط تحت النص ام لا

الامر Datepicker font strick out  
يستخدم لتحديد هل الخط بارز ام لا

الحدث Got Focus  
يتحقق عندما يمتلك ملتقط التاريخ البؤرة

## الحدث Lost focus

يتحقق عندما يفقد ملتقط التاريخ البؤرة

## الحدث Change

يتحقق عندما تتغير القيمة (التاريخ)

## الحدث Enter

يتحقق عندما يتم الضغط على مفتاح الادخال

التعليمات التالية هي المتوفرة للتعامل مع الوصلة التشعبية HyperLink

HyperLink

-----

```
* I want hyperlink
* Hyperlink caption is ....data as value....
* Hyperlink top is ....data as value....
* Hyperlink left is ....data as value....
* Hyperlink width is ....data as value....
* Hyperlink height is ....data as value....
* Hyperlink name is ....data as value....
* Hyperlink font name is ....data as value....
* Hyperlink font size is ....data as value....
* Hyperlink fore color is ....data as value....
* Hyperlink back color is ....data as value....
* Hyperlink font bold is ....data as value....
* Hyperlink font italic is ....data as value....
* Hyperlink font underline is ....data as value....
* Hyperlink font strickout is ....data as value....
* Set hyperlink transparent on
* Set hyperlink transparent off
* ....data as value.... hyperlink selected
* Hyperlink address is ....data as value....
```

## الامر I want hyperlink

يستخدم لانشاء وصلة تشعبية جديدة

## الامر Hyperlink caption is

يستخدم لتحديد النص

## الامر Hyperlink top is

يستخدم لتحديد نقطة البداية العلوية للوصلة التشعبية على سطح النافذة

الامر **Hyperlink left is** يستخدم لتحديد نقطة البداية اليسرى للوصلة التشعبية على سطح النافذة

الامر **Hyperlink width is** يستخدم لتحديد العرض

الامر **Hyperlink Height is** يستخدم لتحديد الارتفاع

الامر **Hyperlink name is** يستخدم لتحديد اسم الكائن لتحديده فيما بعد والتعديل في خواصه

الامر **Hyperlink font name is** يستخدم في تحديد اسم الخط

الامر **Hyperlink font size is** يستخدم في تحديد حجم الخط

الامر **Hyperlink fore color is** يستخدم في تحديد لون الكتابة

الامر **Hyperlink back color is** يستخدم في تحديد لون الخلفية

الامر **Hyperlink font bold is** يستخدم في تحديد هل الخط سميك ام لا عن طريق True او False

الامر **Hyperlink font italic is** يستخدم في تحديد هل الخط مائل ام لا

الامر **Hyperlink font under line is** يستخدم في تحديد هل هناك خط تحت النص ام لا

الامر **Hyperlink font strick out is** يستخدم في تحديد هل الخط بارز ام لا

الامر **Set hyperlink transparent on**  
يستخدم لجعل الوصلة التشعبية شفافة (عدم اظهار لون للخلفية)

الامر **Set hyperlink transparent off**  
يستخدم لالغاء الشفافية للوصلة التشعبية (تفعيل لون الخلفية)

الامر **Hyperlink selected**  
يستخدم لتحديد وصلة تشعبية للتحكم فى خصائصها

الامر **Hyperlink address is**  
يستخدم فى تحديد عنوان الويب للوصلة التشعبية

التعليمات التالية هى المتوفرة للتعامل مع الصندوق الرقمى Spinner

```
Spinner  
-----
```

```
* I want spinner  
* Spinner value is ....data as value....  
* Spinner top is ....data as value....  
* Spinner left is ....data as value....  
* Spinner width is ....data as value....  
* Spinner height is ....data as value....  
* Spinner name is ....data as value....  
* ....data as value.... Spinner selected  
* Spinner fore color is ....data as value....  
* Spinner back color is ....data as value....  
* Spinner font name is ....data as value....  
* Spinner font size is ....data as value....  
* Spinner font bold is ....data as value....  
* Spinner font italic is ....data as value....  
* Spinner font underline is ....data as value....  
* Spinner font strickout is ....data as value....  
* Spinner range minimum is ....data as value....  
* Spinner range maximum is ....data as value....
```

```
Spinner Events  
-----
```

```
gotfocus  
lostfocus  
change
```

الامر **I want spinner**  
يستخدم لانشاء صندوق رقمى جديد

الامر Spinner value is  
يستخدم لتحديد قيمة الصندوق الرقمي

الامر Spinner top is  
يستخدم لتحديد نقطة البداية العلوية للصندوق الرقمي على سطح النافذة

الامر Spinner left is  
يستخدم لتحديد نقطة البداية اليسرى للصندوق الرقمي على سطح النافذة

الامر Spinner width is  
يستخدم لتحديد عرض الصندوق الرقمي

الامر Spinner height is  
يستخدم لتحديد ارتفاع الصندوق الرقمي

الامر Spinner name is  
يستخدم لتحديد اسم الصندوق الرقمي لتحديده فيما بعد والتعديل في خصائصه

الامر Spinner selected  
يستخدم لتحديد صندوق رقمي من خلال اسمه

الامر Spinner fore color is  
يستخدم لتحديد لون الكتابة

الامر Spinner back color is  
يستخدم لتحديد لون الخلفية

الامر Spinner font name is  
يستخدم لتحديد اسم الخط

الامر Spinner font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Spinner font italic is  
يستخدم لتحديد هل الخط مائل ام لا



الامر Spinner font under line is  
يستخدم لتحديد هل هناك خط تحت الرقم ام لا

الامر Spinner font strick out is  
يستخدم لتحديد هل الخط بارز ام لا

الامر Spinner range minimum is  
يستخدم لتحديد الحد الادنى للنطاق المستخدم من قبل الصندوق الرقمي

الامر Spinner range maximum is  
يستخدم لتحديد الحد الاقصى للنطاق المستخدم من قبل الصندوق الرقمي

الحدث Got Focus  
يتحقق عندما يمتلك الصندوق الرقمي البؤرة

الحدث Lost focus  
يتحقق عندما يفقد الصندوق الرقمي البؤرة

الحدث Change  
يتحقق عندما تتغير القيمة الرقمية في الصندوق الرقمي

التعليمات التالية هي المتوفرة للتعامل مع شريط الحالة Statusbar

```
* I want statusbar
* Statusbar font name is ....data as value....
* Statusbar font size is ....data as value....
* Statusbar font bold is ....data as value....
* Statusbar font italic is ....data as value....
* Statusbar font underline is ....data as value....
* Statusbar font strickout is ....data as value....
* I want status item
* I want status clock
* I want status date
* I want status keyboard
* Status item caption is ....data as value....
* Status item width is ....data as value....
* You show statusbar
```

الامر I want status bar  
يستخدم لانشاء شريط الحالة

الامر Statusbar font name is  
يستخدم لتحديد اسم الخط

الامر Statusbar font size is  
يستخدم لتحديد حجم الخط

الامر Statusbar font bold is  
يستخدم لتحديد هل الخط سميك ام لا عن طريق True او False

الامر Statusbar font italic is  
يستخدم هذا الامر لتحديد هل الخط مائل ام لا

الامر Statusbar font under line is  
يستخدم هذا الامر لتحديد هل هناك خط تحت النص ام لا

الامر Statusbar font strick out is  
يستخدم هذا الامر لتحديد هل الخط سميك ام لا

الامر I want status item  
يستخدم لانشاء عنصر فى شريط الحالة

الامر I want status clock  
يستخدم لاضافة ساعة الى شريط الحالة

الامر I want status date  
يستخدم لاضافة التاريخ الى شريط الحالة

الامر I want status keyboard  
يستخدم لاضافة معرفة حالة لوحة المفاتيح الى شريط الحالة

الامر Status item caption is  
يستخدم لتحديد نص العنصر

الامر Status item width is  
يستخدم لتحديد عرض العنصر

الامر You show status bar  
يستخدم لعرض شريط الحالة

## التعليمات التالية هي المتوفرة للتعامل مع شريط القوائم Menubar

```
Menubar
-----

* I want menubar
* I want menu popup
* Menu popup caption is ....data as value....
* I want menu item
* Menu item caption is ....data as value....
* Menu item name is ....data as value....

Menu Item Events
-----

        mouseclick
```

**I want menubar** الامر  
يستخدم لإنشاء شريط القوائم

**I want menu popup** الامر  
يستخدم لإنشاء قائمة

**Menu popup caption is** الامر  
يستخدم لتحديد عنوان للقائمة

**I want menu item** الامر  
يستخدم لإنشاء عنصر في القائمة

**Menu item caption is** الامر  
يستخدم لتحديد نص العنصر

**Menu item name is** الامر  
يستخدم لتحديد اسم العنصر

**Mouse click** الحدث  
يتحقق عندما يتم اختيار عنصر في القائمة

والآن حان الوقت لكي ننطلق في الاطلاع على مجموعة من الامثلة البسيطة

## سنبدأ بمثال يوضح احداث النافذة Window Events

الشفيرة المصدرية كالتالى

```
set window autoshow off

I want window and The Window Name is win1.
The window title is Just move me to see what will happen!.
Window top is 50. and window left is 50.
Window width is 350. and window height is 100.
The window back color is {55,155,205}. and you show window

I want window and The Window Name is win2.
The window title is Oh what will happen to me ?!.
The window back color is {255,205,175}.
Window top is 250. and window left is 250.
Window width is 350. and window height is 100. and You show
window

win1 move. instructions are
    win1. window selected and The window title is "window is
moved!!!"
    win2. window selected and The window backcolor is {0,0,0}.
    The window title is Oh no, i hate the BLACK color mmmmmm.
end of instructions

win1 gotfocus. instructions are
    The window title is Wow i have the focus.
end of instructions

win1 lostfocus. instructions are
    The window title is Oh no i have lost the focus.
end of instructions

win1 mouseclick. instructions are
    The window title is WOW the window is Clicked!.
end of instructions


win1 mousemove. instructions are
    The window title is WOW the mouse is Moved!.
end of instructions





win1 resize. instructions are
    The window title is WOW the window is Resized!.
end of instructions

win2 maximize. instructions are
    The window title is Maximized!.
end of instructions

win2 minimize. instructions are
    The window title is Minimized!.
end of instructions
```

## والتطبيق وقت العمل كالتالى





## والشفيرة المصدرية باللغة العربية كالتالى

امنع الاظهار التلقائى للنافذة  
انا اريد نافذة  
النافذة تحت عنوان هو حركنى لترى ماذا سوف يحدث ان شاء الله.  
يانافذة البداية العلوية هي 50.  
يانافذة البداية اليسرى هي 50.  
يانافذة العرض هو 350.  
يانافذة الارتفاع هو 100.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة الاسم هو "ن1"  
لوسمحت اعرض النافذة  
انا اريد نافذة  
يانافذة الاسم هو "ن2"  
النافذة تحت عنوان هو ياترى ايه الى هيجصلى.  
يانافذة لون الخلفية هو {255,205,175}.  
يانافذة البداية اليسرى هي 250.  
يانافذة البداية العلوية هي 250.  
يانافذة العرض هو 350.  
يانافذة الارتفاع هو 100.  
لوسمحت اعرض النافذة  
"ن1 التحرك" تعليماته هي  
"ن1" نافذة تم تحديدها  
النافذة تحت عنوان هو "انا اتحركت يا جماعة الخير !!!"  
"ن2" نافذة تم تحديدها  
النافذة تحت عنوان هو "لا لا لا انا ايه بيحصلى !!!"  
يانافذة لون الخلفية هو {0,0,0}.  
نهاية التعليمات  
"ن1 امتلاك البؤرة" تعليماته هي  
النافذة تحت عنوان هو "ياسلام انا امتلكت البؤرة"  
نهاية التعليمات  
"ن1 فقدان البؤرة" تعليماته هي  
النافذة تحت عنوان هو "للاسف انا اعانى من فقدان البؤرة"  
نهاية التعليمات  
"ن1 النقر بالفارة" تعليماته هي  
النافذة تحت عنوان هو "ياسلام تم النقر عليا بالفارة ليه !!!"  
نهاية التعليمات  
"ن1 التحرك بالفارة" تعليماته هي  
النافذة تحت عنوان هو "ياسلام ارى الفارة تتحرك على سطحى !!!"  
نهاية التعليمات  
"ن1 اعادة التحجيم" تعليماته هي  
النافذة تحت عنوان هو "ياسلام تم تغيير حجمى ياترى كبرت والا صغرت !!!"  
نهاية التعليمات  
"ن2 التكبير" تعليماته هي  
النافذة تحت عنوان هو "ياسلام هو فى اكبر من كده !!!"  
نهاية التعليمات  
"ن2 التصغير" تعليماته هي  
النافذة تحت عنوان هو "ياسلام على القهر انا صغيرة مرة !!!"  
نهاية التعليمات

والتطبيق وقت التشغيل كالتالى



والان لدينا مثال اخر يشتمل عنوان وزر Label and Button والشفيرة المصدرية كالتالى

```
I want window
The window title is "Label and Button" and The window height is
100.


I want label and the label caption is "Label"
I want button and the button caption is "Button"

The label left is 50. and the button left is 250.
The label width is 200. and the button width is 200.

The label name is lbl1. and lbl1 mouseclick. instructions are
    The button caption is message form label click.
end of instructions

The button name is btn1. and btn1 mouseclick. instructions are
    The label caption is message form button click.
end of instructions
```

والتطبيق وقت التشغيل كالتالى





حيث يكون هناك عنوان عند الضغط عليه يتغير نص الزر الى Message from label click اي رسالة من الضغط على الزر يتغير نص العنوان الى Message from button click اي رسالة من الضغط على الزر



ولدينا مثال الان يشتمل على عنوان وزرين Label and Two Buttons والشفيرة المصدرية كالتالى

```
Set window auto show off

I want window and the window title is "Label and two Buttons"
The window width is 250. and the window height is 200.
The window bgcolor is {100,200,250}.

I want button and the button caption is One (1).
The button top is 100. and the button left is 10.
The button width is 100. and the button height is 50.
The button name is btn1.
The button forecolor is {0,0,255}.
The button font bold is true.

I want button and the button caption is Two (2).
The button top is 100. and the button left is 130.
The button width is 100. and the button height is 50.
The button name is btn2.
The button forecolor is {0,0,255}.
The button font bold is true.




I want label and the label height is 40.
the label top is 30. the label left is 70.
The label font name is arial. and the label font size is 30.
The label forecolor is {255,0,0}. and the label bgcolor is
{255,255,255}.
The label font bold is true.
The label font italic is true.
The label font under line is true.
The label font strick out is true.

You show window

btn1 mouseclick. instructions are
    The label caption is One.
end of instructions

btn2 mouseclick. instructions are
    The label caption is Two.
end of instructions
```

والتطبيق وقت التشغيل كالتالى



حيث لدينا عنوان فارغ  
وهناك زر يحمل النص One (1)  
وهناك زر اخر يحمل النص Two (2)

وعند الضغط على اى زر من الزرين يتغير العنوان ليحمل نص يشير الى الرقم  
الذى يشير اليه الزر.

والآن لدينا مثال اخر يشتمل على عنوان وثلاثة ازرار

والشفيرة المصدرية كالتالى

```
Set window auto show off

I want window and the window title is "Label and Three Buttons"
The window width is 370. and the window height is 200.
The window back color is {170,180,190}.

I want label and the label height is 40.
The label name is lbl1.
the label top is 30. the label left is 130.
The label font name is arial. and the label font size is 30.
The label backcolor is {0,255,0}.
The label font bold is true.
The label font italic is true.
The label font under line is true.
The label font strick out is true.

I want button and the button caption is Red.
The button top is 100. and the button left is 10.
The button width is 100. and the button height is 50.
The button name is btn1.
The button forecolor is {0,0,255}.
The button font bold is true.

I want button and the button caption is Green.
The button top is 100. and the button left is 130.
The button width is 100. and the button height is 50.
The button name is btn2.
The button forecolor is {0,0,255}.
The button font bold is true.

I want button and the button caption is Blue.
The button top is 100. and the button left is 250.
The button width is 100. and the button height is 50.
The button name is btn3.
The button forecolor is {0,0,255}.
The button font bold is true.




You show window

btn1 mouseclick. instructions are
the label backcolor is {255,0,0}.
end of instructions

btn2 mouseclick. instructions are
the label backcolor is {0,255,0}.
end of instructions

btn3 mouseclick. instructions are
lbl1. label selected
the label backcolor is {0,0,255}.
end of instructions
```

والتطبيق وقت التشغيل كالتالى




حيث اننا نجد التطبيق عبارة عن نافذة و تشتمل هذه النافذة على عنوان بدون نص  
بالاضافة الى ثلاثة ازرار وهى  
زر يحمل النص Red اشارة الى اللون الاحمر  
زر يحمل النص Green اشارة الى اللون الاخضر  
زر يحمل النص Blue اشارة الى اللون الازرق  
بالمناسبة (هذه الالوان الثلاثة هى الالوان الاساسية التى تكون بقية الالوان الاخرى)  
عند الضغط على اى زر من هذه الازرار يتغير لون الخلفية للعنوان الذى بدون نص  
لكى يصبح لون الخفيه هو اللون الذى يشير اليه الزر.

## Checkbox and Two buttons مربع فحص وزرين

والشفيرة المصدرية للمثال كالتالى

```
set window auto show off
I want window and the window title is "checkbox"
The window back color is {100,40,200}.
The window width is 260.
The window height is 200.
I want checkbox and the checkbox name is mycheck.
The checkbox top is 50. and the checkbox left is 10.
The checkbox width is 200. and the checkbox height is 30.
The checkbox font name is "arial"
The checkbox font size is 14.
The checkbox forecolor is {255,255,255}.
The checkbox backcolor is {50,200,155}.
The checkbox caption is "Checkbox"
I want button and the button caption is On (1).
The button top is 100. and the button left is 10.
The button width is 100. and the button height is 50.
The button name is btn1.
The button forecolor is {0,0,255}.
The button font bold is true.
I want button and the button caption is Off (0).
The button top is 100. and the button left is 130.
The button width is 100. and the button height is 50.
The button name is btn2.
The button forecolor is {0,0,255}.
The button font bold is true.
You show window
btn1 mouseclick. instructions are
    The checkbox value is true.
end of instructions
btn2 mouseclick. instructions are
    The checkbox value is false.
end of instructions
```

والتطبيق وقت التشغيل كالتالى



نجد لدينا مربع فحص يتم تعديل قيمته من خلال زر للتشغيل On وزر للايقاف Off

والان ننتقل لمثال اخر يوضح كيفية تحديد الالوان الشائعة من خلال اسمها

والشفيرة المصدرية كالتالى

```
i want window and the window title is hello.
the window back color is blue.
the window width = 350. and the window height = 200.

i want label and the label caption is Welcome.
the label font size is 50.
the label width is 300. and the label height is 80.
the label fore color is white.
the label back color is blue.

i want button and the button caption is close.
the button top is 100. and the button left is 20.
the button width is 300. and the button name is btn1.
the button fore color is purple.
the button back color is white.

Btn1 MouseClick. instructions are
    You close window
end of instructions
```

والتطبيق وقت التشغيل كالتالى



والالوان الجاهزة للاستخدام عن طريق الاسم هي

aqua	{0,255,255}	black	{0,0,0}
blue	{0,0,255}	darkgray	{128,128,128}
fuchsia	{255,0,255}	gray	{128,128,128}
green	{0,128,0}	lime	{0,255,0}
lightgray	{192,192,192}	maroon	{128,0,0}
navy	{0,0,128}	purple	{128,0,128}
red	{255,0,0}	silver	{192,192,192}
teal	{128,128,0}	white	{255,255,255}
yellow	{255,255,0}		

## والشفيرة المصدرية باللغة العربية كالتالى

انا اريد نافذة والنافذة تحت عنوان هو مرحبا.  
يانافذة لون الخلفية هو ازرق.  
يانافذة العرض هو 350. يانافذة الارتفاع هو 200.

انا اريد عنوان وياعنوان النص هو مرحبا.  
ياعنوان حجم الخط هو 50.  
ياعنوان العرض هو 300. ياعنوان الارتفاع هو 80.  
ياعنوان لون الكتابة هو ابيض.  
ياعنوان لون الخلفية هو ازرق.

انا اريد زر ويازر النص هو خروج.  
يازر البداية العلوية هي 100.  
يازر البداية اليسرى هي 20.  
يازر العرض هو 300.  
يازر الاسم هو "زر1"  
يازر لون الكتابة هو الارجوان.  
يازر لون الخلفية هو ابيض.

"زر1 النقر بالفارة" تعليماته هي

لوسمحت اغلق النافذة

نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى



والان نتابع المسيرة نحو الاطلاع على المزيد من الامثلة


ولدينا مجموعة من الامثلة التى تبين كيف نقوم بانشاء مجموعة من الكائنات ذات خصائص مشتركة.

المثال التالي ينشئ مجموعة نوافذ ذات خصائص مشتركة

الشفيرة المصدرية كالتالي

```
set window auto show off
i want window and the window name is basewin.
The window title is The main window.
You maximize window and You show window
I want window and the window name is mywin.
mywin init. instructions must done and you show window
mywin init. instructions are
mywin. window selected
    the window top is 50. and the window left is 50.
    the window width is 200. and the window height is 200.
    the window title is wow. and the window back color is {255,0,0}.
end of instructions
mywin mouseclick. instructions are
    I want window and the window name is mywin.
    mywin init. instructions must done and you show window
end of instructions
mywin maximize. instructions are
    the window title is Maximize!!.
end of instructions
```

والتطبيق وقت التشغيل كالتالي







والان لدينا مثال حول انشاء اكثر من مربع نص بنفس الخصائص

والشفيرة المصدرية كالتالى

```
I want window
The window title is "Textbox"
The window width is 230. and the window height is 130.
I want textbox and mytext init. instructions must done
I want textbox and mytext init. instructions must done
The textbox top is 50.
mytext init. instructions are
The textbox name is mytext.
The textbox font name is "arial"
The textbox font size is 14.
The textbox fore color is {0,0,255}.
The textbox back color is {100,255,100}.
end of instructions
mytext lostfocus. instructions are
    The textbox fore color is {50,100,150}.
    The textbox back color is {255,255,255}.
end of instructions
mytext gotfocus. instructions are
    The textbox fore color is {255,255,255}.
    The textbox back color is {50,100,150}.
end of instructions
```

والتطبيق وقت التشغيل كالتالى





والان لدينا مثال على انشاء اكثر من مربع تحرير بنفس الخصائص

والشفيرة المصدرية للمثال كالتالى

```
set window auto show off

I want window and the window title is "Editbox"
The window height is 520. and the window width is 440.

I want editbox and myeditbox init. instructions must done
the editbox bgcolor is {100,200,250}.
the editbox value is "Welcome"

I want editbox and myeditbox init. instructions must done
the editbox left is 220.
the editbox value is "How r u ?"

I want editbox and myeditbox init. instructions must done
the editbox top is 220.
the editbox value is "Thank u"

I want editbox and myeditbox init. instructions must done
the editbox top is 220. and the editbox left is 220.
the editbox bgcolor is {250,200,100}.
the editbox value is "I am fine !!"

I want button and the button top is 440.
The button height is 30. and the button width is 410.
The button caption is "Close"
The button name is mybtn.


you show window

mybtn mouseclick. instructions are
    you close window
end of instructions

myeditbox init. instructions are
    The editbox name is myeditbox.
    The editbox font name is "arial"
    The editbox font size is 14.
end of instructions

myeditbox change. instructions are
    The editbox back color is {50,100,150}.
    The editbox fore color is {255,255,255}.
end of instructions
```

والتطبيق وقت التشغيل كالتالى حيث بالتعديل فى محتوى مربع النص يتغير لونه.



والآن لدينا مثال على انشاء اكثر من مربع قائمة Listbox بنفس الخصائص

والشفيرة المصدرية للمثال كالتالى

```
Set window auto show off

I want window and the window title is "Listbox"
The window width is 500.
The window height is 330.

I want label and the label top is 210.
The label width is 300.
The label font name is "Arial"
The label font size is 14.

I want button and the button caption is Close.
The button name is mybtn. and the button top is 250.

I want listbox and mylist init. instructions must done

I want listbox and mylist init. instructions must done
The listbox left is 230.

I want listbox and mylist init. instructions must done
The listbox left is 460.

You maximize window and you show window

mylist init. instructions are
The listbox name is mylist.
The listbox font name is "arial"
The listbox font size is 14.
The listbox must add item One (1).
The listbox must add item Two (2).
The listbox must add item Three (3).
The listbox must add item Four (4).
The listbox must add item Five (5).
The listbox value is 3.
end of instructions




mylist change. instructions are
    The label caption is "listbox value changed."
end of instructions

mylist gotfocus. instructions are
    The listbox forecolor is {255,255,255}.
    The listbox backcolor is {255,0,0}.
end of instructions

mylist lostfocus. instructions are
    The listbox forecolor is {0,0,0}.
    The listbox backcolor is {255,255,255}.
end of instructions

mybtn mouseclick. instructions are
    You close window
end of instructions
```

## والتطبيق وقت التشغيل كالتالى



## والان لدينا مثال على انشاء اكثر من قائمة منسدلة Combobox بخصائص مشتركة والشفيرة المصدرية كالتالى

```
Set window auto show off

I want window and the window title is "combobox"
The window width is 500.
The window height is 330.

I want label and the label top is 210.
The label width is 300.
The label font name is "Arial"
The label font size is 14.

I want button and the button caption is Close.
The button name is mybtn. and the button top is 250.

I want combobox and mycombo init. instructions must done

I want combobox and mycombo init. instructions must done
The combobox left is 230.

I want combobox and mycombo init. instructions must done
The combobox left is 460.

You maximize window and you show window

mycombo init. instructions are
The combobox name is mycombo.
The combobox width is 200.
The combobox height is 200.
The combobox font name is "arial"
The combobox font size is 14.
The combobox must add item One (1).
The combobox must add item Two (2).
The combobox must add item Three (3).
The combobox must add item Four (4).
The combobox must add item Five (5).
The combobox value is 3.
end of instructions


mycombo change. instructions are
    The label caption is "combobox value changed."
end of instructions

mycombo gotfocus. instructions are
    The combobox forecolor is {255,255,255}.
    The combobox backcolor is {255,0,0}.
end of instructions

mycombo lostfocus. instructions are
    The combobox forecolor is {0,0,0}.
    The combobox backcolor is {255,255,255}.
end of instructions

mybtn mouseclick. instructions are
    You close window
end of instructions
```





## والتطبيق وقت العمل كالتالي



والان لدينا مثال على انشاء اكثر من مربع فحص بنفس الخصائص

```
Set window auto show off
I want window and the window title is "checkbox"
The window width is 500.
The window height is 330.
I want label and the label top is 210.
The label width is 300.
The label font name is "Arial"
The label font size is 14.
I want button and the button caption is Close.
The button name is mybtn. and the button top is 250.
I want checkbox and mycheck init. instructions must done
I want checkbox and mycheck init. instructions must done
The checkbox left is 230.
I want checkbox and mycheck init. instructions must done
The checkbox left is 460.
You maximize window and you show window
mycheck init. instructions are
    The checkbox name is mycheck.
    The checkbox width is 200.
    The checkbox height is 200.
    The checkbox font name is "arial"
    The checkbox font size is 14.
end of instructions
mycheck change. instructions are
    The label caption is "checkbox value changed."
end of instructions
mycheck gotfocus. instructions are
    The checkbox forecolor is {255,255,255}.
    The checkbox backcolor is {255,0,0}.
end of instructions
mycheck lostfocus. instructions are
    The checkbox forecolor is {0,0,0}.
    The checkbox backcolor is {255,255,255}.
end of instructions
mybtn mouseclick. instructions are
    You close window
end of instructions
```

## والتطبيق وقت العمل كالتالى





والان لدينا مجموعة من الامثلة بشفيرة مصدرية باللغة العربية

## مثال على العنوان Label

التطبيق تحت عنوان هو تجربة .

انا اريد نافذة و النافذة تحت عنوان هو تجربة .  
يانافذة الاتجاه هو بداية من اليمين لليسار .  
يانافذة لون الخلفية هو {255,255,255} .  
يانافذة العرض هو 530 .  
يانافذة الارتفاع هو 200 .

انا اريد عنوان  
ياعنوان النص هو السلام عليكم .  
ياعنوان سمك الخط هو "1"

## والتطبيق وقت التشغيل كالتالى



## مثال على مربع النص Textbox

التطبيق تحت عنوان هو تجربة .

انا اريد نافذة و النافذة تحت عنوان هو تجربة .  
يانافذة الاتجاه هو بداية من اليمين لليسار .  
يانافذة لون الخلفية هو {255,255,255} .  
يانافذة العرض هو 530 .  
يانافذة الارتفاع هو 200 .

انا اريد مربع نص  
يامربع نص القيمة هي السلام عليكم .

## والتطبيق وقت التشغيل كالتالى



## مثال على مربع تحرير Editbox

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد مربع تحرير  
يامربع تحرير القيمة هي السلام عليكم.

## والتطبيق وقت التشغيل كالتالى



## مثال على زر Button

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 200.

انا اريد زر  
يازر النص هو السلام عليكم.

والتطبيق وقت التشغيل كالتالى



## مثال على مربع قائمة Listbox

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد مربع القائمة  
يامربع القائمة يجب ان تضيف عنصر السلام عليكم.  
يامربع القائمة يجب ان تضيف عنصر السلام عليكم.  
يامربع القائمة يجب ان تضيف عنصر السلام عليكم.

## والتطبيق وقت التشغيل كالتالى



## مثال على قائمة منسدلة Combobox

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد كومبو  
ياكومبو يجب ان تضيف عنصر السلام عليكم.  
ياكومبو يجب ان تضيف عنصر السلام عليكم 2.  
ياكومبو يجب ان تضيف عنصر السلام عليكم 3.  
ياكومبو الارتفاع هو 100.  
ياكومبو القيمة هي 2.  
ياكومبو العرض هو 200.

## والتطبيق وقت التشغيل كالتالى



## مثال على مربع فحص Checkbox

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد مربع فحص  
يامربع فحص النص هو احب العربي.  
يامربع فحص القيمة هي 1.  
يامربع فحص سمك الخط هو 1.

انا اريد مربع فحص  
يامربع فحص البداية العلوية هي 100.  
يامربع فحص النص هو احب العربي.  
يامربع فحص القيمة هي 0.

## والتطبيق وقت التشغيل كالتالي



## مثال على صندوق رقمي Spinner

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد سبينر  
ياسبينر القيمة هي 50.  
ياسبينر بداية النطاق هي 1.  
ياسبينر نهاية النطاق هي 70.

## والتطبيق وقت التشغيل كالتالى



## مثال على السوستة Slider

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.

انا اريد سوستة  
ياسوستة القيمة هي 50.  
ياسوستة بداية النطاق هي 1.  
ياسوستة نهاية النطاق هي 70.

## والتطبيق وقت التشغيل كالتالى



## مثال على ملتقط التاريخ Datepicker

التطبيق تحت عنوان هو تجربة.  
انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.  
انا اريد ملتقط تاريخ

## والتطبيق وقت التشغيل كالتالى



## مثال على الوصلة التشعبية Hyperlink

التطبيق تحت عنوان هو تجربة.  
انا اريد نافذة و النافذة تحت عنوان هو تجربة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 300.  
انا اريد رابط تشعبى  
يارابط تشعبى النص هو "موقع السوبر نوبا على السورس فورج"  
يارابط تشعبى العرض هو 400.  
يارابط تشعبى حجم الخط هو 14.  
يارابط تشعبى رابط الويب هو "http://www.sourceforge.net/projects/supernova"

## والتطبيق وقت التشغيل كالتالى



## مثال على شريط التقدم Progressbar

التطبيق تحت عنوان هو تجربة.

انا اريد نافذة و النافذة تحت عنوان هو تجربة.

يانافذة الاتجاه هو بداية من اليمين لليساار.

يانافذة لون الخلفية هو {255,255,255}.

يانافذة العرض هو 530.

يانافذة الارتفاع هو 300.

انا اريد شريط تقدم

ياشريط تقدم القيمة هي 50.

انا اريد شريط تقدم

ياشريط تقدم البداية العلوية هي 100.


ياشريط تقدم القيمة هي 80.

والتطبيق وقت التشغيل كالتالى

حيث قمنا بانشاء اثنين من اشربة التقدم

تم اسناد القيمة (50) الى الشريط الاول

بينما تم اسناد القيمة (80) الى الشريط الثانى





## مثال على الصورة Picture

التطبيق تحت عنوان هو ياحلاوة الصورة.


انا اريد نافذة و النافذة تحت عنوان هو ياحلاوة الصورة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 530.  
يانافذة الارتفاع هو 550.

انا اريد صورة  
ياصورة العرض هو 317.  
ياصورة الارتفاع هو 490.  
ياصورة الملف هو "cover.jpg"

انا اريد زر  
يازر الاسم هو "زر1"  
يازر النص هو "خروج من البرنامج"  
يازر البداية اليسرى هي 350.  
يازر اسم الخط هو arial.  
يازر حجم الخط هو 14.  
يازر العرض هو 150.

"زر1 النقر بالفارة" تعليماته هي  
لوسمحت اغلق النافذة  
نهاية التعليمات

## والتطبيق وقت التشغيل كالتالى



## مثال شريط القوائم Menubar وشريط الحالة Statusbar

امنع الاظهار التلقائى للنافذة

التطبيق تحت عنوان هو ياحلاوة شريط القائمة.

انا اريد نافذة و النافذة تحت عنوان هو ياحلاوة شريط القائمة.

يانافذة الاتجاه هو بداية من اليمين لليساار.

يانافذة لون الخلفية هو {255,255,255}.

يانافذة العرض هو 530.

يانافذة الارتفاع هو 550.

انا اريد شريط القوائم

انا اريد قائمة فرعية وياقائمة فرعية النص هو ملف.

انا اريد عنصر فى القائمة وياعنصر فى القائمة النص هو خروج.

ياعنصر فى القائمة الاسم هو "زر1"

انا اريد شريط حالة

انا اريد عنصر حالة

ياعنصر حالة النص هو جاهز.

"زر1 النقر بالفارة" تعليماته هي

لوسمحت اغلق النافذة

نهاية التعليمات

لوسمحت قم بتكبير النافذة و لوسمحت اعرض النافذة

والتطبيق وقت التشغيل كالتالى  
حيث لدينا شريط قائمة يشتمل على قائمة ملف بها الاختيار (خروج)  
ولدينا شريط حالة يحتوى على عنصر يحمل النص (جاهز)



والان لدينا مثال يعرض سؤال واختيارات ويقيم الاجابة والشفيرة المصدرية كالتالى

التطبيق تحت عنوان هو "عاصمة مصر ؟".  
امنح الازهار التلقائي للنافذة  
انا اريد نافذة و النافذة تحت عنوان هو مرحبا.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,255,255}.  
يانافذة العرض هو 350.  
انا اريد عنوان  
ياعنون النص هو " عاصمة جمهورية مصر العربية ؟ "  
ياعنون اسم الخط هو arial. و ياعنون حجم الخط هو 14.  
ياعنون العرض هو 300.  
انا اريد زر و يازر الاسم هو "زر1"  
يازر النص هو مدينة الاسكندرية.  
يازر اسم الخط هو arial. و يازر حجم الخط هو 14.  
يازر العرض هو 300.  
يازر البداية العلوية هي 100. و يازر البداية اليسرى هي 30.  
انا اريد زر و يازر الاسم هو "زر2"  
يازر النص هو مدينة سرس الليان - منوفية.  
يازر اسم الخط هو arial. و يازر حجم الخط هو 14.  
يازر العرض هو 300.  
يازر البداية العلوية هي 150. و يازر البداية اليسرى هي 30.  
انا اريد زر و يازر الاسم هو "زر3"  
يازر النص هو مدينة القاهرة.  
يازر اسم الخط هو arial. و يازر حجم الخط هو 14.  
يازر العرض هو 300.  
يازر البداية العلوية هي 200. و يازر البداية اليسرى هي 30.  
انا اريد زر و يازر الاسم هو "زر4"  
يازر النص هو خروج.  
يازر اسم الخط هو arial. و يازر حجم الخط هو 14.  
يازر العرض هو 150.  
يازر البداية العلوية هي 300. و يازر البداية اليسرى هي 10.  
لوسمحت اعرض النافذة  
"زر1 النقر بالفارة" تعليماته هي  
انا اريد نافذة و النافذة تحت عنوان هو النتيجة.  
يانافذة الاتجاه هو بداية من اليمين لليساار.  
يانافذة لون الخلفية هو {255,0,0}.  
يانافذة البداية اليسرى هي 300.  
يانافذة البداية العلوية هي 100.  
يانافذة الارتفاع هو 250.  
انا اريد عنوان  
ياعنون النص هو " الاجابة خطأ "  
ياعنون اسم الخط هو arial. و ياعنون حجم الخط هو 54.  
ياعنون العرض هو 400. و ياعنون الارتفاع هو 100.  
ياعنون لون الخلفية هو {255,0,0}.  
ياعنون لون الكتابة هو {255,255,255}.  
انا اريد زر و يازر الاسم هو "زر4"  
يازر النص هو إغلاق.  
يازر اسم الخط هو arial. و يازر حجم الخط هو 14.

## تابع الشفرة المصدرية

يازر العرض هو 150.  
يازر البداية العلوية هي 150. و  
يازر البداية اليسرى هي 10.  
لوسحت اعرض النافذة

نهاية التعليمات

"زر2 النقر بالفارة" تعليماته هي

"زر1 النقر بالفارة" تعليماته واجب تنفيذها

نهاية التعليمات

"زر3 النقر بالفارة" تعليماته هي

انا اريد نافذة و النافذة تحت عنوان هو النتيجة.  
يانافذة الاتجاه هو بداية من اليمين لليسار.  
يانافذة لون الخلفية هو {0,255,0}.  
يانافذة البداية اليسرى هي 300.  
يانافذة البداية العلوية هي 100.  
يانافذة الارتفاع هو 250.

انا اريد عنوان

ياعنوان النص هو " الاجابة صحيحة "

ياعنوان اسم الخط هو arial. و يعنوان حجم الخط هو 54.

ياعنوان العرض هو 400. و يعنوان الارتفاع هو 100.

ياعنوان لون الخلفية هو {0,255,0}.

ياعنوان لون الكتابة هو {0,0,255}.

انا اريد زر و يازر الاسم هو "زر4"

يازر النص هو إغلاق.

يازر اسم الخط هو arial. و يازر حجم الخط هو 14.

يازر العرض هو 150.

يازر البداية العلوية هي 150. و يازر البداية اليسرى هي 10.

لوسحت اعرض النافذة

نهاية التعليمات


"زر4 النقر بالفارة" تعليماته هي لوسحت اغلق النافذة حسنا

والتطبيق وقت التشغيل كالتالى

حيث يتم عرض نافذة بها السؤال

عاصمة جمهورية مصر العربية ؟


- وهناك ثلاثة اختيارات متاحة هي
- مدينة الاسكندرية
  - مدينة سرس الليان – منوفية
  - مدينة القاهرة



في حالة اختيار مدينة الاسكندرية او مدينة سرس الليان – منوفية نحصل على



في حالة اختيار مدينة القاهرة نحصل على





## إستعمال تقنية COM Component Object Model

يمكن عمل امتداد لتطبيقات السوبرنوبا ايضا من خلال استخدام تقنية الـ COM والتي تفتح نافذة جديدة للتواصل مع العالم الخارجى بالنسبة لتطبيقات السوبرنوبا

والتعليمات التالية هي المتوفرة للتعامل مع تقنية الـ COM حيث تتيح لنا انشاء كائنات جديدة New Objects واستخدامها من خلال التعامل مع الخصائص باسناد قيم اليها ومعرفتها وايضا من خلال التعامل مع الطرق Methods باستعمالها وارسال المعطيات Parameters اليها.

- \* ....data as expression.... COM Object to create of type ....data as expression....
- \* ....data as expression.... COM Object to use
- \* ....data as expression.... COM Property to set ....data as expression....
- \* ....data as expression.... COM Property to get ....data as expression....
- \* ....data as expression.... COM Method to call with parameters ....data as value....

الامر COM Object to create of type يستخدم لانشاء كائن جديد باسم معين ونوع معين (اسم الفئة التي ينتمى اليها Class Name )  
الامر COM Object to use يحدد اسم الكائن الذى سوف نتعامل معه  
الامر COM Property to set يسند قيمة معينة الى خاصية من خواص الكائن  
الامر COM Property to get يخزن قيمة خاصية من خواص الكائن فى متغير  
الامر COM Method to call with parameters يستدعى طريقة من طرق الكائن باستخدام المعطيات المطلوب ارسالها الى هذه الطريقة لتستخدمه فى عمل ما هو مطلوب بالتحديد.



والان نرى مثالا جديدا يقوم بتشغيل برنامج الورد مع امكانية اخفائه و اظهاره.

```
I want window and the window title is COM Test.
I want button and button caption is Open Word Application. button width = 150.
button name = btn1.
I want button and button caption is Hide Word.
button name = btn2. and button left = 200.
I want button and button caption is Show Word.
button name = btn3. and button left = 350.
procedure btn1 mouse click.
    [myobj]. COM Object to create of type "(Word.Application)"
    [myobj]. COM Object to use
    "(Documents:Add)" COM method to call with parameters nothing.
    (visible). COM Property to set (1).
endprocedure
procedure btn2 mouse click.
    [myobj]. COM Object to use and (visible). COM Property to set (0).
endprocedure
procedure btn3 mouse click.
    [myobj]. COM Object to use and (visible). COM Property to set (1).
endprocedure
```

### والمثال بشفيرة مصدرية باللغة العربية كالتالي


انا اريد نافذة والنافذة تحت عنوان هو اختبار.  
انا اريد زر ويأزر النص هو افتح برنامج الورد. يأزر العرض = 150.  
يأزر الاسم هو "زر1"  
انا اريد زر  
يأزر النص هو "عدم اظهار برنامج الورد"  
يأزر الاسم هو "زر2" و يأزر البداية اليسرى هي 200.  
انا اريد زر ويأزر النص هو "اظهر برنامج الورد"  
يأزر الاسم هو "زر3" ويأزر البداية اليسرى هي 350.

تعريف الاجراء زر1 النقر بالفارة.  
[الورد]. كائن لمكون من النوع "(Word.Application)"  
[الورد]. كائن مكون لكي نستعمله  
"(Documents:Add)" طريقة مكون للاستدعاء بالمعطيات .  
(visible). خاصية مكون والقيمة تكون (1).  
نهاية الاجراء


تعريف الاجراء زر2 النقر بالفارة.  
[الورد]. كائن مكون لكي نستعمله و (visible). خاصية مكون والقيمة تكون (0).  
نهاية الاجراء

تعريف الاجراء زر3 النقر بالفارة.  
[الورد]. كائن مكون لكي نستعمله و (visible). خاصية مكون والقيمة تكون (1).  
نهاية الاجراء

والتطبيق وقت التشغيل كالتالى  
الصورة التالية عند تنفيذ الشفيرة المصدرية المكتوبة باللغة الانجليزية



والصورة التالية عند تنفيذ الشفيرة المصدرية المكتوبة باللغة العربية





والآن لدينا مثال يوضح استعمال ADO أى ActiveX Data Object من خلال تقنية الـ COM وذلك كمدخل لتطبيقات قواعد البيانات حيث يقوم المثال بفتح قاعدة بيانات Test.mdb والتعامل مع احد الجداول بالاضافة والحذف والاستعلام عن السجلات وكل ذلك على التوالى مع عرض النتائج.

والشفيرة المصدرية كالتالى

```
I want window contains LISTBOX and the window title is Using ADODB.  
Window width = 600. and WINDOW HEIGHT = 500.  
Window back color is white.
```

```
Listbox width = 350. and LISTBOX HEIGHT = 450.  
Listbox back color is silver.  
Listbox fore color is Black.  
Listbox font name is arial.  
Listbox font size is 14.
```

```
I want button and button caption is Test ADODB.  
Button name = btn1. and button LEFT = 400. and button width = 150.
```

```
procedure btn1 mouse click.
```

```
  :Note: " Connect to the Database Test.MDB"  
  [MYCON]. COM OBJECT TO CREATE OF TYPE "(ADODB.Connection)"  
  [MYCON]. COM Object to use  
  [MYARRAY]. = (1).  
  [MYARRAY](1)(1). = (string).  
  [MYARRAY](1)(2). = '(Provider=MSDASQL.1;Persist Security Info=False;Extended  
Properties="dsn=MS Access Database;uid=";Initial Catalog=test.mdb)'  
  "(OPEN)" COM method to call with parameters MYARRAY.
```

```
  :Note: " Open Table MainTable"  
  [MYRS]. COM OBJECT TO CREATE OF TYPE "(ADODB.Recordset)"  
  [MYRS]. COM Object to use  
  (CursorType). COM Property to set (3).  
  (LockType). COM Property to set (3).  
  (ActiveConnection). COM Property to set [MYCON].  
  [MYARRAY]. = (1).  
  [MYARRAY](1)(1). = (String).  
  [MYARRAY](1)(2). = '(Select * from MainTable)'  
  "(OPEN)" COM method to call with parameters MYARRAY.
```

```

:Note: " Delete All Records"
'(RECORDCOUNT)' COM PROPERTY TO GET [MYMAX].
doif [mymax] > (0).
    [X]. = (1).
    DoWhile [x] <= [MYMAX].
        "(movefirst)" COM method to call with parameters .
        "(delete)" COM method to call with parameters .
        "(update)" COM method to call with parameters .
        [x]. = [x] + (1).
    endwhile
endif

:Note: " Add 10 Records "
[X]. = (1).
DoWhile [x] <= (10).
    "(AddNew)" com method to call with parameters .
    '|fields:item("Number"):value|' COM PROPERTY TO SET [x].
    '|fields:item("Square"):value|' COM PROPERTY TO SET [x] x [x].
    "(UPDATE)" COM method to call with parameters .
    [x]. = [x] + (1).
EndWhile

:Note: " Show All Records "
"(movefirst)" COM method to call with parameters .
'(RECORDCOUNT)' COM PROPERTY TO GET [MYMAX].
listbox must add item -----
listbox must add from (Records Count = ) [MYMAX].
listbox must add item -----
listbox must add item Number                Square.
listbox must add item -----
[X]. = (1).
DoWhile [x] <= [MYMAX].
    '|fields:item("Number"):value|' COM PROPERTY TO GET [V1].
    '|fields:item("Square"):value|' COM PROPERTY TO GET [V2].
    Listbox must add from [v1] (                ) [v2].
    "(movenext)" COM method to call with parameters .
    [x]. = [x] + (1).
EndWhile

:Note: " Close Recordset"
"(CLOSE)" COM method to call with parameters .


:Note: " Close Connection"
[MYCON]. COM Object to use and "(CLOSE)" COM method to call with parameters .

endprocedure

```

والتطبيق وقت التشغيل كالتالى

حيث تم فتح قاعدة البيانات Test.MDB  
ثم تم الاستعلام عن سجلات الجدول MainTable  
من خلال الامر `Select * from MainTable` وهو احد اوامر لغة البحث  
والاستعلام SQL  
ثم بعد ذلك تم حذف جميع السجلات فى الجدول  
ثم اضافة 10 سجلات جديدة  
ثم تم عرض هذه السجلات التى تم اضافتها



Number	Square
1	1
2	4
3	9
4	16
5	25
6	36
7	49
8	64
9	81
10	100

وبهذا اكون انتهيت من توضيح كيفية استخدام تقنية الـ Component = COM فى Object Model فى السوبرنوفوا واستخدام هذه التقنية يفتح الباب لتطوير العديد من التطبيقات الجيدة باذن الله.



## تجربة العمل على Notepad++ لكتابة الشفرة المصدرية

يمكن استخدام المحرر Notepad++ لكتابة الشفرة المصدرية لتطبيقات السوبرنوفيا حيث ان من مميزات هذا البرنامج امكانية دعم لغات برمجة جديدة بسهولة وقد قمت بالفعل بدعم السوبرنوفيا في ساعات قليلة حيث ان الامر بسيط ويتم بشكل مرن من خلال واجهة المحرر.



في البداية عليك بتحميل Notepad++ ثم بعد ذلك تشغيل برنامج الاعداد واثناء الاعداد عليك بتفعيل الاختيار `Don't Use %APPDATA%` وبعد الانتهاء من الاعداد عليك بنسخ الملف `userDefineLang.XML` في المجلد الخاص بالمحرر في مجلد ملفات البرامج



### Program Files\Notepad++

وبهذا تحصل على امكانيات المحرر المتطورة مع لغة السوبرنوفيا وللتسهيل يمكن عمل اختصار لتشغيل برامج اللغة مثلا عند الضغط على `CTRL+R` وايضا يمكن تصدير الشفرة المصدرية الى ملفات `HTML` لنشرها على صفحات الويب او لوضعها بشكل ملون ومنسق في الكتب الالكترونية والمقالات.

ان المحرر رائع يتيح وضع لون مميز لكلمات اللغة وللعلامات والرموز وللتعليقات و يتيح عمل ضغط وفتح لاجزاء معينة من الشفرة المصدرية مثل الاجراءات وجمل التحكم والتكرار.











## رسالة من المؤلف

السلام عليكم ورحمة الله وبركاته

اخى الحبيب

اتمنى ان لا تتردد فى ان تتواصل معى لتبادل الخبرات والافكار  
هذا بريدى الالكترونى على الياهو

[msfclipper@yahoo.com](mailto:msfclipper@yahoo.com)

وهذه صفحتى الشخصية على الفيس بوك

<http://www.facebook.com/msfclipper>

وهذا منتدى تقنية البرمجة بدون كود ويشتمل على قسم للسوبرنوفيا ايضا

<http://pwct.p2h.info/vb>

وهذه مواقع انصحك بزيارتها

(موقع لاجبار تقنية المعلومات اول باول)

<http://www.itwadi.com>

منتديات الفريق العربى للبرمجة

<http://www.arabteam2000-forum.com>

منتدى فيجوال بيسك للعرب

<http://www.vb4arab.com/vb>

موقع كتب الحاسب العربية

<http://www.kutub.info>

موقع للمشاريع المفتوحة المصدر

<http://www.sourceforge.net>

موقع للمقالات التقنية

<http://www.codeproject.com>

تم بحمد الله فى 11 مايو عام 2010

